Rules for integrands of the form $(d+e\,x)^m\,(a+b\,x+c\,x^2)^p$

x: $\int (d+e\,x)^m\,(a+b\,x+c\,x^2)^p\,dx$ when $b^2-4\,a\,c=0\,\wedge\,p\in\mathbb{Z}$

- **Derivation: Algebraic simplification**
- **Basis: If** $b^2-4\,a\,c=0$, **then** $a+b\,x+c\,x^2=\frac{1}{c}\left(\frac{b}{2}+c\,x\right)^2$
- **Rule 1.2.1.2.2.1: If** $b^2-4\,a\,c=0\,\wedge\,p\in\mathbb{Z}$, **then**

$$\int (d+e\,x)^m\,(a+b\,x+c\,x^2)^p\,dx\;\longrightarrow\;\frac{1}{c^p}\int (d+e\,x)^m\left(\frac{b}{2}+c\,x\right)^{2p}dx$$

- **Program code:**

```
(* Int[(d_.+e_.*x_)^m_.*(a_+b_.*x_+c_.*x_^2)^p_.,x_Symbol] :=
  1/c^p*Int[(d+e*x)^m*(b/2+c*x)^(2*p),x] /;
FreeQ[{a,b,c,d,e,m},x] && EqQ[b^2-4*a*c,0] && IntegerQ[p] *)
```

0: $\int (d+e\,x)^m\,(a+b\,x+c\,x^2)^p\,dx$ when $b^2-4\,a\,c\neq 0\,\wedge\,c\,d^2-b\,d\,e+a\,e^2=0\,\wedge\,(p\in\mathbb{Z}\,\vee\,b=0\,\wedge\,a>0\,\wedge\,d>0\,\wedge\,m+p\in\mathbb{Z})$

- **Derivation: Algebraic simplification**
- **Basis: If** $c\,d^2-b\,d\,e+a\,e^2=0$, **then** $a+b\,x+c\,x^2=(d+e\,x)\left(\frac{a}{d}+\frac{c\,x}{e}\right)$
- **Basis: If** $c\,d^2+a\,e^2=0\,\wedge\,a>0\,\wedge\,d>0$, **then** $(a+c\,x^2)^p=\left(a-\frac{a\,e^2\,x^2}{d^2}\right)^p=(d+e\,x)^p\left(\frac{a}{d}+\frac{c\,x}{e}\right)^p$
- **Rule 1.2.1.2.3.1: If** $b^2-4\,a\,c\neq 0\,\wedge\,c\,d^2-b\,d\,e+a\,e^2=0\,\wedge\,(p\in\mathbb{Z}\,\vee\,b=0\,\wedge\,a>0\,\wedge\,d>0\,\wedge\,m+p\in\mathbb{Z})$, **then**

$$\int (d+e\,x)^m\,(a+b\,x+c\,x^2)^p\,dx\;\longrightarrow\;\int (d+e\,x)^{m+p}\left(\frac{a}{d}+\frac{c\,x}{e}\right)^p dx$$

- **Program code:**

```
Int[(d_+e_.*x_)^m_.*(a_.+b_.*x_+c_.*x_^2)^p_.,x_Symbol] :=
  Int[(d+e*x)^(m+p)*(a/d+c/e*x)^p,x] /;
FreeQ[{a,b,c,d,e,m},x] && NeQ[b^2-4*a*c,0] && EqQ[c*d^2-b*d*e+a*e^2,0] && IntegerQ[p]
```

```
Int[(d_+e_.*x_)^m_.*(a_+c_.*x_^2)^p_.,x_Symbol] :=
  Int[(d+e*x)^(m+p)*(a/d+c/e*x)^p,x] /;
FreeQ[{a,c,d,e,m,p},x] && EqQ[c*d^2+a*e^2,0] && (IntegerQ[p] || GtQ[a,0] && GtQ[d,0] && IntegerQ[m+p])
```

1. $\int (d+e\,x)\,\left(a+b\,x+c\,x^2\right)^p\,dx$

1. $\int (d+e\,x)\,\left(a+b\,x+c\,x^2\right)^p\,dx$ when $2\,c\,d-b\,e=0$

1: $\int \frac{d+e\,x}{a+b\,x+c\,x^2}\,dx$ when $2\,c\,d-b\,e=0$

Derivation: Integration by substitution

Basis: If $2\,c\,d-b\,e=0$, **then** $(d+e\,x)\,F\left[a+b\,x+c\,x^2\right]=\frac{d}{b}\,\text{Subst}\left[F[x],\,x,\,a+b\,x+c\,x^2\right]\,\partial_x\left(a+b\,x+c\,x^2\right)$

Rule 1.2.1.2.1.1.1: If $2\,c\,d-b\,e=0$, **then**

$$\int \frac{d+e\,x}{a+b\,x+c\,x^2}\,dx\,\longrightarrow\,\frac{d}{b}\,\text{Subst}\left[\int \frac{1}{x}\,dx,\,x,\,a+b\,x+c\,x^2\right]\,\longrightarrow\,\frac{d\,\text{Log}\left[a+b\,x+c\,x^2\right]}{b}$$

Program code:

```
Int[(d_+e_.*x_)/(a_.+b_.*x_+c_.*x_^2),x_Symbol] :=
  d*Log[RemoveContent[a+b*x+c*x^2,x]]/b /;
FreeQ[{a,b,c,d,e},x] && EqQ[2*c*d-b*e,0]
```

2: $\int (d+e\,x)\,\left(a+b\,x+c\,x^2\right)^p\,dx$ when $2\,c\,d-b\,e=0\,\wedge\,p\neq-1$

Derivation: Integration by substitution

Basis: If $2\,c\,d-b\,e=0$, **then** $(d+e\,x)\,F\left[a+b\,x+c\,x^2\right]=\frac{d}{b}\,\text{Subst}\left[F[x],\,x,\,a+b\,x+c\,x^2\right]\,\partial_x\left(a+b\,x+c\,x^2\right)$

Rule 1.2.1.2.1.1.2: If $2\,c\,d-b\,e=0\,\wedge\,p\neq-1$, **then**

$$\int (d+e\,x)\,\left(a+b\,x+c\,x^2\right)^p\,dx\,\longrightarrow\,\frac{d}{b}\,\text{Subst}\left[\int x^p\,dx,\,x,\,a+b\,x+c\,x^2\right]\,\longrightarrow\,\frac{d\,\left(a+b\,x+c\,x^2\right)^{p+1}}{b\,(p+1)}$$

Program code:

```
Int[(d_+e_.*x_)*(a_.+b_.*x_+c_.*x_^2)^p_.,x_Symbol] :=
  d*(a+b*x+c*x^2)^(p+1)/(b*(p+1)) /;
FreeQ[{a,b,c,d,e,p},x] && EqQ[2*c*d-b*e,0] && NeQ[p,-1]
```

2. $\int (d+e\,x)\,\left(a+b\,x+c\,x^2\right)^p\,dx$ when $2\,c\,d-b\,e\neq 0$

  1. $\int (d+e\,x)\,\left(a+b\,x+c\,x^2\right)^p\,dx$ when $2\,c\,d-b\,e\neq 0\,\wedge\,p\in\mathbb{Z}\,\wedge\,(p>0\,\vee\,a=0)$

    1: $\int (d+e\,x)\,\left(a+b\,x+c\,x^2\right)^p\,dx$ when $2\,c\,d-b\,e\neq 0\,\wedge\,p\in\mathbb{Z}^+\wedge\,c\,d^2-b\,d\,e+a\,e^2=0$

Derivation: Algebraic simplification

Basis: If $c\,d^2-b\,d\,e+a\,e^2=0$, **then** $a+b\,x+c\,x^2=(d+e\,x)\,\left(\frac{a}{d}+\frac{c\,x}{e}\right)$

Rule 1.2.1.2.1.2.1.1: If $2\,c\,d-b\,e\neq 0\,\wedge\,p\in\mathbb{Z}^+\wedge\,c\,d^2-b\,d\,e+a\,e^2=0$, **then**

$$\int (d+e\,x)\,\left(a+b\,x+c\,x^2\right)^p\,dx\,\longrightarrow\,\int (d+e\,x)^{p+1}\left(\frac{a}{d}+\frac{c\,x}{e}\right)^p\,dx$$

Program code:

```
Int[(d_+e_.*x_)*(a_.+b_.*x_+c_.*x_^2)^p_.,x_Symbol] :=
  Int[(d+e*x)^(p+1)*(a/d+c/e*x)^p,x] /;
FreeQ[{a,b,c,d,e},x] && NeQ[2*c*d-b*e,0] && IGtQ[p,0] && EqQ[c*d^2-b*d*e+a*e^2,0]
```

    2: $\int (d+e\,x)\,\left(a+b\,x+c\,x^2\right)^p\,dx$ when $2\,c\,d-b\,e\neq 0\,\wedge\,p\in\mathbb{Z}\,\wedge\,(p>0\,\vee\,a=0)$

Derivation: Algebraic expansion

Rule 1.2.1.2.1.2.1.2: If $2\,c\,d-b\,e\neq 0\,\wedge\,p\in\mathbb{Z}\,\wedge\,(p>0\,\vee\,a=0)$, **then**

$$\int (d+e\,x)\,\left(a+b\,x+c\,x^2\right)^p\,dx\,\longrightarrow\,\int \text{ExpandIntegrand}\left[(d+e\,x)\,\left(a+b\,x+c\,x^2\right)^p,\,x\right]\,dx$$

Program code:

```
Int[(d_.+e_.*x_)*(a_.+b_.*x_+c_.*x_^2)^p_.,x_Symbol] :=
  Int[ExpandIntegrand[(d+e*x)*(a+b*x+c*x^2)^p,x],x] /;
FreeQ[{a,b,c,d,e},x] && NeQ[2*c*d-b*e,0] && IntegerQ[p] && (GtQ[p,0] || EqQ[a,0])
```

2. $\int \frac{d+e\,x}{a+b\,x+c\,x^2}\,dx$ when $2\,c\,d-b\,e\neq 0\,\wedge\,b^2-4\,a\,c\neq 0$

1: $\int \frac{d+e\,x}{a+b\,x+c\,x^2}\,dx$ when $2\,c\,d-b\,e\neq 0\,\wedge\,b^2-4\,a\,c\neq 0\,\wedge\,\text{NiceSqrtQ}\left[b^2-4\,a\,c\right]$

Reference: G&R 2.161.1a & G&R 2.161.3

Derivation: Algebraic expansion

- **Basis: Let** $q=\sqrt{b^2-4\,a\,c}$**, then** $\frac{d+e\,x}{a+b\,x+c\,x^2}==\frac{c\,d-e\left(\frac{b}{2}-\frac{q}{2}\right)}{q\left(\frac{b}{2}-\frac{q}{2}+c\,x\right)}-\frac{c\,d-e\left(\frac{b}{2}+\frac{q}{2}\right)}{q\left(\frac{b}{2}+\frac{q}{2}+c\,x\right)}$

- **Rule 1.2.1.2.1.2.2.1: If** $2\,c\,d-b\,e\neq 0\,\wedge\,b^2-4\,a\,c\neq 0\,\wedge\,\text{NiceSqrtQ}\left[b^2-4\,a\,c\right]$**, let** $q\to\sqrt{b^2-4\,a\,c}$**, then**

$$\int \frac{d+e\,x}{a+b\,x+c\,x^2}\,dx\longrightarrow\frac{c\,d-e\left(\frac{b}{2}-\frac{q}{2}\right)}{q}\int\frac{1}{\frac{b}{2}-\frac{q}{2}+c\,x}\,dx-\frac{c\,d-e\left(\frac{b}{2}+\frac{q}{2}\right)}{q}\int\frac{1}{\frac{b}{2}+\frac{q}{2}+c\,x}\,dx$$

Program code:

```
Int[(d_.+e_.*x_)/(a_+b_.*x_+c_.*x_^2),x_Symbol] :=
  With[{q=Rt[b^2-4*a*c,2]},
  (c*d-e*(b/2-q/2))/q*Int[1/(b/2-q/2+c*x),x] - (c*d-e*(b/2+q/2))/q*Int[1/(b/2+q/2+c*x),x]] /;
FreeQ[{a,b,c,d,e},x] && NeQ[2*c*d-b*e,0] && NeQ[b^2-4*a*c,0] && NiceSqrtQ[b^2-4*a*c]
```

```
Int[(d_+e_.*x_)/(a_+c_.*x_^2),x_Symbol] :=
  With[{q=Rt[-a*c,2]},
  (e/2+c*d/(2*q))*Int[1/(-q+c*x),x] + (e/2-c*d/(2*q))*Int[1/(q+c*x),x]] /;
FreeQ[{a,c,d,e},x] && NiceSqrtQ[-a*c]
```

2: $\int \frac{d+e\,x}{a+b\,x+c\,x^2}\,dx$ when $2\,c\,d-b\,e\neq 0\,\wedge\,b^2-4\,a\,c\neq 0\,\wedge\,\neg\,\text{NiceSqrtQ}\left[b^2-4\,a\,c\right]$

Reference: A&S 3.3.19

Derivation: Algebraic expansion

- **Basis:** $\frac{d+e\,x}{a+b\,x+c\,x^2}==\left(d-\frac{b\,e}{2\,c}\right)\frac{1}{a+b\,x+c\,x^2}+\frac{e\,(b+2\,c\,x)}{2\,c\,(a+b\,x+c\,x^2)}$

Note: $\frac{b+2\,c\,x}{a+b\,x+c\,x^2}$ **is easily integrated using the rules for when** $2\,c\,d-b\,e==0$.

Rule 1.2.1.2.1.2.2.2: If $2\,c\,d-b\,e\neq 0\,\wedge\,b^2-4\,a\,c\neq 0\,\wedge\,\neg\,\text{NiceSqrtQ}\left[b^2-4\,a\,c\right]$**, then**

$$\int \frac{d+e\,x}{a+b\,x+c\,x^2}\,dx \longrightarrow \frac{2\,c\,d-b\,e}{2\,c}\int \frac{1}{a+b\,x+c\,x^2}\,dx+\frac{e}{2\,c}\int \frac{b+2\,c\,x}{a+b\,x+c\,x^2}\,dx$$

Program code:

```
Int[(d_.+e_.*x_)/(a_+b_.*x_+c_.*x_^2),x_Symbol] :=
(* (d-b*e/(2*c))*Int[1/(a+b*x+c*x^2),x] + *)
  (2*c*d-b*e)/(2*c)*Int[1/(a+b*x+c*x^2),x] + e/(2*c)*Int[(b+2*c*x)/(a+b*x+c*x^2),x] /;
FreeQ[{a,b,c,d,e},x] && NeQ[2*c*d-b*e,0] && NeQ[b^2-4*a*c,0] && Not[NiceSqrtQ[b^2-4*a*c]]
```

```
Int[(d_+e_.*x_)/(a_+c_.*x_^2),x_Symbol] :=
  d*Int[1/(a+c*x^2),x] + e*Int[x/(a+c*x^2),x] /;
FreeQ[{a,c,d,e},x] && Not[NiceSqrtQ[-a*c]]
```

3. $\int (d+e\,x)\left(a+b\,x+c\,x^2\right)^p\,dx$ when $2\,c\,d-b\,e\neq 0\,\wedge\,b^2-4\,a\,c\neq 0\,\wedge\,p<-1$

1: $\int \frac{d+e\,x}{\left(a+b\,x+c\,x^2\right)^{3/2}}\,dx$ when $2\,c\,d-b\,e\neq 0\,\wedge\,b^2-4\,a\,c\neq 0$

Derivation: Quadratic recurrence 2a

Rule 1.2.1.2.1.2.3.1: If $2\,c\,d-b\,e\neq 0\,\wedge\,b^2-4\,a\,c\neq 0$, then

$$\int \frac{d+e\,x}{\left(a+b\,x+c\,x^2\right)^{3/2}}\,dx \longrightarrow -\frac{2\,(b\,d-2\,a\,e+(2\,c\,d-b\,e)\,x)}{\left(b^2-4\,a\,c\right)\sqrt{a+b\,x+c\,x^2}}$$

Program code:

```
Int[(d_.+e_.*x_)/(a_.+b_.*x_+c_.*x_^2)^(3/2),x_Symbol] :=
  -2*(b*d-2*a*e+(2*c*d-b*e)*x)/((b^2-4*a*c)*Sqrt[a+b*x+c*x^2]) /;
FreeQ[{a,b,c,d,e},x] && NeQ[2*c*d-b*e,0] && NeQ[b^2-4*a*c,0]
```

```
Int[(d_+e_.*x_)/(a_+c_.*x_^2)^(3/2),x_Symbol] :=
  (-a*e+c*d*x)/(a*c*Sqrt[a+c*x^2]) /;
FreeQ[{a,c,d,e},x]
```

2: $\int (d+e\,x)\,\left(a+b\,x+c\,x^2\right)^p\,dx$ when $2\,c\,d-b\,e\neq 0\,\bigwedge\,b^2-4\,a\,c\neq 0\,\bigwedge\,p<-1\,\bigwedge\,p\neq -\frac{3}{2}$

Derivation: Quadratic recurrence 2a

Rule 1.2.1.2.1.2.3.2: If $2\,c\,d-b\,e\neq 0\,\bigwedge\,b^2-4\,a\,c\neq 0\,\bigwedge\,p<-1\,\bigwedge\,p\neq -\frac{3}{2}$, then

$$\int (d+e\,x)\,\left(a+b\,x+c\,x^2\right)^p\,dx\,\longrightarrow$$

$$\frac{(b\,d-2\,a\,e+(2\,c\,d-b\,e)\,x)\,\left(a+b\,x+c\,x^2\right)^{p+1}}{(p+1)\,\left(b^2-4\,a\,c\right)}-\frac{(2\,p+3)\,(2\,c\,d-b\,e)}{(p+1)\,\left(b^2-4\,a\,c\right)}\int\left(a+b\,x+c\,x^2\right)^{p+1}\,dx$$

Program code:

```
Int[(d_.+e_.*x_)*(a_.+b_.*x_+c_.*x_^2)^p_,x_Symbol] :=
  (b*d-2*a*e+(2*c*d-b*e)*x)/((p+1)*(b^2-4*a*c))*(a+b*x+c*x^2)^(p+1) -
  (2*p+3)*(2*c*d-b*e)/((p+1)*(b^2-4*a*c))*Int[(a+b*x+c*x^2)^(p+1),x] /;
FreeQ[{a,b,c,d,e},x] && NeQ[2*c*d-b*e,0] && NeQ[b^2-4*a*c,0] && LtQ[p,-1] && NeQ[p,-3/2]
```

```
Int[(d_+e_.*x_)*(a_+c_.*x_^2)^p_,x_Symbol] :=
  (a*e-c*d*x)/(2*a*c*(p+1))*(a+c*x^2)^(p+1) +
  d*(2*p+3)/(2*a*(p+1))*Int[(a+c*x^2)^(p+1),x] /;
FreeQ[{a,c,d,e},x] && LtQ[p,-1] && NeQ[p,-3/2]
```

4: $\int (d+e\,x)\,\left(a+b\,x+c\,x^2\right)^p\,dx$ when $2\,c\,d-b\,e\neq 0\,\wedge\,p\neq -1$

Reference: G&R 2.181.1, CRC 119

Derivation: Special quadratic recurrence 3a

Rule 1.2.1.2.1.2.4: If $2\,c\,d-b\,e\neq 0\,\wedge\,p\neq -1$, then

$$\int (d+e\,x)\,\left(a+b\,x+c\,x^2\right)^p\,dx\,\longrightarrow\,\frac{e\,\left(a+b\,x+c\,x^2\right)^{p+1}}{2\,c\,(p+1)}+\frac{2\,c\,d-b\,e}{2\,c}\int\left(a+b\,x+c\,x^2\right)^p\,dx$$

Program code:

```
Int[(d_.+e_.*x_)*(a_.+b_.*x_+c_.*x_^2)^p_,x_Symbol] :=
  e*(a+b*x+c*x^2)^(p+1)/(2*c*(p+1)) + (2*c*d-b*e)/(2*c)*Int[(a+b*x+c*x^2)^p,x] /;
FreeQ[{a,b,c,d,e,p},x] && NeQ[2*c*d-b*e,0] && NeQ[p,-1]
```

```
Int[(d_+e_.*x_)*(a_+c_.*x_^2)^p_.,x_Symbol] :=
  e*(a+c*x^2)^(p+1)/(2*c*(p+1)) + d*Int[(a+c*x^2)^p,x] /;
FreeQ[{a,c,d,e,p},x] && NeQ[p,-1]
```

2. $\int (d+e\,x)^m \left(a+b\,x+c\,x^2\right)^p\,dx$ when $b^2-4\,a\,c=0\,\land\,p\notin\mathbb{Z}$

1. $\int (d+e\,x)^m \left(a+b\,x+c\,x^2\right)^p\,dx$ when $b^2-4\,a\,c=0\,\land\,p\notin\mathbb{Z}\,\land\,2\,c\,d-b\,e=0$

1. $\int (d+e\,x)^m \left(a+b\,x+c\,x^2\right)^p\,dx$ when $b^2-4\,a\,c=0\,\land\,p\notin\mathbb{Z}\,\land\,2\,c\,d-b\,e=0\,\land\,m\in\mathbb{Z}$

1: $\int (d+e\,x)^m \left(a+b\,x+c\,x^2\right)^p\,dx$ when $b^2-4\,a\,c=0\,\bigwedge\,p\notin\mathbb{Z}\,\bigwedge\,2\,c\,d-b\,e=0\,\bigwedge\,\frac{m}{2}\in\mathbb{Z}$

Derivation: Algebraic simplification

- **Basis: If** $b^2-4\,a\,c=0\,\bigwedge\,2\,c\,d-b\,e=0\,\bigwedge\,\frac{m}{2}\in\mathbb{Z}$, **then** $(d+e\,x)^m \left(a+b\,x+c\,x^2\right)^p = \frac{e^m}{c^{m/2}} \left(a+b\,x+c\,x^2\right)^{p+\frac{m}{2}}$
- **Rule 1.2.1.2.2.2.1.1.1: If** $b^2-4\,a\,c=0\,\bigwedge\,p\notin\mathbb{Z}\,\bigwedge\,2\,c\,d-b\,e=0\,\bigwedge\,\frac{m}{2}\in\mathbb{Z}$, **then**

$$\int (d+e\,x)^m \left(a+b\,x+c\,x^2\right)^p\,dx \longrightarrow \frac{e^m}{c^{m/2}} \int \left(a+b\,x+c\,x^2\right)^{p+\frac{m}{2}}\,dx$$

Program code:

```
Int[(d_+e_.*x_)^m_*(a_+b_.*x_+c_.*x_^2)^p_,x_Symbol] :=
  e^m/c^(m/2)*Int[(a+b*x+c*x^2)^(p+m/2),x] /;
FreeQ[{a,b,c,d,e,p},x] && EqQ[b^2-4*a*c,0] && Not[IntegerQ[p]] && EqQ[2*c*d-b*e,0] && IntegerQ[m/2]
```

2: $\int (d+e\,x)^m \left(a+b\,x+c\,x^2\right)^p dx$ when $b^2-4\,a\,c==0 \bigwedge p\notin\mathbb{Z} \bigwedge 2\,c\,d-b\,e==0 \bigwedge \frac{m-1}{2}\in\mathbb{Z} \bigwedge m\neq 1$

Derivation: Algebraic simplification

- **Basis: If** $b^2-4\,a\,c==0 \bigwedge 2\,c\,d-b\,e==0 \bigwedge \frac{m-1}{2}\in\mathbb{Z}$**, then** $(d+e\,x)^m \left(a+b\,x+c\,x^2\right)^p == \frac{e^{m-1}}{c^{\frac{m-1}{2}}} (d+e\,x) \left(a+b\,x+c\,x^2\right)^{p+\frac{m-1}{2}}$
- **Rule 1.2.1.2.2.2.1.1.2: If** $b^2-4\,a\,c==0 \bigwedge p\notin\mathbb{Z} \bigwedge 2\,c\,d-b\,e==0 \bigwedge \frac{m-1}{2}\in\mathbb{Z} \bigwedge m\neq 1$**, then**

$$\int (d+e\,x)^m \left(a+b\,x+c\,x^2\right)^p dx \longrightarrow \frac{e^{m-1}}{c^{\frac{m-1}{2}}} \int (d+e\,x) \left(a+b\,x+c\,x^2\right)^{p+\frac{m-1}{2}} dx$$

Program code:

```
Int[(d_+e_.*x_)^m_*(a_+b_.*x_+c_.*x_^2)^p_,x_Symbol] :=
  e^(m-1)/c^((m-1)/2)*Int[(d+e*x)*(a+b*x+c*x^2)^(p+(m-1)/2),x] /;
FreeQ[{a,b,c,d,e,p},x] && EqQ[b^2-4*a*c,0] && Not[IntegerQ[p]] && EqQ[2*c*d-b*e,0] && IntegerQ[(m-1)/2]
```

2: $\int (d+e\,x)^m \left(a+b\,x+c\,x^2\right)^p dx$ when $b^2-4\,a\,c==0 \wedge p\notin\mathbb{Z} \wedge 2\,c\,d-b\,e==0 \wedge m\notin\mathbb{Z}$

Derivation: Piecewise constant extraction

- **Basis: If** $b^2-4\,a\,c==0 \wedge 2\,c\,d-b\,e==0$**, then** $\partial_x \frac{(a+b\,x+c\,x^2)^p}{(d+e\,x)^{2\,p}} == 0$
- **Rule 1.2.1.2.2.2.1.2: If** $b^2-4\,a\,c==0 \wedge p\notin\mathbb{Z} \wedge 2\,c\,d-b\,e==0 \wedge m\notin\mathbb{Z}$**, then**

$$\int (d+e\,x)^m \left(a+b\,x+c\,x^2\right)^p dx \longrightarrow \frac{\left(a+b\,x+c\,x^2\right)^p}{(d+e\,x)^{2\,p}} \int (d+e\,x)^{m+2\,p} dx$$

Program code:

```
Int[(d_+e_.*x_)^m_*(a_+b_.*x_+c_.*x_^2)^p_,x_Symbol] :=
  (a+b*x+c*x^2)^p/(d+e*x)^(2*p)*Int[(d+e*x)^(m+2*p),x] /;
FreeQ[{a,b,c,d,e,m,p},x] && EqQ[b^2-4*a*c,0] && Not[IntegerQ[p]] && EqQ[2*c*d-b*e,0] && Not[IntegerQ[m]]
```

2. $\int (d+e x)^m \left(a+b x+c x^2\right)^p \, dx$ when $b^2-4 a c = 0 \wedge p \notin \mathbb{Z} \wedge 2 c d-b e \neq 0$

1: $\int (d+e x)^m \left(a+b x+c x^2\right)^p \, dx$ when $b^2-4 a c = 0 \wedge p \notin \mathbb{Z} \wedge 2 c d-b e \neq 0 \wedge m \in \mathbb{Z}^+ \wedge m-2 p+1 = 0$

Derivation: Piecewise constant extraction and algebraic expansion

- **Basis:** If $b^2-4 a c = 0$, then $\partial_x \frac{\left(a+b x+c x^2\right)^p}{\left(\frac{b}{2}+c x\right)^{2 p}} = 0$

- **Rule 1.2.1.2.2.2.2.1:** If $b^2-4 a c = 0 \wedge p \notin \mathbb{Z} \wedge 2 c d-b e \neq 0 \wedge m \in \mathbb{Z}^+ \wedge m-2 p+1 = 0$, then

$$\int (d+e x)^m \left(a+b x+c x^2\right)^p \, dx \rightarrow \frac{\left(a+b x+c x^2\right)^{\text{FracPart}[p]}}{c^{\text{IntPart}[p]} \left(\frac{b}{2}+c x\right)^{2\,\text{FracPart}[p]}} \int \text{ExpandLinearProduct}\left[\left(\frac{b}{2}+c x\right)^{2 p}, (d+e x)^m, \frac{b}{2}, c, x\right] dx$$

Program code:

```
Int[(d_.+e_.*x_)^m_*(a_+b_.*x_+c_.*x_^2)^p_,x_Symbol] :=
  (a+b*x+c*x^2)^FracPart[p]/(c^IntPart[p]*(b/2+c*x)^(2*FracPart[p]))*
    Int[ExpandLinearProduct[(b/2+c*x)^(2*p),(d+e*x)^m,b/2,c,x],x] /;
FreeQ[{a,b,c,d,e,m,p},x] && EqQ[b^2-4*a*c,0] && Not[IntegerQ[p]] && NeQ[2*c*d-b*e,0] && IGtQ[m,0] && EqQ[m-2*p+1,0]
```

2: $\int (d+e x)^m \left(a+b x+c x^2\right)^p \, dx$ when $b^2-4 a c = 0 \wedge p \notin \mathbb{Z} \wedge 2 c d-b e \neq 0$

Derivation: Piecewise constant extraction

- **Basis:** If $b^2-4 a c = 0$, then $\partial_x \frac{\left(a+b x+c x^2\right)^p}{\left(\frac{b}{2}+c x\right)^{2 p}} = 0$

- **Rule 1.2.1.2.2.2.2.2:** If $b^2-4 a c = 0 \wedge p \notin \mathbb{Z} \wedge 2 c d-b e \neq 0$, then

$$\int (d+e x)^m \left(a+b x+c x^2\right)^p \, dx \rightarrow \frac{\left(a+b x+c x^2\right)^{\text{FracPart}[p]}}{c^{\text{IntPart}[p]} \left(\frac{b}{2}+c x\right)^{2\,\text{FracPart}[p]}} \int (d+e x)^m \left(\frac{b}{2}+c x\right)^{2 p} dx$$

Program code:

```
Int[(d_.+e_.*x_)^m_*(a_+b_.*x_+c_.*x_^2)^p_,x_Symbol] :=
  (a+b*x+c*x^2)^FracPart[p]/(c^IntPart[p]*(b/2+c*x)^(2*FracPart[p]))*Int[(d+e*x)^m*(b/2+c*x)^(2*p),x] /;
FreeQ[{a,b,c,d,e,m,p},x] && EqQ[b^2-4*a*c,0] && Not[IntegerQ[p]] && NeQ[2*c*d-b*e,0]
```

3. $\int (d+e\,x)^m \left(a+b\,x+c\,x^2\right)^p dx$ when $b^2-4\,a\,c\neq 0\,\land\,c\,d^2-b\,d\,e+a\,e^2==0$

0: $\int (e\,x)^m \left(b\,x+c\,x^2\right)^p dx$ when $p\in\mathbb{Z}$

Derivation: Algebraic simplification

Rule 1.2.1.2.3.0: If $p\in\mathbb{Z}$, then

$$\int (e\,x)^m \left(b\,x+c\,x^2\right)^p dx\;\longrightarrow\;\frac{1}{e^p}\int (e\,x)^{m+p}\,(b+c\,x)^p\,dx$$

Program code:

```
Int[(e_.*x_)^m_.*(b_.*x_+c_.*x_^2)^p_.,x_Symbol] :=
  1/e^p*Int[(e*x)^(m+p)*(b+c*x)^p,x] /;
FreeQ[{b,c,e,m},x] && IntegerQ[p]
```

1: $\int (d+e\,x)^m \left(a+b\,x+c\,x^2\right)^p dx$ when $b^2-4\,a\,c\neq 0\,\land\,c\,d^2-b\,d\,e+a\,e^2==0\,\land\,p\notin\mathbb{Z}\,\land\,m+p==0$

Reference: G&R 2.181.1, CRC 119 with $c\,d^2-b\,d\,e+a\,e^2==0\,\land\,m+p==0$

Derivation: Special quadratic recurrence 2a or 3a with $m+p==0$

Rule 1.2.1.2.3.2.1: If $b^2-4\,a\,c\neq 0\,\land\,c\,d^2-b\,d\,e+a\,e^2==0\,\land\,p\notin\mathbb{Z}\,\land\,m+p==0$, then

$$\int (d+e\,x)^m \left(a+b\,x+c\,x^2\right)^p dx\;\longrightarrow\;\frac{e\,(d+e\,x)^{m-1}\left(a+b\,x+c\,x^2\right)^{p+1}}{c\,(p+1)}$$

Program code:

```
Int[(d_.+e_.*x_)^m_*(a_.+b_.*x_+c_.*x_^2)^p_,x_Symbol] :=
  e*(d+e*x)^(m-1)*(a+b*x+c*x^2)^(p+1)/(c*(p+1)) /;
FreeQ[{a,b,c,d,e,m,p},x] && NeQ[b^2-4*a*c,0] && EqQ[c*d^2-b*d*e+a*e^2,0] && Not[IntegerQ[p]] && EqQ[m+p,0]
```

```
Int[(d_+e_.*x_)^m_*(a_+c_.*x_^2)^p_,x_Symbol] :=
  e*(d+e*x)^(m-1)*(a+c*x^2)^(p+1)/(c*(p+1)) /;
FreeQ[{a,c,d,e,m,p},x] && EqQ[c*d^2+a*e^2,0] && Not[IntegerQ[p]] && EqQ[m+p,0]
```

2: $\int (d+e\,x)^m \left(a+b\,x+c\,x^2\right)^p dx$ when $b^2-4\,a\,c\neq 0\,\land\,c\,d^2-b\,d\,e+a\,e^2==0\,\land\,p\notin\mathbb{Z}\,\land\,m+2\,p+2==0$

Reference: G&R 2.181.4.4

Derivation: Special quadratic recurrence 2b or 3b with $m+2\,p+2==0$

Note: If $m+2\,p+2==0$ and $m\neq 0$, then $p+1\neq 0$.

Rule 1.2.1.2.3.2.2: If $b^2-4\,a\,c\neq 0\,\land\,c\,d^2-b\,d\,e+a\,e^2==0\,\land\,p\notin\mathbb{Z}\,\land\,m+2\,p+2==0$, then

$$\int (d+e\,x)^m \left(a+b\,x+c\,x^2\right)^p dx \,\longrightarrow\, \frac{e\,(d+e\,x)^m \left(a+b\,x+c\,x^2\right)^{p+1}}{(p+1)\,(2\,c\,d-b\,e)}$$

Program code:

```
Int[(d_.+e_.*x_)^m_*(a_.+b_.*x_+c_.*x_^2)^p_,x_Symbol] :=
  e*(d+e*x)^m*(a+b*x+c*x^2)^(p+1)/((p+1)*(2*c*d-b*e)) /;
FreeQ[{a,b,c,d,e,m,p},x] && NeQ[b^2-4*a*c,0] && EqQ[c*d^2-b*d*e+a*e^2,0] && Not[IntegerQ[p]] && EqQ[m+2*p+2,0]
```

```
Int[(d_+e_.*x_)^m_*(a_+c_.*x_^2)^p_,x_Symbol] :=
  e*(d+e*x)^m*(a+c*x^2)^(p+1)/(2*c*d*(p+1)) /;
FreeQ[{a,c,d,e,m,p},x] && EqQ[c*d^2+a*e^2,0] && Not[IntegerQ[p]] && EqQ[m+2*p+2,0]
```

3: $\int (d+e\,x)^2 \left(a+b\,x+c\,x^2\right)^p dx$ when $b^2-4\,a\,c\neq 0\,\land\,c\,d^2-b\,d\,e+a\,e^2==0\,\land\,p\notin\mathbb{Z}\,\land\,p<-1$

Derivation: Special quadratic recurrence 2a

Rule 1.2.1.2.3.2.3: If $b^2-4\,a\,c\neq 0\,\land\,c\,d^2-b\,d\,e+a\,e^2==0\,\land\,p\notin\mathbb{Z}\,\land\,p<-1$, then

$$\int (d+e\,x)^2 \left(a+b\,x+c\,x^2\right)^p dx \,\longrightarrow\, \frac{e\,(d+e\,x)\,\left(a+b\,x+c\,x^2\right)^{p+1}}{c\,(p+1)} - \frac{e^2\,(p+2)}{c\,(p+1)} \int \left(a+b\,x+c\,x^2\right)^{p+1} dx$$

Program code:

```
Int[(d_.+e_.*x_)^2*(a_.+b_.*x_+c_.*x_^2)^p_,x_Symbol] :=
  e*(d+e*x)*(a+b*x+c*x^2)^(p+1)/(c*(p+1)) - e^2*(p+2)/(c*(p+1))*Int[(a+b*x+c*x^2)^(p+1),x] /;
FreeQ[{a,b,c,d,e,p},x] && NeQ[b^2-4*a*c,0] && EqQ[c*d^2-b*d*e+a*e^2,0] && Not[IntegerQ[p]] && LtQ[p,-1]
```

```
Int[(d_+e_.*x_)^2*(a_+c_.*x_^2)^p_,x_Symbol] :=
  e*(d+e*x)*(a+c*x^2)^(p+1)/(c*(p+1)) - e^2*(p+2)/(c*(p+1))*Int[(a+c*x^2)^(p+1),x] /;
FreeQ[{a,c,d,e,p},x] && EqQ[c*d^2+a*e^2,0] && Not[IntegerQ[p]] && LtQ[p,-1]
```

4: $\int (d+e\,x)^m \left(a+b\,x+c\,x^2\right)^p dx$ when $b^2-4\,a\,c\neq 0\,\wedge\,c\,d^2-b\,d\,e+a\,e^2=0\,\wedge\,p\notin\mathbb{Z}\,\wedge\,m\in\mathbb{Z}\,\wedge\,(0<-m<p\,\vee\,p<-m<0)$

Derivation: Algebraic simplification

- **Basis: If** $c\,d^2-b\,d\,e+a\,e^2=0$, **then** $d+e\,x=\frac{a+b\,x+c\,x^2}{\frac{a}{d}+\frac{c\,x}{e}}$

- **Basis: If** $c\,d^2+a\,e^2=0$, **then** $d+e\,x=\frac{d^2\,(a+c\,x^2)}{a\,(d-e\,x)}$

- **Rule 1.2.1.2.3.2.4: If** $b^2-4\,a\,c\neq 0\,\wedge\,c\,d^2-b\,d\,e+a\,e^2=0\,\wedge\,p\notin\mathbb{Z}\,\wedge\,m\in\mathbb{Z}\,\wedge\,(0<-m<p\,\vee\,p<-m<0)$, **then**

$$\int (d+e\,x)^m \left(a+b\,x+c\,x^2\right)^p dx \,\longrightarrow\, \int \frac{\left(a+b\,x+c\,x^2\right)^{m+p}}{\left(\frac{a}{d}+\frac{c\,x}{e}\right)^m}\,dx$$

Program code:

```
Int[(d_+e_.*x_)^m_*(a_.+b_.*x_+c_.*x_^2)^p_,x_Symbol] :=
  Int[(a+b*x+c*x^2)^(m+p)/(a/d+c*x/e)^m,x] /;
FreeQ[{a,b,c,d,e},x] && NeQ[b^2-4*a*c,0] && EqQ[c*d^2-b*d*e+a*e^2,0] && Not[IntegerQ[p]] && IntegerQ[m] &&
  RationalQ[p] && (LtQ[0,-m,p] || LtQ[p,-m,0]) && NeQ[m,2] && NeQ[m,-1]
```

```
Int[(d_+e_.*x_)^m_*(a_+c_.*x_^2)^p_,x_Symbol] :=
  d^(2*m)/a^m*Int[(a+c*x^2)^(m+p)/(d-e*x)^m,x] /;
FreeQ[{a,c,d,e,m,p},x] && EqQ[c*d^2+a*e^2,0] && Not[IntegerQ[p]] && IntegerQ[m] &&
  RationalQ[p] && (LtQ[0,-m,p] || LtQ[p,-m,0]) && NeQ[m,2] && NeQ[m,-1]
```

5: $\int (d+e\,x)^m \left(a+b\,x+c\,x^2\right)^p dx$ when $b^2-4\,a\,c\neq 0\,\wedge\,c\,d^2-b\,d\,e+a\,e^2=0\,\wedge\,p\notin\mathbb{Z}\,\wedge\,m+p\in\mathbb{Z}^+$

- **Reference: G&R 2.181.1, CRC 119 with** $a\,e^2-b\,d\,e+c\,d^2=0$

- **Derivation: Special quadratic recurrence 3a**

Note: If $p\notin\mathbb{Z}\,\wedge\,m+p\in\mathbb{Z}^+$, **then** $m+2\,p+1\neq 0$.

- **Rule 1.2.1.2.3.2.5: If** $b^2-4\,a\,c\neq 0\,\wedge\,c\,d^2-b\,d\,e+a\,e^2=0\,\wedge\,p\notin\mathbb{Z}\,\wedge\,m+p\in\mathbb{Z}^+$, **then**

$$\int (d+e\,x)^m \left(a+b\,x+c\,x^2\right)^p dx \longrightarrow$$

$$\frac{e\,(d+e\,x)^{m-1}\left(a+b\,x+c\,x^2\right)^{p+1}}{c\,(m+2\,p+1)}+\frac{(m+p)\,(2\,c\,d-b\,e)}{c\,(m+2\,p+1)}\int (d+e\,x)^{m-1}\left(a+b\,x+c\,x^2\right)^p dx$$

Program code:

```
Int[(d_.+e_.*x_)^m_*(a_.+b_.*x_+c_.*x_^2)^p_,x_Symbol] :=
  e*(d+e*x)^(m-1)*(a+b*x+c*x^2)^(p+1)/(c*(m+2*p+1)) +
  Simplify[m+p]*(2*c*d-b*e)/(c*(m+2*p+1))*Int[(d+e*x)^(m-1)*(a+b*x+c*x^2)^p,x] /;
FreeQ[{a,b,c,d,e,m,p},x] && NeQ[b^2-4*a*c,0] && EqQ[c*d^2-b*d*e+a*e^2,0] && Not[IntegerQ[p]] && IGtQ[Simplify[m+p],0]
```

```
Int[(d_+e_.*x_)^m_*(a_+c_.*x_^2)^p_,x_Symbol] :=
  e*(d+e*x)^(m-1)*(a+c*x^2)^(p+1)/(c*(m+2*p+1)) +
  2*c*d*Simplify[m+p]/(c*(m+2*p+1))*Int[(d+e*x)^(m-1)*(a+c*x^2)^p,x] /;
FreeQ[{a,c,d,e,m,p},x] && EqQ[c*d^2+a*e^2,0] && Not[IntegerQ[p]] && IGtQ[Simplify[m+p],0]
```

6: $\int (d+e\,x)^m \left(a+b\,x+c\,x^2\right)^p dx$ when $b^2-4\,a\,c\neq 0\,\wedge\,c\,d^2-b\,d\,e+a\,e^2=0\,\wedge\,p\notin\mathbb{Z}\,\wedge\,m+2\,p+2\in\mathbb{Z}^-$

Reference: G&R 2.181.4.4

Derivation: Special quadratic recurrence 3b

Note: If $b^2-4\,a\,c\neq 0\,\wedge\,c\,d^2-b\,d\,e+a\,e^2=0$, **then** $2\,c\,d-b\,e\neq 0$.

Note: If $p\notin\mathbb{Z}\,\wedge\,m+2\,p+2\in\mathbb{Z}^-$, **then** $m+p+1\neq 0$.

Rule 1.2.1.2.3.2.6: If $b^2-4\,a\,c\neq 0\,\wedge\,c\,d^2-b\,d\,e+a\,e^2=0\,\wedge\,p\notin\mathbb{Z}\,\wedge\,m+2\,p+2\in\mathbb{Z}^-$, **then**

$$\int (d+e\,x)^m \left(a+b\,x+c\,x^2\right)^p dx \longrightarrow$$

$$-\frac{e\,(d+e\,x)^m\left(a+b\,x+c\,x^2\right)^{p+1}}{(m+p+1)\,(2\,c\,d-b\,e)}+\frac{c\,(m+2\,p+2)}{(m+p+1)\,(2\,c\,d-b\,e)}\int (d+e\,x)^{m+1}\left(a+b\,x+c\,x^2\right)^p dx$$

Program code:

```
Int[(d_.+e_.*x_)^m_*(a_.+b_.*x_+c_.*x_^2)^p_,x_Symbol] :=
  -e*(d+e*x)^m*(a+b*x+c*x^2)^(p+1)/((m+p+1)*(2*c*d-b*e)) +
  c*Simplify[m+2*p+2]/((m+p+1)*(2*c*d-b*e))*Int[(d+e*x)^(m+1)*(a+b*x+c*x^2)^p,x] /;
FreeQ[{a,b,c,d,e,m,p},x] && NeQ[b^2-4*a*c,0] && EqQ[c*d^2-b*d*e+a*e^2,0] && Not[IntegerQ[p]] && ILtQ[Simplify[m+2*p+2],0]
```

```
Int[(d_+e_.*x_)^m_*(a_+c_.*x_^2)^p_,x_Symbol] :=
  -e*(d+e*x)^m*(a+c*x^2)^(p+1)/(2*c*d*(m+p+1)) +
  Simplify[m+2*p+2]/(2*d*(m+p+1))*Int[(d+e*x)^(m+1)*(a+c*x^2)^p,x] /;
FreeQ[{a,c,d,e,m,p},x] && EqQ[c*d^2+a*e^2,0] && Not[IntegerQ[p]] && ILtQ[Simplify[m+2*p+2],0]
```

7: $\int \frac{1}{\sqrt{d+e x}\,\sqrt{a+b x+c x^2}}\,dx$ when $b^2-4 a c \neq 0 \land c d^2-b d e+a e^2 == 0$

Derivation: Integration by substitution

- **Basis: If** $c d^2-b d e+a e^2 == 0$, **then** $\frac{1}{\sqrt{d+e x}\,\sqrt{a+b x+c x^2}} == 2 e \,\text{Subst}\left[\frac{1}{2 c d-b e+e^2 x^2},\, x,\, \frac{\sqrt{a+b x+c x^2}}{\sqrt{d+e x}}\right] \partial_x \frac{\sqrt{a+b x+c x^2}}{\sqrt{d+e x}}$

- **Rule 1.2.1.2.3.2.7: If** $b^2-4 a c \neq 0 \land c d^2-b d e+a e^2 == 0$, **then**

$$\int \frac{1}{\sqrt{d+e x}\,\sqrt{a+b x+c x^2}}\,dx \longrightarrow 2 e \,\text{Subst}\left[\int \frac{1}{2 c d-b e+e^2 x^2}\,dx,\, x,\, \frac{\sqrt{a+b x+c x^2}}{\sqrt{d+e x}}\right]$$

Program code:

```
Int[1/(Sqrt[d_.+e_.*x_]*Sqrt[a_.+b_.*x_+c_.*x_^2]),x_Symbol] :=
  2*e*Subst[Int[1/(2*c*d-b*e+e^2*x^2),x],x,Sqrt[a+b*x+c*x^2]/Sqrt[d+e*x]] /;
FreeQ[{a,b,c,d,e},x] && NeQ[b^2-4*a*c,0] && EqQ[c*d^2-b*d*e+a*e^2,0]
```

```
Int[1/(Sqrt[d_+e_.*x_]*Sqrt[a_+c_.*x_^2]),x_Symbol] :=
  2*e*Subst[Int[1/(2*c*d+e^2*x^2),x],x,Sqrt[a+c*x^2]/Sqrt[d+e*x]] /;
FreeQ[{a,c,d,e},x] && EqQ[c*d^2+a*e^2,0]
```

8. $\int (d+e\,x)^m\,\left(a+b\,x+c\,x^2\right)^p\,dx$ when $b^2-4\,a\,c\neq 0\,\land\,c\,d^2-b\,d\,e+a\,e^2==0\,\land\,p>0\,\land\,m<0$

1: $\int (d+e\,x)^m\,\left(a+b\,x+c\,x^2\right)^p\,dx$ when $b^2-4\,a\,c\neq 0\,\land\,c\,d^2-b\,d\,e+a\,e^2==0\,\land\,p>0\,\land\,(m<-2\,\lor\,m+2\,p+1==0)\,\land\,m+p+1\neq 0$

Reference: G&R 2.265b

Derivation: Special quadratic recurrence 1a

Rule 1.2.1.2.3.2.8.1: If $b^2-4\,a\,c\neq 0\,\land\,c\,d^2-b\,d\,e+a\,e^2==0\,\land\,p>0\,\land\,(m<-2\,\lor\,m+2\,p+1==0)\,\land\,m+p+1\neq 0$**, then**

$$\int (d+e\,x)^m\,\left(a+b\,x+c\,x^2\right)^p\,dx\,\longrightarrow\,\frac{(d+e\,x)^{m+1}\,\left(a+b\,x+c\,x^2\right)^p}{e\,(m+p+1)}-\frac{c\,p}{e^2\,(m+p+1)}\int (d+e\,x)^{m+2}\,\left(a+b\,x+c\,x^2\right)^{p-1}\,dx$$

Program code:

```
Int[(d_.+e_.*x_)^m_*(a_.+b_.*x_+c_.*x_^2)^p_,x_Symbol] :=
  (d+e*x)^(m+1)*(a+b*x+c*x^2)^p/(e*(m+p+1)) -
  c*p/(e^2*(m+p+1))*Int[(d+e*x)^(m+2)*(a+b*x+c*x^2)^(p-1),x] /;
FreeQ[{a,b,c,d,e},x] && NeQ[b^2-4*a*c,0] && EqQ[c*d^2-b*d*e+a*e^2,0] && GtQ[p,0] && (LtQ[m,-2] || EqQ[m+2*p+1,0]) && NeQ[m+p+1,0] &
```

```
Int[(d_+e_.*x_)^m_*(a_+c_.*x_^2)^p_,x_Symbol] :=
  (d+e*x)^(m+1)*(a+c*x^2)^p/(e*(m+p+1)) -
  c*p/(e^2*(m+p+1))*Int[(d+e*x)^(m+2)*(a+c*x^2)^(p-1),x] /;
FreeQ[{a,c,d,e},x] && EqQ[c*d^2+a*e^2,0] && GtQ[p,0] && (LtQ[m,-2] || EqQ[m+2*p+1,0]) && NeQ[m+p+1,0] && IntegerQ[2*p]
```

2: $\int (d+e\,x)^m \left(a+b\,x+c\,x^2\right)^p\,dx$ when $b^2-4\,a\,c\neq 0\,\wedge\,c\,d^2-b\,d\,e+a\,e^2==0\,\wedge\,p>0\,\wedge\,(-2\leq m<0\,\vee\,m+p+1==0)\,\wedge\,m+2\,p+1\neq 0$

Derivation: Special quadratic recurrence 1b

Rule 1.2.1.2.3.2.8.2: If $b^2-4\,a\,c\neq 0\,\wedge\,c\,d^2-b\,d\,e+a\,e^2==0\,\wedge\,p>0\,\wedge\,(-2\leq m<0\,\vee\,m+p+1==0)\,\wedge\,m+2\,p+1\neq 0$, then

$$\int (d+e\,x)^m \left(a+b\,x+c\,x^2\right)^p\,dx \longrightarrow$$

$$\frac{(d+e\,x)^{m+1}\left(a+b\,x+c\,x^2\right)^p}{e\,(m+2\,p+1)}-\frac{p\,(2\,c\,d-b\,e)}{e^2\,(m+2\,p+1)}\int (d+e\,x)^{m+1}\left(a+b\,x+c\,x^2\right)^{p-1}\,dx$$

Program code:

```
Int[(d_.+e_.*x_)^m_*(a_.+b_.*x_+c_.*x_^2)^p_,x_Symbol] :=
  (d+e*x)^(m+1)*(a+b*x+c*x^2)^p/(e*(m+2*p+1)) -
  p*(2*c*d-b*e)/(e^2*(m+2*p+1))*Int[(d+e*x)^(m+1)*(a+b*x+c*x^2)^(p-1),x] /;
FreeQ[{a,b,c,d,e},x] && NeQ[b^2-4*a*c,0] && EqQ[c*d^2-b*d*e+a*e^2,0] && GtQ[p,0] && (LeQ[-2,m,0] || EqQ[m+p+1,0]) && NeQ[m+2*p+1,0]
```

```
Int[(d_+e_.*x_)^m_*(a_+c_.*x_^2)^p_,x_Symbol] :=
  (d+e*x)^(m+1)*(a+c*x^2)^p/(e*(m+2*p+1)) -
  2*c*d*p/(e^2*(m+2*p+1))*Int[(d+e*x)^(m+1)*(a+c*x^2)^(p-1),x] /;
FreeQ[{a,c,d,e},x] && EqQ[c*d^2+a*e^2,0] && GtQ[p,0] && (LeQ[-2,m,0] || EqQ[m+p+1,0]) && NeQ[m+2*p+1,0] && IntegerQ[2*p]
```

9. $\int (d+e\,x)^m \left(a+b\,x+c\,x^2\right)^p dx$ when $b^2-4\,a\,c\neq 0\ \land\ c\,d^2-b\,d\,e+a\,e^2=0\ \land\ p<-1\ \land\ m>0$

1: $\int (d+e\,x)^m \left(a+b\,x+c\,x^2\right)^p dx$ when $b^2-4\,a\,c\neq 0\ \land\ c\,d^2-b\,d\,e+a\,e^2=0\ \land\ p<-1\ \land\ 0<m<1$

Derivation: Special quadratic recurrence 2b

Rule 1.2.1.2.3.2.9.1: If $b^2-4\,a\,c\neq 0\ \land\ c\,d^2-b\,d\,e+a\,e^2=0\ \land\ p<-1\ \land\ 0<m<1$**, then**

$$\int (d+e\,x)^m \left(a+b\,x+c\,x^2\right)^p dx\ \longrightarrow$$

$$\frac{(2\,c\,d-b\,e)\,(d+e\,x)^m \left(a+b\,x+c\,x^2\right)^{p+1}}{e\,(p+1)\left(b^2-4\,a\,c\right)}-\frac{(2\,c\,d-b\,e)\,(m+2\,p+2)}{(p+1)\left(b^2-4\,a\,c\right)}\int (d+e\,x)^{m-1}\left(a+b\,x+c\,x^2\right)^{p+1}dx$$

Program code:

```
Int[(d_.+e_.*x_)^m_*(a_.+b_.*x_+c_.*x_^2)^p_,x_Symbol] :=
  (2*c*d-b*e)*(d+e*x)^m*(a+b*x+c*x^2)^(p+1)/(e*(p+1)*(b^2-4*a*c)) -
  (2*c*d-b*e)*(m+2*p+2)/((p+1)*(b^2-4*a*c))*Int[(d+e*x)^(m-1)*(a+b*x+c*x^2)^(p+1),x] /;
FreeQ[{a,b,c,d,e},x] && NeQ[b^2-4*a*c,0] && EqQ[c*d^2-b*d*e+a*e^2,0] && LtQ[p,-1] && LtQ[0,m,1] && IntegerQ[2*p]
```

```
Int[(d_+e_.*x_)^m_*(a_+c_.*x_^2)^p_,x_Symbol] :=
  -d*(d+e*x)^m*(a+c*x^2)^(p+1)/(2*a*e*(p+1)) +
  d*(m+2*p+2)/(2*a*(p+1))*Int[(d+e*x)^(m-1)*(a+c*x^2)^(p+1),x] /;
FreeQ[{a,c,d,e},x] && EqQ[c*d^2+a*e^2,0] && LtQ[p,-1] && LtQ[0,m,1] && IntegerQ[2*p]
```

2: $\int (d+e\,x)^m \left(a+b\,x+c\,x^2\right)^p dx$ when $b^2-4\,a\,c\neq 0\,\wedge\,c\,d^2-b\,d\,e+a\,e^2==0\,\wedge\,p<-1\,\wedge\,m>1$

Derivation: Special quadratic recurrence 2a

Rule 1.2.1.2.2.3.9.2: If $b^2-4\,a\,c\neq 0\,\wedge\,c\,d^2-b\,d\,e+a\,e^2==0\,\wedge\,p<-1\,\wedge\,m>1$**, then**

$$\int (d+e\,x)^m \left(a+b\,x+c\,x^2\right)^p dx\,\longrightarrow$$

$$\frac{e\,(d+e\,x)^{m-1}\left(a+b\,x+c\,x^2\right)^{p+1}}{c\,(p+1)}-\frac{e^2\,(m+p)}{c\,(p+1)}\int (d+e\,x)^{m-2}\left(a+b\,x+c\,x^2\right)^{p+1}dx$$

Program code:

```
Int[(d_.+e_.*x_)^m_*(a_.+b_.*x_+c_.*x_^2)^p_,x_Symbol] :=
  e*(d+e*x)^(m-1)*(a+b*x+c*x^2)^(p+1)/(c*(p+1)) -
  e^2*(m+p)/(c*(p+1))*Int[(d+e*x)^(m-2)*(a+b*x+c*x^2)^(p+1),x] /;
FreeQ[{a,b,c,d,e},x] && NeQ[b^2-4*a*c,0] && EqQ[c*d^2-b*d*e+a*e^2,0] && LtQ[p,-1] && GtQ[m,1] && IntegerQ[2*p]
```

```
Int[(d_+e_.*x_)^m_*(a_+c_.*x_^2)^p_,x_Symbol] :=
  e*(d+e*x)^(m-1)*(a+c*x^2)^(p+1)/(c*(p+1)) -
  e^2*(m+p)/(c*(p+1))*Int[(d+e*x)^(m-2)*(a+c*x^2)^(p+1),x] /;
FreeQ[{a,c,d,e},x] && EqQ[c*d^2+a*e^2,0] && LtQ[p,-1] && GtQ[m,1] && IntegerQ[2*p]
```

10: $\int (d+e\,x)^m \left(a+b\,x+c\,x^2\right)^p \,dx$ when $b^2-4\,a\,c\neq 0 \land c\,d^2-b\,d\,e+a\,e^2==0 \land m>1 \land m+2\,p+1\neq 0$

Reference: G&R 2.181.1, CRC 119 with $a\,e^2-b\,d\,e+c\,d^2==0$

Derivation: Special quadratic recurrence 3a

Rule 1.2.1.2.3.2.10: If $b^2-4\,a\,c\neq 0 \land c\,d^2-b\,d\,e+a\,e^2==0 \land m>1 \land m+2\,p+1\neq 0$, then

$$\int (d+e\,x)^m \left(a+b\,x+c\,x^2\right)^p \,dx \longrightarrow$$

$$\frac{e\,(d+e\,x)^{m-1}\left(a+b\,x+c\,x^2\right)^{p+1}}{c\,(m+2\,p+1)}+\frac{(m+p)\,(2\,c\,d-b\,e)}{c\,(m+2\,p+1)}\int (d+e\,x)^{m-1}\left(a+b\,x+c\,x^2\right)^p \,dx$$

Program code:

```
Int[(d_.+e_.*x_)^m_*(a_.+b_.*x_+c_.*x_^2)^p_,x_Symbol] :=
  e*(d+e*x)^(m-1)*(a+b*x+c*x^2)^(p+1)/(c*(m+2*p+1)) +
  (m+p)*(2*c*d-b*e)/(c*(m+2*p+1))*Int[(d+e*x)^(m-1)*(a+b*x+c*x^2)^p,x] /;
FreeQ[{a,b,c,d,e,p},x] && NeQ[b^2-4*a*c,0] && EqQ[c*d^2-b*d*e+a*e^2,0] && GtQ[m,1] && NeQ[m+2*p+1,0] && IntegerQ[2*p]
```

```
Int[(d_+e_.*x_)^m_*(a_+c_.*x_^2)^p_,x_Symbol] :=
  e*(d+e*x)^(m-1)*(a+c*x^2)^(p+1)/(c*(m+2*p+1)) +
  2*c*d*(m+p)/(c*(m+2*p+1))*Int[(d+e*x)^(m-1)*(a+c*x^2)^p,x] /;
FreeQ[{a,c,d,e,p},x] && EqQ[c*d^2+a*e^2,0] && GtQ[m,1] && NeQ[m+2*p+1,0] && IntegerQ[2*p]
```

11: $\int (d+e\,x)^m \left(a+b\,x+c\,x^2\right)^p\,dx$ when $b^2-4\,a\,c\neq 0\,\land\,c\,d^2-b\,d\,e+a\,e^2==0\,\land\,m<0\,\land\,m+p+1\neq 0$

Reference: G&R 2.181.4.4

Derivation: Special quadratic recurrence 3b

Note: If $b^2-4\,a\,c\neq 0\,\land\,c\,d^2-b\,d\,e+a\,e^2==0$**, then** $2\,c\,d-b\,e\neq 0$

Rule 1.2.1.2.3.2.11: If $b^2-4\,a\,c\neq 0\,\land\,c\,d^2-b\,d\,e+a\,e^2==0\,\land\,m<0\,\land\,m+p+1\neq 0$**, then**

$$\int (d+e\,x)^m \left(a+b\,x+c\,x^2\right)^p\,dx \longrightarrow$$

$$-\frac{e\,(d+e\,x)^m \left(a+b\,x+c\,x^2\right)^{p+1}}{(m+p+1)\,(2\,c\,d-b\,e)}+\frac{c\,(m+2\,p+2)}{(m+p+1)\,(2\,c\,d-b\,e)}\int (d+e\,x)^{m+1} \left(a+b\,x+c\,x^2\right)^p\,dx$$

Program code:

```
Int[(d_.+e_.*x_)^m_*(a_.+b_.*x_+c_.*x_^2)^p_,x_Symbol] :=
  -e*(d+e*x)^m*(a+b*x+c*x^2)^(p+1)/((m+p+1)*(2*c*d-b*e)) +
  c*(m+2*p+2)/((m+p+1)*(2*c*d-b*e))*Int[(d+e*x)^(m+1)*(a+b*x+c*x^2)^p,x] /;
FreeQ[{a,b,c,d,e,p},x] && NeQ[b^2-4*a*c,0] && EqQ[c*d^2-b*d*e+a*e^2,0] && LtQ[m,0] && NeQ[m+p+1,0] && IntegerQ[2*p]
```

```
Int[(d_+e_.*x_)^m_*(a_+c_.*x_^2)^p_,x_Symbol] :=
  -e*(d+e*x)^m*(a+c*x^2)^(p+1)/(2*c*d*(m+p+1)) +
  (m+2*p+2)/(2*d*(m+p+1))*Int[(d+e*x)^(m+1)*(a+c*x^2)^p,x] /;
FreeQ[{a,c,d,e,p},x] && EqQ[c*d^2+a*e^2,0] && LtQ[m,0] && NeQ[m+p+1,0] && IntegerQ[2*p]
```

12. $\int (d+e\,x)^m\,\left(a+b\,x+c\,x^2\right)^p\,dx$ when $b^2-4\,a\,c\neq 0\,\wedge\,c\,d^2-b\,d\,e+a\,e^2==0\,\wedge\,p\notin\mathbb{Z}$

1: $\int (e\,x)^m\,\left(b\,x+c\,x^2\right)^p\,dx$ when $p\notin\mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{(e\,x)^m\,(b\,x+c\,x^2)^p}{x^{m+p}\,(b+c\,x)^p}==0$

Rule 1.2.1.2.3.2.12.1: If $p\notin\mathbb{Z}$, then

$$\int (e\,x)^m\,\left(b\,x+c\,x^2\right)^p\,dx\;\longrightarrow\;\frac{(e\,x)^m\,\left(b\,x+c\,x^2\right)^p}{x^{m+p}\,(b+c\,x)^p}\int x^{m+p}\,(b+c\,x)^p\,dx$$

Program code:

```
Int[(e_.*x_)^m_*(b_.*x_+c_.*x_^2)^p_,x_Symbol] :=
  (e*x)^m*(b*x+c*x^2)^p/(x^(m+p)*(b+c*x)^p)*Int[x^(m+p)*(b+c*x)^p,x] /;
FreeQ[{b,c,e,m},x] && Not[IntegerQ[p]]
```

??2: $\int (d+e\,x)^m\,\left(a+c\,x^2\right)^p\,dx$ when $c\,d^2+a\,e^2==0\,\wedge\,p\notin\mathbb{Z}\,\wedge\,a>0\,\wedge\,d>0$

Derivation: Algebraic simplification

Basis: If $c\,d^2+a\,e^2==0\,\wedge\,a>0\,\wedge\,d>0$, then $\left(a+c\,x^2\right)^p==\left(a-\frac{a\,e^2\,x^2}{d^2}\right)^p==(d+e\,x)^p\,\left(\frac{a}{d}+\frac{c\,x}{e}\right)^p$

Rule 1.2.1.2.3.2.12.2: If $c\,d^2+a\,e^2==0\,\wedge\,p\notin\mathbb{Z}\,\wedge\,a>0\,\wedge\,d>0$, then

$$\int (d+e\,x)^m\,\left(a+c\,x^2\right)^p\,dx\;\longrightarrow\;\int (d+e\,x)^{m+p}\,\left(\frac{a}{d}+\frac{c\,x}{e}\right)^p\,dx$$

Program code:

```
Int[(d_+e_.*x_)^m_*(a_+c_.*x_^2)^p_,x_Symbol] :=
  Int[(d+e*x)^(m+p)*(a/d+c/e*x)^p,x] /;
FreeQ[{a,c,d,e,m,p},x] && EqQ[c*d^2+a*e^2,0] && Not[IntegerQ[p]] && GtQ[a,0] && GtQ[d,0] && Not[IGtQ[m,0]]
```

?. $\int (d+e\,x)^m \left(a+b\,x+c\,x^2\right)^p dx$ when $b^2-4\,a\,c\neq 0\,\land\,c\,d^2-b\,d\,e+a\,e^2==0\,\land\,p\notin\mathbb{Z}\,\land\,(m\in\mathbb{Z}\,\lor\,d>0)$

1: $\int (d+e\,x)^m \left(a+c\,x^2\right)^p dx$ when $c\,d^2+a\,e^2==0\,\land\,p\notin\mathbb{Z}\,\land\,(m\in\mathbb{Z}\,\lor\,d>0)\,\land\,a>0$

Derivation: Piecewise constant extraction

- **Basis: If** $c\,d^2+a\,e^2==0$, **then** $\partial_x \frac{\left(a+c\,x^2\right)^{p+1}}{\left(1+\frac{e\,x}{d}\right)^{p+1}\left(\frac{a}{d}+\frac{c\,x}{e}\right)^{p+1}}==0$
- **Basis: If** $c\,d^2+a\,e^2==0\,\land\,a>0$, **then** $\frac{\left(a+c\,x^2\right)^{p+1}}{\left(1+\frac{e\,x}{d}\right)^{p+1}}==a^{p+1}\left(\frac{d-e\,x}{d}\right)^{p+1}$
- **Note: If** $c\,d^2-b\,d\,e+a\,e^2==0\,\land\,m\in\mathbb{Z}^+\land\,(3\,p\in\mathbb{Z}\,\lor\,4\,p\in\mathbb{Z})$, **then** $(d+e\,x)^m\left(a+b\,x+c\,x^2\right)^p$ **is integrable in terms of non-hypergeometric functions.**
- **Rule 1.2.1.2.3.2.12.3: If** $c\,d^2+a\,e^2==0\,\land\,p\notin\mathbb{Z}\,\land\,(m\in\mathbb{Z}\,\lor\,d>0)\,\land\,a>0$, **then**

$$\int (d+e\,x)^m\left(a+c\,x^2\right)^p dx\,\longrightarrow\,\frac{d^{m-1}\left(a+c\,x^2\right)^{p+1}}{\left(1+\frac{e\,x}{d}\right)^{p+1}\left(\frac{a}{d}+\frac{c\,x}{e}\right)^{p+1}}\int\left(1+\frac{e\,x}{d}\right)^{m+p}\left(\frac{a}{d}+\frac{c\,x}{e}\right)^p dx$$

$$\longrightarrow\,\frac{a^{p+1}\,d^{m-1}\left(\frac{d-e\,x}{d}\right)^{p+1}}{\left(\frac{a}{d}+\frac{c\,x}{e}\right)^{p+1}}\int\left(1+\frac{e\,x}{d}\right)^{m+p}\left(\frac{a}{d}+\frac{c\,x}{e}\right)^p dx$$

Program code:

```
Int[(d_+e_.*x_)^m_*(a_+c_.*x_^2)^p_,x_Symbol] :=
  a^(p+1)*d^(m-1)*((d-e*x)/d)^(p+1)/(a/d+c*x/e)^(p+1)*Int[(1+e*x/d)^(m+p)*(a/d+c/e*x)^p,x] /;
FreeQ[{a,c,d,e,m},x] && EqQ[c*d^2+a*e^2,0] && Not[IntegerQ[p]] && (IntegerQ[m] || GtQ[d,0]) && GtQ[a,0] &&
  Not[IGtQ[m,0] && (IntegerQ[3*p] || IntegerQ[4*p])]
```

2: $\int (d+e\,x)^m \left(a+b\,x+c\,x^2\right)^p dx$ when $b^2-4\,a\,c\neq 0\,\land\,c\,d^2-b\,d\,e+a\,e^2==0\,\land\,p\notin\mathbb{Z}\,\land\,(m\in\mathbb{Z}\,\lor\,d>0)$

Derivation: Piecewise constant extraction

- **Basis: If** $c\,d^2-b\,d\,e+a\,e^2==0$, **then** $\partial_x \frac{\left(a+b\,x+c\,x^2\right)^{p}}{\left(1+\frac{e\,x}{d}\right)^{p}\left(\frac{a}{d}+\frac{c\,x}{e}\right)^{p}}==0$
- **Note: If** $c\,d^2-b\,d\,e+a\,e^2==0\,\land\,m\in\mathbb{Z}^+\land\,(3\,p\in\mathbb{Z}\,\lor\,4\,p\in\mathbb{Z})$, **then** $(d+e\,x)^m\left(a+b\,x+c\,x^2\right)^p$ **is integrable in terms of non-hypergeometric functions.**
- **Rule 1.2.1.2.3.2.12.3: If** $b^2-4\,a\,c\neq 0\,\land\,c\,d^2-b\,d\,e+a\,e^2==0\,\land\,p\notin\mathbb{Z}\,\land\,(m\in\mathbb{Z}\,\lor\,d>0)$, **then**

$$\int (d+e\,x)^m \left(a+b\,x+c\,x^2\right)^p dx \longrightarrow \frac{d^m \left(a+b\,x+c\,x^2\right)^{\text{FracPart}[p]}}{\left(1+\frac{e\,x}{d}\right)^{\text{FracPart}[p]} \left(\frac{a}{d}+\frac{c\,x}{e}\right)^{\text{FracPart}[p]}} \int \left(1+\frac{e\,x}{d}\right)^{m+p} \left(\frac{a}{d}+\frac{c\,x}{e}\right)^p dx$$

Program code:

```
Int[(d_+e_.*x_)^m_*(a_.+b_.*x_+c_.*x_^2)^p_,x_Symbol] :=
  d^m*(a+b*x+c*x^2)^FracPart[p]/((1+e*x/d)^FracPart[p]*(a/d+(c*x)/e)^FracPart[p])*Int[(1+e*x/d)^(m+p)*(a/d+c/e*x)^p,x] /;
FreeQ[{a,b,c,d,e,m},x] && NeQ[b^2-4*a*c,0] && EqQ[c*d^2-b*d*e+a*e^2,0] && Not[IntegerQ[p]] && (IntegerQ[m] || GtQ[d,0]) &&
  Not[IGtQ[m,0] && (IntegerQ[3*p] || IntegerQ[4*p])]
```

```
Int[(d_+e_.*x_)^m_*(a_+c_.*x_^2)^p_,x_Symbol] :=
  d^(m-1)*(a+c*x^2)^(p+1)/((1+e*x/d)^(p+1)*(a/d+(c*x)/e)^(p+1))*Int[(1+e*x/d)^(m+p)*(a/d+c/e*x)^p,x] /;
FreeQ[{a,c,d,e,m},x] && EqQ[c*d^2+a*e^2,0] && Not[IntegerQ[p]] && (IntegerQ[m] || GtQ[d,0]) && Not[IGtQ[m,0] && (IntegerQ[3*p] || I
```

3: $\int (d+e\,x)^m \left(a+b\,x+c\,x^2\right)^p dx$ when $b^2-4\,a\,c\neq 0\,\land\,c\,d^2-b\,d\,e+a\,e^2==0\,\land\,p\notin\mathbb{Z}\,\land\,\neg\,(m\in\mathbb{Z}\,\lor\,d>0)$

Derivation: Piecewise constant extraction

- **Basis:** $\partial_x \frac{(d+e\,x)^m}{\left(1+\frac{e\,x}{d}\right)^m} == 0$
- **Rule 1.2.1.2.3.2.12.3: If** $b^2-4\,a\,c\neq 0\,\land\,c\,d^2-b\,d\,e+a\,e^2==0\,\land\,p\notin\mathbb{Z}\,\land\,\neg\,(m\in\mathbb{Z}\,\lor\,d>0)$**, then**

$$\int (d+e\,x)^m \left(a+b\,x+c\,x^2\right)^p dx \longrightarrow \frac{d^{\text{IntPart}[m]}\,(d+e\,x)^{\text{FracPart}[m]}}{\left(1+\frac{e\,x}{d}\right)^{\text{FracPart}[m]}} \int \left(1+\frac{e\,x}{d}\right)^m \left(a+b\,x+c\,x^2\right)^p dx$$

- **Program code:**

```
Int[(d_+e_.*x_)^m_*(a_.+b_.*x_+c_.*x_^2)^p_,x_Symbol] :=
  d^IntPart[m]*(d+e*x)^FracPart[m]/(1+e*x/d)^FracPart[m]*Int[(1+e*x/d)^m*(a+b*x+c*x^2)^p,x] /;
FreeQ[{a,b,c,d,e,m},x] && NeQ[b^2-4*a*c,0] && EqQ[c*d^2-b*d*e+a*e^2,0] && Not[IntegerQ[p]] && Not[IntegerQ[m] || GtQ[d,0]]
```

```
Int[(d_+e_.*x_)^m_*(a_+c_.*x_^2)^p_,x_Symbol] :=
  d^IntPart[m]*(d+e*x)^FracPart[m]/(1+e*x/d)^FracPart[m]*Int[(1+e*x/d)^m*(a+c*x^2)^p,x] /;
FreeQ[{a,c,d,e,m},x] && EqQ[c*d^2+a*e^2,0] && Not[IntegerQ[p]] && Not[IntegerQ[m] || GtQ[d,0]]
```

4. $\int (d+e\,x)^m \left(a+b\,x+c\,x^2\right)^p dx$ when $b^2-4\,a\,c\neq 0\,\land\,2\,c\,d-b\,e==0$

  1. $\int (d+e\,x)^m \left(a+b\,x+c\,x^2\right)^p dx$ when $b^2-4\,a\,c\neq 0\,\land\,2\,c\,d-b\,e==0\,\land\,m+2\,p+3==0$

1: $\int \frac{1}{(d+e\,x)\,\left(a+b\,x+c\,x^2\right)}\,dx$ when $b^2-4\,a\,c\neq 0\,\wedge\,2\,c\,d-b\,e==0$

Derivation: Algebraic expansion

Basis: If $2\,c\,d-b\,e==0$**, then** $\frac{1}{(d+e\,x)\,(a+b\,x+c\,x^2)}==-\frac{4\,b\,c}{d\,(b^2-4\,a\,c)\,(b+2\,c\,x)}+\frac{b^2\,(d+e\,x)}{d^2\,(b^2-4\,a\,c)\,(a+b\,x+c\,x^2)}$

Rule 1.2.1.2.3.1.1: If $b^2-4\,a\,c\neq 0\,\wedge\,2\,c\,d-b\,e==0$**, then**

$$\int \frac{1}{(d+e\,x)\,\left(a+b\,x+c\,x^2\right)}\,dx\,\longrightarrow\,-\frac{4\,b\,c}{d\,\left(b^2-4\,a\,c\right)}\int \frac{1}{b+2\,c\,x}\,dx+\frac{b^2}{d^2\,\left(b^2-4\,a\,c\right)}\int \frac{d+e\,x}{a+b\,x+c\,x^2}\,dx$$

Program code:

```
Int[1/((d_+e_.*x_)*(a_.+b_.*x_+c_.*x_^2)),x_Symbol] :=
  -4*b*c/(d*(b^2-4*a*c))*Int[1/(b+2*c*x),x] +
  b^2/(d^2*(b^2-4*a*c))*Int[(d+e*x)/(a+b*x+c*x^2),x] /;
FreeQ[{a,b,c,d,e},x] && NeQ[b^2-4*a*c,0] && EqQ[2*c*d-b*e,0]
```

2: $\int (d+e\,x)^m\,\left(a+b\,x+c\,x^2\right)^p\,dx$ when $b^2-4\,a\,c\neq 0\,\wedge\,2\,c\,d-b\,e==0\,\wedge\,m+2\,p+3==0\,\wedge\,p\neq -1$

Derivation: Derivative divides quadratic recurrence 2b or 3b with $m+2\,p+3==0$

Rule 1.2.1.2.3.1.2: If $b^2-4\,a\,c\neq 0\,\wedge\,2\,c\,d-b\,e==0\,\wedge\,m+2\,p+3==0\,\wedge\,p\neq -1$**, then**

$$\int (d+e\,x)^m\,\left(a+b\,x+c\,x^2\right)^p\,dx\,\longrightarrow\,\frac{2\,c\,(d+e\,x)^{m+1}\,\left(a+b\,x+c\,x^2\right)^{p+1}}{e\,(p+1)\,\left(b^2-4\,a\,c\right)}$$

Program code:

```
Int[(d_+e_.*x_)^m_*(a_.+b_.*x_+c_.*x_^2)^p_.,x_Symbol] :=
  2*c*(d+e*x)^(m+1)*(a+b*x+c*x^2)^(p+1)/(e*(p+1)*(b^2-4*a*c)) /;
FreeQ[{a,b,c,d,e,m,p},x] && NeQ[b^2-4*a*c,0] && EqQ[2*c*d-b*e,0] && EqQ[m+2*p+3,0] && NeQ[p,-1]
```

2: $\int (d+e\,x)^m\,(a+b\,x+c\,x^2)^p\,dx$ when $b^2-4\,a\,c\neq 0\ \wedge\ 2\,c\,d-b\,e==0\ \wedge\ p\in\mathbb{Z}^+$

Derivation: Algebraic expansion

Rule 1.2.1.2.3.2: If $b^2-4\,a\,c\neq 0\ \wedge\ 2\,c\,d-b\,e==0\ \wedge\ p\in\mathbb{Z}^+$**, then**

$$\int (d+e\,x)^m\,(a+b\,x+c\,x^2)^p\,dx\ \longrightarrow\ \int \text{ExpandIntegrand}\left[(d+e\,x)^m\,(a+b\,x+c\,x^2)^p,\ x\right]dx$$

Program code:

```
Int[(d_.+e_.*x_)^m_*(a_.+b_.*x_+c_.*x_^2)^p_.,x_Symbol] :=
  Int[ExpandIntegrand[(d+e*x)^m*(a+b*x+c*x^2)^p,x],x] /;
FreeQ[{a,b,c,d,e,m},x] && NeQ[b^2-4*a*c,0] && EqQ[2*c*d-b*e,0] && IGtQ[p,0] && Not[EqQ[m,3] && NeQ[p,1]]
```

3. $\int (d+e\,x)^m\,(a+b\,x+c\,x^2)^p\,dx$ when $b^2-4\,a\,c\neq 0\ \wedge\ 2\,c\,d-b\,e==0\ \wedge\ m+2\,p+3\neq 0\ \wedge\ p>0$

1: $\int (d+e\,x)^m\,(a+b\,x+c\,x^2)^p\,dx$ when $b^2-4\,a\,c\neq 0\ \wedge\ 2\,c\,d-b\,e==0\ \wedge\ m+2\,p+3\neq 0\ \wedge\ p>0\ \wedge\ m<-1$

Derivation: Derivative divides quadratic recurrence 1a

Derivation: Inverted integration by parts

Rule 1.2.1.2.3.3.1: If $b^2-4\,a\,c\neq 0\ \wedge\ 2\,c\,d-b\,e==0\ \wedge\ m+2\,p+3\neq 0\ \wedge\ p>0\ \wedge\ m<-1$**, then**

$$\int (d+e\,x)^m\,(a+b\,x+c\,x^2)^p\,dx\ \longrightarrow\ \frac{(d+e\,x)^{m+1}\,(a+b\,x+c\,x^2)^p}{e\,(m+1)}-\frac{b\,p}{d\,e\,(m+1)}\int (d+e\,x)^{m+2}\,(a+b\,x+c\,x^2)^{p-1}\,dx$$

Program code:

```
Int[(d_+e_.*x_)^m_*(a_.+b_.*x_+c_.*x_^2)^p_.,x_Symbol] :=
  (d+e*x)^(m+1)*(a+b*x+c*x^2)^p/(e*(m+1)) -
  b*p/(d*e*(m+1))*Int[(d+e*x)^(m+2)*(a+b*x+c*x^2)^(p-1),x] /;
FreeQ[{a,b,c,d,e},x] && NeQ[b^2-4*a*c,0] && EqQ[2*c*d-b*e,0] && NeQ[m+2*p+3,0] && GtQ[p,0] && LtQ[m,-1] &&
  Not[IntegerQ[m/2] && LtQ[m+2*p+3,0]] && IntegerQ[2*p]
```

2: $\int (d+e\,x)^m \left(a+b\,x+c\,x^2\right)^p dx$ when $b^2-4\,a\,c \neq 0 \wedge 2\,c\,d-b\,e == 0 \wedge m+2\,p+3 \neq 0 \wedge p>0 \wedge m \not< -1$

Derivation: Derivative divides quadratic recurrence 1b

Rule 1.2.1.2.3.3.2: If $b^2-4\,a\,c \neq 0 \wedge 2\,c\,d-b\,e == 0 \wedge m+2\,p+3 \neq 0 \wedge p>0 \wedge m \not< -1$**, then**

$$\int (d+e\,x)^m \left(a+b\,x+c\,x^2\right)^p dx \longrightarrow \frac{(d+e\,x)^{m+1} \left(a+b\,x+c\,x^2\right)^p}{e\,(m+2\,p+1)} - \frac{d\,p\,\left(b^2-4\,a\,c\right)}{b\,e\,(m+2\,p+1)} \int (d+e\,x)^m \left(a+b\,x+c\,x^2\right)^{p-1} dx$$

Program code:

```
Int[(d_+e_.*x_)^m_*(a_.+b_.*x_+c_.*x_^2)^p_.,x_Symbol] :=
  (d+e*x)^(m+1)*(a+b*x+c*x^2)^p/(e*(m+2*p+1)) -
  d*p*(b^2-4*a*c)/(b*e*(m+2*p+1))*Int[(d+e*x)^m*(a+b*x+c*x^2)^(p-1),x] /;
FreeQ[{a,b,c,d,e,m},x] && NeQ[b^2-4*a*c,0] && EqQ[2*c*d-b*e,0] && NeQ[m+2*p+3,0] && GtQ[p,0] &&
  Not[LtQ[m,-1]] && Not[IGtQ[(m-1)/2,0] && (Not[IntegerQ[p]] || LtQ[m,2*p])] && RationalQ[m] && IntegerQ[2*p]
```

4. $\int (d+e\,x)^m \left(a+b\,x+c\,x^2\right)^p dx$ when $b^2-4\,a\,c \neq 0 \wedge 2\,c\,d-b\,e == 0 \wedge m+2\,p+3 \neq 0 \wedge p<-1$

1: $\int (d+e\,x)^m \left(a+b\,x+c\,x^2\right)^p dx$ when $b^2-4\,a\,c \neq 0 \wedge 2\,c\,d-b\,e == 0 \wedge m+2\,p+3 \neq 0 \wedge p<-1 \wedge m>1$

Derivation: Derivative divides quadratic recurrence 2a

Derivation: Integration by parts

Rule 1.2.1.2.3.4.1: If $b^2-4\,a\,c \neq 0 \wedge 2\,c\,d-b\,e == 0 \wedge m+2\,p+3 \neq 0 \wedge p<-1 \wedge m>1$**, then**

$$\int (d+e\,x)^m \left(a+b\,x+c\,x^2\right)^p dx \longrightarrow \frac{d\,(d+e\,x)^{m-1} \left(a+b\,x+c\,x^2\right)^{p+1}}{b\,(p+1)} - \frac{d\,e\,(m-1)}{b\,(p+1)} \int (d+e\,x)^{m-2} \left(a+b\,x+c\,x^2\right)^{p+1} dx$$

Program code:

```
Int[(d_+e_.*x_)^m_*(a_.+b_.*x_+c_.*x_^2)^p_,x_Symbol] :=
  d*(d+e*x)^(m-1)*(a+b*x+c*x^2)^(p+1)/(b*(p+1)) -
  d*e*(m-1)/(b*(p+1))*Int[(d+e*x)^(m-2)*(a+b*x+c*x^2)^(p+1),x] /;
FreeQ[{a,b,c,d,e},x] && NeQ[b^2-4*a*c,0] && EqQ[2*c*d-b*e,0] && NeQ[m+2*p+3,0] && LtQ[p,-1] && GtQ[m,1] && IntegerQ[2*p]
```

2: $\int (d+e\,x)^m \left(a+b\,x+c\,x^2\right)^p\,dx$ when $b^2-4\,a\,c\neq 0\;\land\;2\,c\,d-b\,e==0\;\land\;m+2\,p+3\neq 0\;\land\;p<-1\;\land\;m\not>1$

Derivation: Derivative divides quadratic recurrence 2b

Rule 1.2.1.2.3.4.2: If $b^2-4\,a\,c\neq 0\;\land\;2\,c\,d-b\,e==0\;\land\;m+2\,p+3\neq 0\;\land\;p<-1\;\land\;m\not>1$**, then**

$$\int (d+e\,x)^m \left(a+b\,x+c\,x^2\right)^p\,dx\;\longrightarrow\;\frac{2\,c\,(d+e\,x)^{m+1}\left(a+b\,x+c\,x^2\right)^{p+1}}{e\,(p+1)\left(b^2-4\,a\,c\right)}-\frac{2\,c\,e\,(m+2\,p+3)}{e\,(p+1)\left(b^2-4\,a\,c\right)}\int (d+e\,x)^m\left(a+b\,x+c\,x^2\right)^{p+1}\,dx$$

Program code:

```
Int[(d_+e_.*x_)^m_*(a_.+b_.*x_+c_.*x_^2)^p_.,x_Symbol] :=
  2*c*(d+e*x)^(m+1)*(a+b*x+c*x^2)^(p+1)/(e*(p+1)*(b^2-4*a*c)) -
  2*c*e*(m+2*p+3)/(e*(p+1)*(b^2-4*a*c))*Int[(d+e*x)^m*(a+b*x+c*x^2)^(p+1),x] /;
FreeQ[{a,b,c,d,e,m},x] && NeQ[b^2-4*a*c,0] && EqQ[2*c*d-b*e,0] && NeQ[m+2*p+3,0] && LtQ[p,-1] && Not[GtQ[m,1]] && RationalQ[m] && I
```

5: $\int \frac{1}{(d+e\,x)\,\sqrt{a+b\,x+c\,x^2}}\,dx$ when $b^2-4\,a\,c\neq 0\;\land\;2\,c\,d-b\,e==0$

Derivation: Integration by substitution

Basis: If $2\,c\,d-b\,e==0$**, then** $\frac{1}{(d+e\,x)\,\sqrt{a+b\,x+c\,x^2}}==4\,c\,\text{Subst}\left[\frac{1}{b^2\,e-4\,a\,c\,e+4\,c\,e\,x^2},\,x,\,\sqrt{a+b\,x+c\,x^2}\right]\partial_x\sqrt{a+b\,x+c\,x^2}$

Rule 1.2.1.2.3.5: If $b^2-4\,a\,c\neq 0\;\land\;2\,c\,d-b\,e==0$**, then**

$$\int \frac{1}{(d+e\,x)\,\sqrt{a+b\,x+c\,x^2}}\,dx\;\longrightarrow\;4\,c\,\text{Subst}\left[\int\frac{1}{b^2\,e-4\,a\,c\,e+4\,c\,e\,x^2}\,dx,\,x,\,\sqrt{a+b\,x+c\,x^2}\right]$$

Program code:

```
Int[1/((d_+e_.*x_)*Sqrt[a_.+b_.*x_+c_.*x_^2]),x_Symbol] :=
  4*c*Subst[Int[1/(b^2*e-4*a*c*e+4*c*e*x^2),x],x,Sqrt[a+b*x+c*x^2]] /;
FreeQ[{a,b,c,d,e},x] && NeQ[b^2-4*a*c,0] && EqQ[2*c*d-b*e,0]
```

6. $\int \frac{(d+e\,x)^m}{\sqrt{a+b\,x+c\,x^2}}\,dx$ when $b^2-4\,a\,c\neq 0\;\bigwedge\;2\,c\,d-b\,e==0\;\bigwedge\;m^2==\frac{1}{4}$

1. $\int \frac{(d+e\,x)^m}{\sqrt{a+b\,x+c\,x^2}}\,dx$ when $b^2-4\,a\,c\neq 0\;\bigwedge\;2\,c\,d-b\,e==0\;\bigwedge\;m^2==\frac{1}{4}\;\bigwedge\;\frac{c}{b^2-4\,a\,c}<0$

1: $\int \frac{1}{\sqrt{d+e\,x}\,\sqrt{a+b\,x+c\,x^2}}\,dx$ when $b^2-4\,a\,c\neq 0\,\bigwedge\,2\,c\,d-b\,e=0\,\bigwedge\,\frac{c}{b^2-4\,a\,c}<0$

Derivation: Integration by substitution

- **Basis: If** $2\,c\,d-b\,e=0\,\bigwedge\,\frac{c}{b^2-4\,a\,c}<0$**, then** $\frac{1}{\sqrt{d+e\,x}\,\sqrt{a+b\,x+c\,x^2}}=\frac{4}{e}\sqrt{-\frac{c}{b^2-4\,a\,c}}\,\text{Subst}\left[\frac{1}{\sqrt{1-\frac{b^2\,x^4}{d^2\,(b^2-4\,a\,c)}}},\,x,\,\sqrt{d+e\,x}\right]\partial_x\sqrt{d+e\,x}$

- **Rule 1.2.1.2.3.6.1.1: If** $b^2-4\,a\,c\neq 0\,\bigwedge\,2\,c\,d-b\,e=0\,\bigwedge\,\frac{c}{b^2-4\,a\,c}<0$**, then**

$$\int \frac{1}{\sqrt{d+e\,x}\,\sqrt{a+b\,x+c\,x^2}}\,dx\to\frac{4}{e}\sqrt{-\frac{c}{b^2-4\,a\,c}}\,\text{Subst}\left[\int\frac{1}{\sqrt{1-\frac{b^2\,x^4}{d^2\,(b^2-4\,a\,c)}}}\,dx,\,x,\,\sqrt{d+e\,x}\right]$$

Program code:

```
Int[1/(Sqrt[d_+e_.*x_]*Sqrt[a_.+b_.*x_+c_.*x_^2]),x_Symbol] :=
  4/e*Sqrt[-c/(b^2-4*a*c)]*Subst[Int[1/Sqrt[Simp[1-b^2*x^4/(d^2*(b^2-4*a*c)),x]],x],x,Sqrt[d+e*x]] /;
FreeQ[{a,b,c,d,e},x] && NeQ[b^2-4*a*c,0] && EqQ[2*c*d-b*e,0] && LtQ[c/(b^2-4*a*c),0]
```

2: $\int \frac{\sqrt{d+e\,x}}{\sqrt{a+b\,x+c\,x^2}}\,dx$ when $b^2-4\,a\,c\neq 0\,\bigwedge\,2\,c\,d-b\,e=0\,\bigwedge\,\frac{c}{b^2-4\,a\,c}<0$

Derivation: Integration by substitution

- **Basis: If** $2\,c\,d-b\,e=0\,\bigwedge\,\frac{c}{b^2-4\,a\,c}<0$**, then** $\frac{\sqrt{d+e\,x}}{\sqrt{a+b\,x+c\,x^2}}=\frac{4}{e}\sqrt{-\frac{c}{b^2-4\,a\,c}}\,\text{Subst}\left[\frac{x^2}{\sqrt{1-\frac{b^2\,x^4}{d^2\,(b^2-4\,a\,c)}}},\,x,\,\sqrt{d+e\,x}\right]\partial_x\sqrt{d+e\,x}$

- **Rule 1.2.1.2.3.6.1.2: If** $b^2-4\,a\,c\neq 0\,\bigwedge\,2\,c\,d-b\,e=0\,\bigwedge\,\frac{c}{b^2-4\,a\,c}<0$**, then**

$$\int \frac{\sqrt{d+e\,x}}{\sqrt{a+b\,x+c\,x^2}}\,dx\to\frac{4}{e}\sqrt{-\frac{c}{b^2-4\,a\,c}}\,\text{Subst}\left[\int\frac{x^2}{\sqrt{1-\frac{b^2\,x^4}{d^2\,(b^2-4\,a\,c)}}}\,dx,\,x,\,\sqrt{d+e\,x}\right]$$

Program code:

```
Int[Sqrt[d_+e_.*x_]/Sqrt[a_.+b_.*x_+c_.*x_^2],x_Symbol] :=
  4/e*Sqrt[-c/(b^2-4*a*c)]*Subst[Int[x^2/Sqrt[Simp[1-b^2*x^4/(d^2*(b^2-4*a*c)),x]],x],x,Sqrt[d+e*x]] /;
FreeQ[{a,b,c,d,e},x] && NeQ[b^2-4*a*c,0] && EqQ[2*c*d-b*e,0] && LtQ[c/(b^2-4*a*c),0]
```

2: $\int \frac{(d+e\,x)^m}{\sqrt{a+b\,x+c\,x^2}}\,dx$ when $b^2-4\,a\,c\neq 0\;\bigwedge\;2\,c\,d-b\,e==0\;\bigwedge\;m^2==\frac{1}{4}\;\bigwedge\;\frac{c}{b^2-4\,a\,c}\nless 0$

- **Derivation: Piecewise constant extraction**

- **Basis:** $\partial_x \frac{\sqrt{-\frac{c\,(a+b\,x+c\,x^2)}{b^2-4\,a\,c}}}{\sqrt{a+b\,x+c\,x^2}}==0$

- **Rule 1.2.1.2.3.6.2: If** $b^2-4\,a\,c\neq 0\;\bigwedge\;2\,c\,d-b\,e==0\;\bigwedge\;m^2==\frac{1}{4}\;\bigwedge\;\frac{c}{b^2-4\,a\,c}\nless 0$**, then**

$$\int \frac{(d+e\,x)^m}{\sqrt{a+b\,x+c\,x^2}}\,dx\;\longrightarrow\;\frac{\sqrt{-\frac{c\,(a+b\,x+c\,x^2)}{b^2-4\,a\,c}}}{\sqrt{a+b\,x+c\,x^2}}\int \frac{(d+e\,x)^m}{\sqrt{-\frac{a\,c}{b^2-4\,a\,c}-\frac{b\,c\,x}{b^2-4\,a\,c}-\frac{c^2\,x^2}{b^2-4\,a\,c}}}\,dx$$

- **Program code:**

```
Int[(d_+e_.*x_)^m_/Sqrt[a_.+b_.*x_+c_.*x_^2],x_Symbol] :=
  Sqrt[-c*(a+b*x+c*x^2)/(b^2-4*a*c)]/Sqrt[a+b*x+c*x^2]*
    Int[(d+e*x)^m/Sqrt[-a*c/(b^2-4*a*c)-b*c*x/(b^2-4*a*c)-c^2*x^2/(b^2-4*a*c)],x] /;
FreeQ[{a,b,c,d,e},x] && NeQ[b^2-4*a*c,0] && EqQ[2*c*d-b*e,0] && EqQ[m^2,1/4]
```

7: $\int (d+e\,x)^m\,\left(a+b\,x+c\,x^2\right)^p\,dx$ when $b^2-4\,a\,c\neq 0\,\wedge\,2\,c\,d-b\,e==0\,\wedge\,m+2\,p+3\neq 0\,\wedge\,m>1\,\wedge\,p\nless -1$

Derivation: Derivative divides quadratic recurrence 3a

- **Derivation: Integration by parts**

- **Rule 1.2.1.2.3.7: If** $b^2-4\,a\,c\neq 0\,\wedge\,2\,c\,d-b\,e==0\,\wedge\,m+2\,p+3\neq 0\,\wedge\,m>1\,\wedge\,p\nless -1$**, then**

$$\int (d+e\,x)^m\,\left(a+b\,x+c\,x^2\right)^p\,dx\;\longrightarrow\;\frac{2\,d\,(d+e\,x)^{m-1}\,\left(a+b\,x+c\,x^2\right)^{p+1}}{b\,(m+2\,p+1)}+\frac{d^2\,(m-1)\,\left(b^2-4\,a\,c\right)}{b^2\,(m+2\,p+1)}\int (d+e\,x)^{m-2}\,\left(a+b\,x+c\,x^2\right)^p\,dx$$

Program code:

```
Int[(d_+e_.*x_)^m_*(a_.+b_.*x_+c_.*x_^2)^p_.,x_Symbol] :=
  2*d*(d+e*x)^(m-1)*(a+b*x+c*x^2)^(p+1)/(b*(m+2*p+1)) +
  d^2*(m-1)*(b^2-4*a*c)/(b^2*(m+2*p+1))*Int[(d+e*x)^(m-2)*(a+b*x+c*x^2)^p,x] /;
FreeQ[{a,b,c,d,e,p},x] && NeQ[b^2-4*a*c,0] && EqQ[2*c*d-b*e,0] && NeQ[m+2*p+3,0] && GtQ[m,1] &&
  NeQ[m+2*p+1,0] && (IntegerQ[2*p] || IntegerQ[m] && RationalQ[p] || OddQ[m])
```

8: $\int (d+e x)^m \left(a+b x+c x^2\right)^p dx$ when $b^2-4 a c \neq 0 \land 2 c d-b e == 0 \land m+2 p+3 \neq 0 \land m<-1 \land p \not> 0$

Derivation: Derivative divides quadratic recurrence 3b

Rule 1.2.1.2.3.8: If $b^2-4 a c \neq 0 \land 2 c d-b e == 0 \land m+2 p+3 \neq 0 \land m<-1 \land p \not> 0$**, then**

$$\int (d+e x)^m \left(a+b x+c x^2\right)^p dx \to -\frac{2 b d (d+e x)^{m+1} \left(a+b x+c x^2\right)^{p+1}}{d^2 (m+1) \left(b^2-4 a c\right)}+\frac{b^2 (m+2 p+3)}{d^2 (m+1) \left(b^2-4 a c\right)} \int (d+e x)^{m+2} \left(a+b x+c x^2\right)^p dx$$

Program code:

```
Int[(d_+e_.*x_)^m_*(a_.+b_.*x_+c_.*x_^2)^p_.,x_Symbol] :=
  -2*b*d*(d+e*x)^(m+1)*(a+b*x+c*x^2)^(p+1)/(d^2*(m+1)*(b^2-4*a*c)) +
  b^2*(m+2*p+3)/(d^2*(m+1)*(b^2-4*a*c))*Int[(d+e*x)^(m+2)*(a+b*x+c*x^2)^p,x] /;
FreeQ[{a,b,c,d,e,p},x] && NeQ[b^2-4*a*c,0] && EqQ[2*c*d-b*e,0] && NeQ[m+2*p+3,0] && LtQ[m,-1] &&
  (IntegerQ[2*p] || IntegerQ[m] && RationalQ[p] || IntegerQ[(m+2*p+3)/2])
```

9: $\int (d+e x)^m \left(a+b x+c x^2\right)^p dx$ when $b^2-4 a c \neq 0 \land 2 c d-b e == 0$

Derivation: Integration by substitution

Basis: If $2 c d-b e == 0$**, then** $F\left[a+b x+c x^2\right] == \frac{1}{e} \text{Subst}\left[F\left[a-\frac{b^2}{4 c}+\frac{c x^2}{e^2}\right], x, d+e x\right] \partial_x (d+e x)$

Rule 1.2.1.2.3.9: If $b^2-4 a c \neq 0 \land 2 c d-b e == 0$**, then**

$$\int (d+e x)^m \left(a+b x+c x^2\right)^p dx \to \frac{1}{e} \text{Subst}\left[\int x^m \left(a-\frac{b^2}{4 c}+\frac{c x^2}{e^2}\right)^p dx, x, d+e x\right]$$

Program code:

```
Int[(d_+e_.*x_)^m_*(a_.+b_.*x_+c_.*x_^2)^p_.,x_Symbol] :=
  1/e*Subst[Int[x^m*(a-b^2/(4*c)+(c*x^2)/e^2)^p,x],x,d+e*x] /;
FreeQ[{a,b,c,d,e,m,p},x] && NeQ[b^2-4*a*c,0] && EqQ[2*c*d-b*e,0]
```

?: $\int \frac{1}{(d+e x) \left(a+c x^2\right)^{1/4}} dx$ when $c d^2+2 a e^2 == 0 \land a<0$

Reference: Eneström index number E688 in The Euler Archive

Rule 1.2.1.2.?: If $c d^2+2 a e^2 == 0 \land a<0$**, then**

$$\int \frac{1}{(d+e\,x)\,\left(a+c\,x^2\right)^{1/4}}\,dx \;\longrightarrow\; \frac{1}{2\,(-a)^{1/4}\,e}\,\mathrm{ArcTan}\!\left[\frac{\left(-1-\frac{c\,x^2}{a}\right)^{1/4}}{1-\frac{c\,d\,x}{2\,a\,e}-\sqrt{-1-\frac{c\,x^2}{a}}}\right]+\frac{1}{4\,(-a)^{1/4}\,e}\,\mathrm{Log}\!\left[\frac{1-\frac{c\,d\,x}{2\,a\,e}+\sqrt{-1-\frac{c\,x^2}{a}}-\left(-1-\frac{c\,x^2}{a}\right)^{1/4}}{1-\frac{c\,d\,x}{2\,a\,e}+\sqrt{-1-\frac{c\,x^2}{a}}+\left(-1-\frac{c\,x^2}{a}\right)^{1/4}}\right]$$

Program code:

```
Int[1/((d_+e_.*x_)*(a_.+c_.*x_^2)^(1/4)),x_Symbol] :=
  1/(2*(-a)^(1/4)*e)*ArcTan[(-1-c*x^2/a)^(1/4)/(1-c*d*x/(2*a*e)-Sqrt[-1-c*x^2/a])] +
  1/(4*(-a)^(1/4)*e)*Log[(1-c*d*x/(2*a*e)+Sqrt[-1-c*x^2/a]-(-1-c*x^2/a)^(1/4))/
    (1-c*d*x/(2*a*e)+Sqrt[-1-c*x^2/a]+(-1-c*x^2/a)^(1/4))] /;
FreeQ[{a,c,d,e},x] && EqQ[c*d^2+2*a*e^2,0] && LtQ[a,0]
```

5. $\int (d+e\,x)^m\,\left(a+b\,x+c\,x^2\right)^p\,dx$ when $b^2-4\,a\,c\neq 0\;\wedge\;c\,d^2-b\,d\,e+a\,e^2\neq 0\;\wedge\;2\,c\,d-b\,e\neq 0\;\wedge\;p\in\mathbb{Z}\;\wedge\;(p>0\;\vee\;m\in\mathbb{Z})$

1. $\int (d+e\,x)^m\,\left(a+c\,x^2\right)^p\,dx$ when $c\,d^2+a\,e^2\neq 0\;\wedge\;p\in\mathbb{Z}^+$

1: $\int (d+e\,x)^m\,\left(a+c\,x^2\right)^p\,dx$ when $c\,d^2+a\,e^2\neq 0\;\wedge\;p-1\in\mathbb{Z}^+\;\wedge\;m\in\mathbb{Z}^+\;\wedge\;m\leq p$

Derivation: Algebraic expansion and power rule for integration

Note: This rule removes the one degree term from the polynomial $(d+e\,x)^m$.

Rule: If $c\,d^2+a\,e^2\neq 0\;\wedge\;p-1\in\mathbb{Z}^+\;\wedge\;m\in\mathbb{Z}^+\;\wedge\;m\leq p$, then

$$\int (d+e\,x)^m\,\left(a+c\,x^2\right)^p\,dx \;\longrightarrow\; e\,m\,d^{m-1}\int x\,\left(a+c\,x^2\right)^p\,dx+\int\left((d+e\,x)^m-e\,m\,d^{m-1}\,x\right)\,\left(a+c\,x^2\right)^p\,dx$$

$$\longrightarrow\;\frac{e\,m\,d^{m-1}\,\left(a+c\,x^2\right)^{p+1}}{2\,c\,(p+1)}+\int\left((d+e\,x)^m-e\,m\,d^{m-1}\,x\right)\,\left(a+c\,x^2\right)^p\,dx$$

Program code:

```
Int[(d_+e_.*x_)^m_*(a_+c_.*x_^2)^p_,x_Symbol] :=
  e*m*d^(m-1)*(a+c*x^2)^(p+1)/(2*c*(p+1)) +
  Int[((d+e*x)^m-e*m*d^(m-1)*x)*(a+c*x^2)^p,x] /;
FreeQ[{a,c,d,e},x] && NeQ[c*d^2+a*e^2,0] && IGtQ[p,1] && IGtQ[m,0] && LeQ[m,p]
```

2: $\int (d + e\,x)^m \left(a + c\,x^2\right)^p \,dx$ when $c\,d^2 + a\,e^2 \neq 0 \wedge p \in \mathbb{Z}^+$

Derivation: Algebraic expansion

Rule 1.2.1.2.5.2: If $c\,d^2 + a\,e^2 \neq 0 \wedge p \in \mathbb{Z}^+$, then

$$\int (d + e\,x)^m \left(a + b\,x + c\,x^2\right)^p \,dx \longrightarrow \int \text{ExpandIntegrand}\left[(d + e\,x)^m \left(a + b\,x + c\,x^2\right)^p,\ x\right] dx$$

Program code:

```
Int[(d_+e_.*x_)^m_*(a_+c_.*x_^2)^p_.,x_Symbol] :=
  Int[ExpandIntegrand[(d+e*x)^m*(a+c*x^2)^p,x],x] /;
FreeQ[{a,c,d,e,m},x] && NeQ[c*d^2+a*e^2,0] && IGtQ[p,0]
```

2: $\int (d + e\,x)^m \left(a + b\,x + c\,x^2\right)^p \,dx$ when $b^2 - 4\,a\,c \neq 0 \wedge c\,d^2 - b\,d\,e + a\,e^2 \neq 0 \wedge 2\,c\,d - b\,e \neq 0 \wedge p \in \mathbb{Z} \wedge (p > 0 \vee a == 0 \wedge m \in \mathbb{Z})$

Derivation: Algebraic expansion

Rule 1.2.1.2.5.2: If $b^2 - 4\,a\,c \neq 0 \wedge c\,d^2 - b\,d\,e + a\,e^2 \neq 0 \wedge 2\,c\,d - b\,e \neq 0 \wedge p \in \mathbb{Z} \wedge (p > 0 \vee a == 0 \wedge m \in \mathbb{Z})$, then

$$\int (d + e\,x)^m \left(a + b\,x + c\,x^2\right)^p \,dx \longrightarrow \int \text{ExpandIntegrand}\left[(d + e\,x)^m \left(a + b\,x + c\,x^2\right)^p,\ x\right] dx$$

Program code:

```
Int[(d_.+e_.*x_)^m_*(a_.+b_.*x_+c_.*x_^2)^p_.,x_Symbol] :=
  Int[ExpandIntegrand[(d+e*x)^m*(a+b*x+c*x^2)^p,x],x] /;
FreeQ[{a,b,c,d,e,m},x] && NeQ[b^2-4*a*c,0] && NeQ[c*d^2-b*d*e+a*e^2,0] && NeQ[2*c*d-b*e,0] && IntegerQ[p] && (GtQ[p,0] || EqQ[a,0]
```

6. $\int \frac{(d+e\,x)^m}{a+b\,x+c\,x^2}\,dx$ when $b^2-4\,a\,c\neq 0\,\wedge\,c\,d^2-b\,d\,e+a\,e^2\neq 0\,\wedge\,2\,c\,d-b\,e\neq 0$

1. $\int \frac{(d+e\,x)^m}{a+b\,x+c\,x^2}\,dx$ when $b^2-4\,a\,c\neq 0\,\wedge\,c\,d^2-b\,d\,e+a\,e^2\neq 0\,\wedge\,2\,c\,d-b\,e\neq 0\,\wedge\,m>0$

x. $\int \frac{\sqrt{d+e\,x}}{a+b\,x+c\,x^2}\,dx$ when $b^2-4\,a\,c\neq 0\,\wedge\,c\,d^2-b\,d\,e+a\,e^2\neq 0\,\wedge\,2\,c\,d-b\,e\neq 0$

1: $\int \frac{\sqrt{d+e\,x}}{a+b\,x+c\,x^2}\,dx$ when $b^2-4\,a\,c\neq 0\,\wedge\,c\,d^2-b\,d\,e+a\,e^2\neq 0\,\wedge\,2\,c\,d-b\,e\neq 0\,\wedge\,b^2-4\,a\,c<0$

- **Derivation: Algebraic expansion**
- **Basis:** $\sqrt{d+e\,x} == \frac{d+q+e\,x}{2\sqrt{d+e\,x}}+\frac{d-q+e\,x}{2\sqrt{d+e\,x}}$
- **Note: Resulting integrands are of the form** $\frac{A+B\,x}{\sqrt{d+e\,x}\,(a+b\,x+c\,x^2)}$ **where** $A^2\,c\,e-2\,A\,B\,c\,d+B^2\,(b\,d-a\,e) == 0$.
- **Note: Although use of this rule when** $b^2-4\,a\,c<0$ **results in antiderivatives superficially free of the imaginary unit but significantly more complicated than those produced by the following rule.**
- **Rule 1.2.1.2.6.1.x.1: If** $b^2-4\,a\,c\neq 0\,\wedge\,c\,d^2-b\,d\,e+a\,e^2\neq 0\,\wedge\,2\,c\,d-b\,e\neq 0\,\wedge\,b^2-4\,a\,c<0$**, let** $q\to\sqrt{\frac{c\,d^2-b\,d\,e+a\,e^2}{c}}$ **, then**

$$\int \frac{\sqrt{d+e\,x}}{a+b\,x+c\,x^2}\,dx \longrightarrow \frac{1}{2}\int \frac{d+q+e\,x}{\sqrt{d+e\,x}\,\left(a+b\,x+c\,x^2\right)}\,dx+\frac{1}{2}\int \frac{d-q+e\,x}{\sqrt{d+e\,x}\,\left(a+b\,x+c\,x^2\right)}\,dx$$

- **Program code:**

```
(* Int[Sqrt[d_.+e_.*x_]/(a_.+b_.*x_+c_.*x_^2),x_Symbol] :=
  With[{q=Rt[(c*d^2-b*d*e+a*e^2)/c,2]},
  1/2*Int[(d+q+e*x)/(Sqrt[d+e*x]*(a+b*x+c*x^2)),x] +
  1/2*Int[(d-q+e*x)/(Sqrt[d+e*x]*(a+b*x+c*x^2)),x]] /;
FreeQ[{a,b,c,d,e},x] && NeQ[b^2-4*a*c,0] && NeQ[c*d^2-b*d*e+a*e^2,0] && NeQ[2*c*d-b*e,0] && LtQ[b^2-4*a*c,0] *)
```

```
(* Int[Sqrt[d_+e_.*x_]/(a_+c_.*x_^2),x_Symbol] :=
  With[{q=Rt[(c*d^2+a*e^2)/c,2]},
  1/2*Int[(d+q+e*x)/(Sqrt[d+e*x]*(a+c*x^2)),x] +
  1/2*Int[(d-q+e*x)/(Sqrt[d+e*x]*(a+c*x^2)),x]] /;
FreeQ[{a,c,d,e},x] && NeQ[c*d^2+a*e^2,0] && LtQ[-a*c,0] *)
```

2: $\int \frac{\sqrt{d+e\,x}}{a+b\,x+c\,x^2}\,dx$ when $b^2-4\,a\,c\neq 0\,\wedge\,c\,d^2-b\,d\,e+a\,e^2\neq 0\,\wedge\,2\,c\,d-b\,e\neq 0\,\wedge\,\neg\left(b^2-4\,a\,c<0\right)$

Derivation: Algebraic expansion

- **Basis: If** $q=\sqrt{b^2-4\,a\,c}$, **then** $\frac{\sqrt{d+e\,x}}{a+b\,x+c\,x^2}==\frac{2\,c\,d-b\,e+e\,q}{q\,\sqrt{d+e\,x}\,(b-q+2\,c\,x)}-\frac{2\,c\,d-b\,e-e\,q}{q\,\sqrt{d+e\,x}\,(b+q+2\,c\,x)}$

- **Rule 1.2.1.2.6.1.x.2: If** $b^2-4\,a\,c\neq 0\,\wedge\,c\,d^2-b\,d\,e+a\,e^2\neq 0\,\wedge\,2\,c\,d-b\,e\neq 0\,\wedge\,\neg\left(b^2-4\,a\,c<0\right)$, **let** $q\to\sqrt{b^2-4\,a\,c}$, **then**

$$\int \frac{\sqrt{d+e\,x}}{a+b\,x+c\,x^2}\,dx\,\longrightarrow\,\frac{2\,c\,d-b\,e+e\,q}{q}\int\frac{1}{\sqrt{d+e\,x}\,(b-q+2\,c\,x)}\,dx-\frac{2\,c\,d-b\,e-e\,q}{q}\int\frac{1}{\sqrt{d+e\,x}\,(b+q+2\,c\,x)}\,dx$$

- **Program code:**

```
(* Int[Sqrt[d_.+e_.*x_]/(a_.+b_.*x_+c_.*x_^2),x_Symbol] :=
  With[{q=Rt[b^2-4*a*c,2]},
  (2*c*d-b*e+e*q)/q*Int[1/(Sqrt[d+e*x]*(b-q+2*c*x)),x] -
  (2*c*d-b*e-e*q)/q*Int[1/(Sqrt[d+e*x]*(b+q+2*c*x)),x]] /;
FreeQ[{a,b,c,d,e},x] && NeQ[b^2-4*a*c,0] && NeQ[c*d^2-b*d*e+a*e^2,0] && NeQ[2*c*d-b*e,0] (* && Not[LtQ[b^2-4*a*c,0]] *) *)
```

```
(* Int[Sqrt[d_+e_.*x_]/(a_+c_.*x_^2),x_Symbol] :=
  With[{q=Rt[-a*c,2]},
  (c*d+e*q)/(2*q)*Int[1/(Sqrt[d+e*x]*(-q+c*x)),x] -
  (c*d-e*q)/(2*q)*Int[1/(Sqrt[d+e*x]*(+q+c*x)),x]] /;
FreeQ[{a,c,d,e},x] && NeQ[c*d^2+a*e^2,0] (* && Not[LtQ[-a*c,0]] *) *)
```

1: $\int \frac{\sqrt{d+e x}}{a+b x+c x^2} \, dx$ when $b^2-4 a c \neq 0 \land c d^2-b d e+a e^2 \neq 0 \land 2 c d-b e \neq 0$

Derivation: Integration by substitution

- **Basis:** $(d+e x)^m F[x] = \frac{2}{e} \text{Subst}\left[x^{2 m+1} F\left[\frac{-d+x^2}{e}\right], x, \sqrt{d+e x}\right] \partial_x \sqrt{d+e x}$

Rule 1.2.1.2.6.1.1: If $b^2-4 a c \neq 0 \land c d^2-b d e+a e^2 \neq 0 \land 2 c d-b e \neq 0$

$$\int \frac{\sqrt{d+e x}}{a+b x+c x^2} \, dx \to 2 e \, \text{Subst}\left[\int \frac{x^2}{c d^2-b d e+a e^2-(2 c d-b e) x^2+c x^4} \, dx, x, \sqrt{d+e x}\right]$$

Program code:

```
Int[Sqrt[d_.+e_.*x_]/(a_.+b_.*x_+c_.*x_^2),x_Symbol] :=
  2*e*Subst[Int[x^2/(c*d^2-b*d*e+a*e^2-(2*c*d-b*e)*x^2+c*x^4),x],x,Sqrt[d+e*x]] /;
FreeQ[{a,b,c,d,e},x] && NeQ[b^2-4*a*c,0] && NeQ[c*d^2-b*d*e+a*e^2,0] && NeQ[2*c*d-b*e,0]
```

```
Int[Sqrt[d_+e_.*x_]/(a_+c_.*x_^2),x_Symbol] :=
  2*e*Subst[Int[x^2/(c*d^2+a*e^2-2*c*d*x^2+c*x^4),x],x,Sqrt[d+e*x]] /;
FreeQ[{a,c,d,e},x] && NeQ[c*d^2+a*e^2,0]
```

2. $\int \frac{(d+e x)^m}{a+b x+c x^2} \, dx$ when $b^2-4 a c \neq 0 \land c d^2-b d e+a e^2 \neq 0 \land 2 c d-b e \neq 0 \land m > 1$

1: $\int \frac{(d+e x)^m}{a+b x+c x^2} \, dx$ when $b^2-4 a c \neq 0 \land c d^2-b d e+a e^2 \neq 0 \land 2 c d-b e \neq 0 \land m \in \mathbb{Z} \land m > 1 \land (d \neq 0 \lor m > 2)$

Derivation: Algebraic expansion

Rule 1.2.1.2.6.1.2.1: If $b^2-4 a c \neq 0 \land c d^2-b d e+a e^2 \neq 0 \land 2 c d-b e \neq 0 \land m \in \mathbb{Z} \land m > 1 \land (d \neq 0 \lor m > 2)$**, then**

$$\int \frac{(d+e x)^m}{a+b x+c x^2} \, dx \to \int \text{PolynomialDivide}\left[(d+e x)^m, a+b x+c x^2, x\right] dx$$

Program code:

```
Int[(d_.+e_.*x_)^m_/(a_.+b_.*x_+c_.*x_^2),x_Symbol] :=
  Int[PolynomialDivide[(d+e*x)^m,a+b*x+c*x^2,x],x] /;
FreeQ[{a,b,c,d,e},x] && NeQ[b^2-4*a*c,0] && NeQ[c*d^2-b*d*e+a*e^2,0] && NeQ[2*c*d-b*e,0] && IGtQ[m,1] && (NeQ[d,0] || GtQ[m,2])
```

```
Int[(d_+e_.*x_)^m_/(a_+c_.*x_^2),x_Symbol] :=
  Int[PolynomialDivide[(d+e*x)^m,a+c*x^2,x],x] /;
FreeQ[{a,c,d,e},x] && NeQ[c*d^2+a*e^2,0] && IGtQ[m,1] && (NeQ[d,0] || GtQ[m,2])
```

2: $\int \frac{(d+e\,x)^m}{a+b\,x+c\,x^2}\,dx$ when $b^2-4\,a\,c\neq 0\;\wedge\;c\,d^2-b\,d\,e+a\,e^2\neq 0\;\wedge\;2\,c\,d-b\,e\neq 0\;\wedge\;m>1$

Reference: G&R 2.160.3, G&R 2.174.1, CRC 119

Derivation: Quadratic recurrence 3a with $A=d$, $B=e$, $m=m-1$ and $p=-1$

Note: G&R 2.174.1 is a special case of G&R 2.160.3.

Rule 1.2.1.2.6.1.2.2: If $b^2-4\,a\,c\neq 0\;\wedge\;c\,d^2-b\,d\,e+a\,e^2\neq 0\;\wedge\;2\,c\,d-b\,e\neq 0\;\wedge\;m>1$, then

$$\int \frac{(d+e\,x)^m}{a+b\,x+c\,x^2}\,dx \longrightarrow \frac{e\,(d+e\,x)^{m-1}}{c\,(m-1)}+\frac{1}{c}\int \frac{(d+e\,x)^{m-2}\left(c\,d^2-a\,e^2+e\,(2\,c\,d-b\,e)\,x\right)}{a+b\,x+c\,x^2}\,dx$$

Program code:

```
Int[(d_.+e_.*x_)^m_/(a_.+b_.*x_+c_.*x_^2),x_Symbol] :=
  e*(d+e*x)^(m-1)/(c*(m-1)) +
  1/c*Int[(d+e*x)^(m-2)*Simp[c*d^2-a*e^2+e*(2*c*d-b*e)*x,x]/(a+b*x+c*x^2),x] /;
FreeQ[{a,b,c,d,e},x] && NeQ[b^2-4*a*c,0] && NeQ[c*d^2-b*d*e+a*e^2,0] && NeQ[2*c*d-b*e,0] && GtQ[m,1]
```

```
Int[(d_+e_.*x_)^m_/(a_+c_.*x_^2),x_Symbol] :=
  e*(d+e*x)^(m-1)/(c*(m-1)) +
  1/c*Int[(d+e*x)^(m-2)*Simp[c*d^2-a*e^2+2*c*d*e*x,x]/(a+c*x^2),x] /;
FreeQ[{a,c,d,e},x] && NeQ[c*d^2+a*e^2,0] && GtQ[m,1]
```

2. $\int \frac{(d+e\,x)^m}{a+b\,x+c\,x^2}\,dx$ when $b^2-4\,a\,c\neq 0\;\wedge\;c\,d^2-b\,d\,e+a\,e^2\neq 0\;\wedge\;2\,c\,d-b\,e\neq 0\;\wedge\;m<0$

1: $\int \frac{1}{(d+e\,x)\left(a+b\,x+c\,x^2\right)}\,dx$ when $b^2-4\,a\,c\neq 0\;\wedge\;c\,d^2-b\,d\,e+a\,e^2\neq 0\;\wedge\;2\,c\,d-b\,e\neq 0$

Derivation: Algebraic expansion

Basis: $\frac{1}{(d+e\,x)\,(a+b\,x+c\,x^2)} == \frac{e^2}{(c\,d^2-b\,d\,e+a\,e^2)\,(d+e\,x)}+\frac{c\,d-b\,e-c\,e\,x}{(c\,d^2-b\,d\,e+a\,e^2)\,(a+b\,x+c\,x^2)}$

Rule 1.2.1.2.6.2.1: If $b^2-4\,a\,c\neq 0\;\wedge\;c\,d^2-b\,d\,e+a\,e^2\neq 0\;\wedge\;2\,c\,d-b\,e\neq 0$, then

$$\int \frac{1}{(d+e\,x)\,\left(a+b\,x+c\,x^2\right)}\,dx \longrightarrow \frac{e^2}{c\,d^2-b\,d\,e+a\,e^2}\int \frac{1}{d+e\,x}\,dx+\frac{1}{c\,d^2-b\,d\,e+a\,e^2}\int \frac{c\,d-b\,e-c\,e\,x}{a+b\,x+c\,x^2}\,dx$$

– **Program code:**

```
Int[1/((d_.+e_.*x_)*(a_.+b_.*x_+c_.*x_^2)),x_Symbol] :=
  e^2/(c*d^2-b*d*e+a*e^2)*Int[1/(d+e*x),x] +
  1/(c*d^2-b*d*e+a*e^2)*Int[(c*d-b*e-c*e*x)/(a+b*x+c*x^2),x] /;
FreeQ[{a,b,c,d,e},x] && NeQ[b^2-4*a*c,0] && NeQ[c*d^2-b*d*e+a*e^2,0] && NeQ[2*c*d-b*e,0]
```

```
Int[1/((d_+e_.*x_)*(a_+c_.*x_^2)),x_Symbol] :=
  e^2/(c*d^2+a*e^2)*Int[1/(d+e*x),x] +
  1/(c*d^2+a*e^2)*Int[(c*d-c*e*x)/(a+c*x^2),x] /;
FreeQ[{a,c,d,e},x] && NeQ[c*d^2+a*e^2,0]
```

x. $\int \frac{1}{\sqrt{d+e x}\,\left(a+b x+c x^2\right)}\,dx$ when $b^2-4 a c \neq 0 \land c d^2-b d e+a e^2 \neq 0 \land 2 c d-b e \neq 0$

1: $\int \frac{1}{\sqrt{d+e x}\,\left(a+b x+c x^2\right)}\,dx$ when $b^2-4 a c \neq 0 \land c d^2-b d e+a e^2 \neq 0 \land 2 c d-b e \neq 0 \land b^2-4 a c<0$

Derivation: Algebraic expansion

- **Basis:** $1 == \frac{d+q+e x}{2 q}-\frac{d-q+e x}{2 q}$
- **Note: Resulting integrands are of the form** $\frac{A+B x}{\sqrt{d+e x}\,\left(a+b x+c x^2\right)}$ **where** $A^2 c e-2 A B c d+B^2 (b d-a e) == 0$**.**
- **Note: Although use of this rule when** $b^2-4 a c<0$ **results in antiderivatives superficially free of the imaginary unit but significantly more complicated than those produced by the following rule.**

- **Rule 1.2.1.2.6.2.x.1: If** $b^2-4 a c \neq 0 \land c d^2-b d e+a e^2 \neq 0 \land 2 c d-b e \neq 0 \land b^2-4 a c<0$**, let** $q \to \sqrt{\frac{c d^2-b d e+a e^2}{c}}$**, then**

$$\int \frac{1}{\sqrt{d+e x}\,\left(a+b x+c x^2\right)}\,dx \longrightarrow \frac{1}{2 q}\int \frac{d+q+e x}{\sqrt{d+e x}\,\left(a+b x+c x^2\right)}\,dx-\frac{1}{2 q}\int \frac{d-q+e x}{\sqrt{d+e x}\,\left(a+b x+c x^2\right)}\,dx$$

Program code:

```
(* Int[1/(Sqrt[d_.+e_.*x_]*(a_.+b_.*x_+c_.*x_^2)),x_Symbol] :=
  With[{q=Rt[(c*d^2-b*d*e+a*e^2)/c,2]},
  1/(2*q)*Int[(d+q+e*x)/(Sqrt[d+e*x]*(a+b*x+c*x^2)),x] -
  1/(2*q)*Int[(d-q+e*x)/(Sqrt[d+e*x]*(a+b*x+c*x^2)),x]] /;
FreeQ[{a,b,c,d,e},x] && NeQ[b^2-4*a*c,0] && NeQ[c*d^2-b*d*e+a*e^2,0] && NeQ[2*c*d-b*e,0] && LtQ[b^2-4*a*c,0] *)
```

```
(* Int[1/(Sqrt[d_+e_.*x_]*(a_+c_.*x_^2)),x_Symbol] :=
  With[{q=Rt[(c*d^2+a*e^2)/c,2]},
  1/(2*q)*Int[(d+q+e*x)/(Sqrt[d+e*x]*(a+c*x^2)),x] -
  1/(2*q)*Int[(d-q+e*x)/(Sqrt[d+e*x]*(a+c*x^2)),x]] /;
FreeQ[{a,c,d,e},x] && NeQ[c*d^2+a*e^2,0] && LtQ[-a*c,0] *)
```

2: $\int \frac{1}{\sqrt{d+e\,x}\,\left(a+b\,x+c\,x^2\right)}\,dx$ when $b^2-4\,a\,c\neq 0\,\wedge\,c\,d^2-b\,d\,e+a\,e^2\neq 0\,\wedge\,2\,c\,d-b\,e\neq 0\,\wedge\,\neg\left(b^2-4\,a\,c<0\right)$

Derivation: Algebraic expansion

- **Basis: If** $q=\sqrt{b^2-4\,a\,c}$, **then** $\frac{1}{a+b\,x+c\,x^2}==\frac{2\,c}{q\,(b-q+2\,c\,x)}-\frac{2\,c}{q\,(b+q+2\,c\,x)}$

- **Rule 1.2.1.2.6.2.x.2: If** $b^2-4\,a\,c\neq 0\,\wedge\,c\,d^2-b\,d\,e+a\,e^2\neq 0\,\wedge\,2\,c\,d-b\,e\neq 0\,\wedge\,\neg\left(b^2-4\,a\,c<0\right)$**, let** $q\to\sqrt{b^2-4\,a\,c}$**, then**

$$\int \frac{1}{\sqrt{d+e\,x}\,\left(a+b\,x+c\,x^2\right)}\,dx\,\longrightarrow\,\frac{2\,c}{q}\int \frac{1}{\sqrt{d+e\,x}\,(b-q+2\,c\,x)}\,dx-\frac{2\,c}{q}\int \frac{1}{\sqrt{d+e\,x}\,(b+q+2\,c\,x)}\,dx$$

Program code:

```
(* Int[1/(Sqrt[d_.+e_.*x_]*(a_.+b_.*x_+c_.*x_^2)),x_Symbol] :=
  With[{q=Rt[b^2-4*a*c,2]},
  2*c/q*Int[1/(Sqrt[d+e*x]*(b-q+2*c*x)),x] -
  2*c/q*Int[1/(Sqrt[d+e*x]*(b+q+2*c*x)),x]] /;
FreeQ[{a,b,c,d,e},x] && NeQ[b^2-4*a*c,0] && NeQ[c*d^2-b*d*e+a*e^2,0] && NeQ[2*c*d-b*e,0] (* && Not[LtQ[b^2-4*a*c,0]] *) *)
```

```
(* Int[1/(Sqrt[d_+e_.*x_]*(a_+c_.*x_^2)),x_Symbol] :=
  With[{q=Rt[-a*c,2]},
  c/(2*q)*Int[1/(Sqrt[d+e*x]*(-q+c*x)),x] -
  c/(2*q)*Int[1/(Sqrt[d+e*x]*(q+c*x)),x]] /;
FreeQ[{a,c,d,e},x] && NeQ[c*d^2+a*e^2,0] (* && Not[LtQ[-a*c,0]] *) *)
```

2: $\int \frac{1}{\sqrt{d+e\,x}\,\left(a+b\,x+c\,x^2\right)}\,dx$ when $b^2-4\,a\,c\neq 0\,\wedge\,c\,d^2-b\,d\,e+a\,e^2\neq 0\,\wedge\,2\,c\,d-b\,e\neq 0$

Derivation: Integration by substitution

Basis: $(d+e\,x)^m\,F[x]=\frac{2}{e}\,\text{Subst}\left[x^{2\,m+1}\,F\left[\frac{-d+x^2}{e}\right],\,x,\,\sqrt{d+e\,x}\right]\,\partial_x\sqrt{d+e\,x}$

Rule 1.2.1.2.6.2.2: If $b^2-4\,a\,c\neq 0\,\wedge\,c\,d^2-b\,d\,e+a\,e^2\neq 0\,\wedge\,2\,c\,d-b\,e\neq 0$

$$\int \frac{1}{\sqrt{d+e\,x}\,\left(a+b\,x+c\,x^2\right)}\,dx\,\longrightarrow\,2\,e\,\text{Subst}\left[\int \frac{1}{c\,d^2-b\,d\,e+a\,e^2-(2\,c\,d-b\,e)\,x^2+c\,x^4}\,dx,\,x,\,\sqrt{d+e\,x}\right]$$

Program code:

```
Int[1/(Sqrt[d_.+e_.*x_]*(a_.+b_.*x_+c_.*x_^2)),x_Symbol] :=
  2*e*Subst[Int[1/(c*d^2-b*d*e+a*e^2-(2*c*d-b*e)*x^2+c*x^4),x],x,Sqrt[d+e*x]] /;
FreeQ[{a,b,c,d,e},x] && NeQ[b^2-4*a*c,0] && NeQ[c*d^2-b*d*e+a*e^2,0] && NeQ[2*c*d-b*e,0]
```

```
Int[1/(Sqrt[d_+e_.*x_]*(a_+c_.*x_^2)),x_Symbol] :=
  2*e*Subst[Int[1/(c*d^2+a*e^2-2*c*d*x^2+c*x^4),x],x,Sqrt[d+e*x]] /;
FreeQ[{a,c,d,e},x] && NeQ[c*d^2+a*e^2,0]
```

3: $\int \frac{(d+e\,x)^m}{a+b\,x+c\,x^2}\,dx$ when $b^2-4\,a\,c\neq 0\,\wedge\,c\,d^2-b\,d\,e+a\,e^2\neq 0\,\wedge\,2\,c\,d-b\,e\neq 0\,\wedge\,m<-1$

Reference: G&R 2.176, CRC 123

Derivation: Quadratic recurrence 3b with $A=1$, $B=0$ and $p=-1$

Rule 1.2.1.2.6.2.3: If $b^2-4\,a\,c\neq 0\,\wedge\,c\,d^2-b\,d\,e+a\,e^2\neq 0\,\wedge\,2\,c\,d-b\,e\neq 0\,\wedge\,m<-1$**, then**

$$\int \frac{(d+e\,x)^m}{a+b\,x+c\,x^2}\,dx\,\longrightarrow\,\frac{e\,(d+e\,x)^{m+1}}{(m+1)\,\left(c\,d^2-b\,d\,e+a\,e^2\right)}+\frac{1}{c\,d^2-b\,d\,e+a\,e^2}\int \frac{(d+e\,x)^{m+1}\,(c\,d-b\,e-c\,e\,x)}{a+b\,x+c\,x^2}\,dx$$

Program code:

```
Int[(d_.+e_.*x_)^m_/(a_.+b_.*x_+c_.*x_^2),x_Symbol] :=
  e*(d+e*x)^(m+1)/((m+1)*(c*d^2-b*d*e+a*e^2)) +
  1/(c*d^2-b*d*e+a*e^2)*Int[(d+e*x)^(m+1)*Simp[c*d-b*e-c*e*x,x]/(a+b*x+c*x^2),x] /;
FreeQ[{a,b,c,d,e,m},x] && NeQ[b^2-4*a*c,0] && NeQ[c*d^2-b*d*e+a*e^2,0] && NeQ[2*c*d-b*e,0] && LtQ[m,-1]
```

```
Int[(d_+e_.*x_)^m_/(a_+c_.*x_^2),x_Symbol] :=
  e*(d+e*x)^(m+1)/((m+1)*(c*d^2+a*e^2)) +
  c/(c*d^2+a*e^2)*Int[(d+e*x)^(m+1)*(d-e*x)/(a+c*x^2),x] /;
FreeQ[{a,c,d,e,m},x] && NeQ[c*d^2+a*e^2,0] && LtQ[m,-1]
```

3: $\int \frac{(d+e x)^m}{a+b x+c x^2} \, dx$ when $b^2-4 a c \neq 0 \wedge c d^2-b d e+a e^2 \neq 0 \wedge 2 c d-b e \neq 0 \wedge m \notin \mathbb{Z}$

Derivation: Algebraic expansion

- **Basis: If** $q=\sqrt{b^2-4 a c}$, **then** $\frac{1}{a+b z+c z^2} = \frac{2 c}{q (b-q+2 c z)} - \frac{2 c}{q (b+q+2 c z)}$

- **Rule 1.2.1.2.6.3: If** $b^2-4 a c \neq 0 \wedge c d^2-b d e+a e^2 \neq 0 \wedge 2 c d-b e \neq 0 \wedge m \notin \mathbb{Z}$**, then**

$$\int \frac{(d+e x)^m}{a+b x+c x^2} \, dx \longrightarrow \int (d+e x)^m \operatorname{ExpandIntegrand}\left[\frac{1}{a+b x+c x^2}, x\right] dx$$

Program code:

```
Int[(d_.+e_.*x_)^m_/(a_.+b_.*x_+c_.*x_^2),x_Symbol] :=
  Int[ExpandIntegrand[(d+e*x)^m,1/(a+b*x+c*x^2),x],x] /;
FreeQ[{a,b,c,d,e,m},x] && NeQ[b^2-4*a*c,0] && NeQ[c*d^2-b*d*e+a*e^2,0] && NeQ[2*c*d-b*e,0] && Not[IntegerQ[m]]
```

```
Int[(d_+e_.*x_)^m_/(a_+c_.*x_^2),x_Symbol] :=
  Int[ExpandIntegrand[(d+e*x)^m,1/(a+c*x^2),x],x] /;
FreeQ[{a,c,d,e,m},x] && NeQ[c*d^2+a*e^2,0] && Not[IntegerQ[m]]
```

7: $\int (d+e x)^m \left(a+b x+c x^2\right)^p dx$ when $b^2-4 a c \neq 0 \wedge c d^2-b d e+a e^2 \neq 0 \wedge 2 c d-b e \neq 0 \wedge b d+a e == 0 \wedge c d+b e == 0 \wedge m-p \in \mathbb{Z}$

Derivation: Piecewise constant extraction

- **Basis: If** $b d+a e == 0 \wedge c d+b e == 0$, **then** $\partial_x \frac{(d+e x)^p \left(a+b x+c x^2\right)^p}{\left(a d+c e x^3\right)^p} == 0$

Rule 1.2.1.2.7: If $b d+a e == 0 \wedge c d+b e == 0 \wedge m-p \in \mathbb{Z}$, **then**

$$\int (d+e x)^m \left(a+b x+c x^2\right)^p dx \longrightarrow \frac{(d+e x)^{\text{FracPart}[p]} \left(a+b x+c x^2\right)^{\text{FracPart}[p]}}{\left(a d+c e x^3\right)^{\text{FracPart}[p]}} \int (d+e x)^{m-p} \left(a d+c e x^3\right)^p dx$$

Program code:

```
Int[(d_.+e_.*x_)^m_*(a_+b_.*x_+c_.*x_^2)^p_,x_Symbol] :=
  (d+e*x)^FracPart[p]*(a+b*x+c*x^2)^FracPart[p]/(a*d+c*e*x^3)^FracPart[p]*Int[(d+e*x)^(m-p)*(a*d+c*e*x^3)^p,x] /;
FreeQ[{a,b,c,d,e,m,p},x] && EqQ[b*d+a*e,0] && EqQ[c*d+b*e,0] && IGtQ[m-p+1,0] && Not[IntegerQ[p]]
```

8. $\int \frac{(d+e\,x)^m}{\sqrt{a+b\,x+c\,x^2}}\,dx$ when $b^2-4\,a\,c\neq 0\,\bigwedge\,c\,d^2-b\,d\,e+a\,e^2\neq 0\,\bigwedge\,2\,c\,d-b\,e\neq 0\,\bigwedge\,m^2==\frac{1}{4}$

1. $\int \frac{(d+e\,x)^m}{\sqrt{b\,x+c\,x^2}}\,dx$ when $c\,d-b\,e\neq 0\,\bigwedge\,2\,c\,d-b\,e\neq 0\,\bigwedge\,m^2==\frac{1}{4}$

1: $\int \frac{(d+e\,x)^m}{\sqrt{b\,x+c\,x^2}}\,dx$ when $c\,d-b\,e\neq 0\,\bigwedge\,2\,c\,d-b\,e\neq 0\,\bigwedge\,m^2==\frac{1}{4}\,\bigwedge\,c<0\,\bigwedge\,b\in\mathbb{R}$

- **Derivation: Algebraic expansion**

- **Basis: If** $c<0\,\wedge\,b>0$, **then** $\sqrt{b\,x+c\,x^2}==\sqrt{x}\,\sqrt{b+c\,x}$

- **Basis: If** $c<0\,\wedge\,b<0$, **then** $\sqrt{b\,x+c\,x^2}==\sqrt{-x}\,\sqrt{-b-c\,x}$

- **Basis: If** $c<0\,\wedge\,b\in\mathbb{R}$, **then** $\sqrt{b\,x+c\,x^2}==\sqrt{b\,x}\,\sqrt{1+\frac{c\,x}{b}}$

- **Rule 1.2.1.2.8.1.1: If** $c\,d-b\,e\neq 0\,\bigwedge\,2\,c\,d-b\,e\neq 0\,\bigwedge\,m^2==\frac{1}{4}\,\bigwedge\,c<0\,\bigwedge\,b\in\mathbb{R}$, **then**

$$\int \frac{(d+e\,x)^m}{\sqrt{b\,x+c\,x^2}}\,dx\,\longrightarrow\,\int \frac{(d+e\,x)^m}{\sqrt{b\,x}\,\sqrt{1+\frac{c\,x}{b}}}\,dx$$

Program code:

```
Int[(d_.+e_.*x_)^m_/Sqrt[b_.*x_+c_.*x_^2],x_Symbol] :=
  Int[(d+e*x)^m/(Sqrt[b*x]*Sqrt[1+c/b*x]),x] /;
FreeQ[{b,c,d,e},x] && NeQ[c*d-b*e,0] && NeQ[2*c*d-b*e,0] && EqQ[m^2,1/4] && LtQ[c,0] && RationalQ[b]
```

2: $\int \frac{(d+e\,x)^m}{\sqrt{b\,x+c\,x^2}}\,dx$ when $c\,d-b\,e\neq 0 \bigwedge 2\,c\,d-b\,e\neq 0 \bigwedge m^2==\frac{1}{4}$

Derivation: Piecewise constant extraction

- **Basis:** $\partial_x \frac{\sqrt{x}\,\sqrt{b+c\,x}}{\sqrt{b\,x+c\,x^2}}==0$

- **Rule 1.2.1.2.8.1.2: If** $c\,d-b\,e\neq 0 \bigwedge 2\,c\,d-b\,e\neq 0 \bigwedge m^2==\frac{1}{4}$**, then**

$$\int \frac{(d+e\,x)^m}{\sqrt{b\,x+c\,x^2}}\,dx \longrightarrow \frac{\sqrt{x}\,\sqrt{b+c\,x}}{\sqrt{b\,x+c\,x^2}}\int \frac{(d+e\,x)^m}{\sqrt{x}\,\sqrt{b+c\,x}}\,dx$$

Program code:

```
Int[(d_.+e_.*x_)^m_/Sqrt[b_.*x_+c_.*x_^2],x_Symbol] :=
  Sqrt[x]*Sqrt[b+c*x]/Sqrt[b*x+c*x^2]*Int[(d+e*x)^m/(Sqrt[x]*Sqrt[b+c*x]),x] /;
FreeQ[{b,c,d,e},x] && NeQ[c*d-b*e,0] && NeQ[2*c*d-b*e,0] && EqQ[m^2,1/4]
```

2. $\int \frac{(e\,x)^m}{\sqrt{a+b\,x+c\,x^2}}\,dx$ when $b^2-4\,a\,c\neq 0 \bigwedge m^2==\frac{1}{4}$

1: $\int \frac{x^m}{\sqrt{a+b\,x+c\,x^2}}\,dx$ when $b^2-4\,a\,c\neq 0 \bigwedge m^2==\frac{1}{4}$

Derivation: Integration by substitution

- **Basis:** $x^m\,F[x]==2\,\text{Subst}\left[x^{2\,m+1}\,F\left[x^2\right],\,x,\,\sqrt{x}\right]\,\partial_x\sqrt{x}$

- **Rule 1.2.1.2.8.2.1: If** $b^2-4\,a\,c\neq 0 \bigwedge m^2==\frac{1}{4}$**, then**

$$\int \frac{x^m}{\sqrt{a+b\,x+c\,x^2}}\,dx \longrightarrow 2\,\text{Subst}\left[\int \frac{x^{2\,m+1}}{\sqrt{a+b\,x^2+c\,x^4}}\,dx,\,x,\,\sqrt{x}\right]$$

Program code:

```
Int[x_^m_/Sqrt[a_+b_.*x_+c_.*x_^2],x_Symbol] :=
  2*Subst[Int[x^(2*m+1)/Sqrt[a+b*x^2+c*x^4],x],x,Sqrt[x]] /;
FreeQ[{a,b,c},x] && NeQ[b^2-4*a*c,0] && EqQ[m^2,1/4]
```

2: $\int \frac{(e\,x)^m}{\sqrt{a+b\,x+c\,x^2}}\,dx$ when $b^2-4\,a\,c\neq 0\ \land\ m^2==\frac{1}{4}$

Derivation: Piecewise constant extraction

- **Basis:** $\partial_x \frac{(e\,x)^m}{x^m}==0$
- **Rule 1.2.1.2.8.2.2: If** $b^2-4\,a\,c\neq 0\ \land\ m^2==\frac{1}{4}$**, then**

$$\int \frac{(e\,x)^m}{\sqrt{a+b\,x+c\,x^2}}\,dx \longrightarrow \frac{(e\,x)^m}{x^m}\int \frac{x^m}{\sqrt{a+b\,x+c\,x^2}}\,dx$$

```
Int[(e_*x_)^m_/Sqrt[a_+b_.*x_+c_.*x_^2],x_Symbol] :=
  (e*x)^m/x^m*Int[x^m/Sqrt[a+b*x+c*x^2], x] /;
FreeQ[{a,b,c,e},x] && NeQ[b^2-4*a*c,0] && EqQ[m^2,1/4]
```

3: $\int \frac{(d+e\,x)^m}{\sqrt{a+b\,x+c\,x^2}}\,dx$ when $b^2-4\,a\,c\neq 0\ \land\ c\,d^2-b\,d\,e+a\,e^2\neq 0\ \land\ 2\,c\,d-b\,e\neq 0\ \land\ m^2==\frac{1}{4}$

Derivation: Piecewise constant extraction and integration by substitution

- **Basis:** $\partial_x \frac{(d+e\,x)^m\sqrt{-\frac{c\,(a+b\,x+c\,x^2)}{b^2-4\,a\,c}}}{\sqrt{a+b\,x+c\,x^2}\left(\frac{2\,c\,(d+e\,x)}{2\,c\,d-b\,e-e\,\sqrt{b^2-4\,a\,c}}\right)^m}==0$
- **Basis:** $\frac{\left(\frac{2\,c\,(d+e\,x)}{2\,c\,d-b\,e-e\,\sqrt{b^2-4\,a\,c}}\right)^m}{\sqrt{-\frac{c\,(a+b\,x+c\,x^2)}{b^2-4\,a\,c}}}==\frac{2\,\sqrt{b^2-4\,a\,c}}{c}\,\text{Subst}\left[\frac{\left(1+\frac{2\,e\,\sqrt{b^2-4\,a\,c}\;x^2}{2\,c\,d-b\,e-e\,\sqrt{b^2-4\,a\,c}}\right)^m}{\sqrt{1-x^2}},\,x,\,\sqrt{\frac{b+\sqrt{b^2-4\,a\,c}+2\,c\,x}{2\,\sqrt{b^2-4\,a\,c}}}\right]\partial_x\sqrt{\frac{b+\sqrt{b^2-4\,a\,c}+2\,c\,x}{2\,\sqrt{b^2-4\,a\,c}}}$
- **Rule 1.2.1.2.8.3: If** $b^2-4\,a\,c\neq 0\ \land\ c\,d^2-b\,d\,e+a\,e^2\neq 0\ \land\ 2\,c\,d-b\,e\neq 0\ \land\ m^2==\frac{1}{4}$**, then**

$$\int \frac{(d+e\,x)^m}{\sqrt{a+b\,x+c\,x^2}}\,dx \longrightarrow \frac{(d+e\,x)^m\sqrt{-\frac{c\,(a+b\,x+c\,x^2)}{b^2-4\,a\,c}}}{\sqrt{a+b\,x+c\,x^2}\left(\frac{2\,c\,(d+e\,x)}{2\,c\,d-b\,e-e\,\sqrt{b^2-4\,a\,c}}\right)^m}\int \frac{\left(\frac{2\,c\,(d+e\,x)}{2\,c\,d-b\,e-e\,\sqrt{b^2-4\,a\,c}}\right)^m}{\sqrt{-\frac{c\,(a+b\,x+c\,x^2)}{b^2-4\,a\,c}}}\,dx$$

$$\to \frac{2\sqrt{b^2-4ac}\,(d+ex)^m\sqrt{-\frac{c(a+bx+cx^2)}{b^2-4ac}}}{c\sqrt{a+bx+cx^2}\left(\frac{2c(d+ex)}{2cd-be-e\sqrt{b^2-4ac}}\right)^m}\,\text{Subst}\left[\int\frac{\left(1+\frac{2e\sqrt{b^2-4ac}\,x^2}{2cd-be-e\sqrt{b^2-4ac}}\right)^m}{\sqrt{1-x^2}}\,dx,\,x,\,\sqrt{\frac{b+\sqrt{b^2-4ac}+2cx}{2\sqrt{b^2-4ac}}}\right]$$

- **Rule 1.2.1.2.8.3: If $c d^2 + a e^2 \neq 0 \wedge m^2 == \frac{1}{4}$, then**

$$\int\frac{(d+ex)^m}{\sqrt{a+cx^2}}\,dx \to \frac{(d+ex)^m\sqrt{1+\frac{cx^2}{a}}}{\sqrt{a+cx^2}\left(\frac{c(d+ex)}{cd-ae\sqrt{-c/a}}\right)^m}\int\frac{\left(\frac{c(d+ex)}{cd-ae\sqrt{-c/a}}\right)^m}{\sqrt{\frac{a+cx^2}{a}}}\,dx$$

$$\to \frac{2a\sqrt{-c/a}\,(d+ex)^m\sqrt{1+\frac{cx^2}{a}}}{c\sqrt{a+cx^2}\left(\frac{c(d+ex)}{cd-ae\sqrt{-c/a}}\right)^m}\,\text{Subst}\left[\int\frac{\left(1+\frac{2ae\sqrt{-c/a}\,x^2}{cd-ae\sqrt{-c/a}}\right)^m}{\sqrt{1-x^2}}\,dx,\,x,\,\sqrt{\frac{1-\sqrt{-c/a}\,x}{2}}\right]$$

Program code:

```
Int[(d_.+e_.*x_)^m_/Sqrt[a_.+b_.*x_+c_.*x_^2],x_Symbol] :=
  2*Rt[b^2-4*a*c,2]*(d+e*x)^m*Sqrt[-c*(a+b*x+c*x^2)/(b^2-4*a*c)]/
    (c*Sqrt[a+b*x+c*x^2]*(2*c*(d+e*x)/(2*c*d-b*e-e*Rt[b^2-4*a*c,2]))^m)*
   Subst[Int[(1+2*e*Rt[b^2-4*a*c,2]*x^2/(2*c*d-b*e-e*Rt[b^2-4*a*c,2]))^m/Sqrt[1-x^2],x],x,
     Sqrt[(b+Rt[b^2-4*a*c,2]+2*c*x)/(2*Rt[b^2-4*a*c,2])]] /;
FreeQ[{a,b,c,d,e},x] && NeQ[b^2-4*a*c,0] && NeQ[c*d^2-b*d*e+a*e^2,0] && NeQ[2*c*d-b*e,0] && EqQ[m^2,1/4]
```

```
Int[(d_+e_.*x_)^m_/Sqrt[a_+c_.*x_^2],x_Symbol] :=
  2*a*Rt[-c/a,2]*(d+e*x)^m*Sqrt[1+c*x^2/a]/(c*Sqrt[a+c*x^2]*(c*(d+e*x)/(c*d-a*e*Rt[-c/a,2]))^m)*
   Subst[Int[(1+2*a*e*Rt[-c/a,2]*x^2/(c*d-a*e*Rt[-c/a,2]))^m/Sqrt[1-x^2],x],x,Sqrt[(1-Rt[-c/a,2]*x)/2]] /;
FreeQ[{a,c,d,e},x] && NeQ[c*d^2+a*e^2,0] && EqQ[m^2,1/4]
```

9. $\int(d+ex)^m\left(a+bx+cx^2\right)^p\,dx$ when $b^2-4ac\neq 0 \wedge cd^2-bde+ae^2\neq 0 \wedge 2cd-be\neq 0 \wedge m+2p+2==0$

1: $\int(d+ex)^m\left(a+bx+cx^2\right)^p\,dx$ when $b^2-4ac\neq 0 \wedge cd^2-bde+ae^2\neq 0 \wedge 2cd-be\neq 0 \wedge m+2p+2==0 \wedge p>0$

Derivation: Quadratic recurrence 2a with $A = d$, $B = e$, $m = m-1$ and $m+2p+2 == 0$ inverted

Rule 1.2.1.2.9.1: If $b^2-4ac\neq 0 \wedge cd^2-bde+ae^2\neq 0 \wedge 2cd-be\neq 0 \wedge m+2p+2==0 \wedge p>0 \wedge p\notin\mathbb{Z}$, then

$$\int (d+e\,x)^m \left(a+b\,x+c\,x^2\right)^p dx \longrightarrow$$
$$-\frac{(d+e\,x)^{m+1}\,(d\,b-2\,a\,e+(2\,c\,d-b\,e)\,x)\left(a+b\,x+c\,x^2\right)^p}{2\,(m+1)\left(c\,d^2-b\,d\,e+a\,e^2\right)}+\frac{p\left(b^2-4\,a\,c\right)}{2\,(m+1)\left(c\,d^2-b\,d\,e+a\,e^2\right)}\int (d+e\,x)^{m+2}\left(a+b\,x+c\,x^2\right)^{p-1} dx$$

Program code:

```
Int[(d_.+e_.*x_)^m_*(a_.+b_.*x_+c_.*x_^2)^p_,x_Symbol] :=
  -(d+e*x)^(m+1)*(d*b-2*a*e+(2*c*d-b*e)*x)*(a+b*x+c*x^2)^p/(2*(m+1)*(c*d^2-b*d*e+a*e^2)) +
  p*(b^2-4*a*c)/(2*(m+1)*(c*d^2-b*d*e+a*e^2))*Int[(d+e*x)^(m+2)*(a+b*x+c*x^2)^(p-1),x] /;
FreeQ[{a,b,c,d,e},x] && NeQ[b^2-4*a*c,0] && NeQ[c*d^2-b*d*e+a*e^2,0] && NeQ[2*c*d-b*e,0] && EqQ[m+2*p+2,0] && GtQ[p,0]
```

```
Int[(d_+e_.*x_)^m_*(a_+c_.*x_^2)^p_,x_Symbol] :=
  -(d+e*x)^(m+1)*(-2*a*e+(2*c*d)*x)*(a+c*x^2)^p/(2*(m+1)*(c*d^2+a*e^2)) -
  4*a*c*p/(2*(m+1)*(c*d^2+a*e^2))*Int[(d+e*x)^(m+2)*(a+c*x^2)^(p-1),x] /;
FreeQ[{a,c,d,e},x] && NeQ[c*d^2+a*e^2,0] && EqQ[m+2*p+2,0] && GtQ[p,0]
```

2: $\int (d+e\,x)^m \left(a+b\,x+c\,x^2\right)^p dx$ when $b^2-4\,a\,c\neq 0\,\land\,c\,d^2-b\,d\,e+a\,e^2\neq 0\,\land\,2\,c\,d-b\,e\neq 0\,\land\,m+2\,p+2==0\,\land\,p<-1$

Derivation: Quadratic recurrence 2a with $A=d$, $B=e$, $m=m-1$ and $m+2\,p+2==0$

Rule 1.2.1.2.9.2: If $b^2-4\,a\,c\neq 0\,\land\,c\,d^2-b\,d\,e+a\,e^2\neq 0\,\land\,2\,c\,d-b\,e\neq 0\,\land\,m+2\,p+2==0\,\land\,p<-1$, then

$$\int (d+e\,x)^m \left(a+b\,x+c\,x^2\right)^p dx \longrightarrow$$
$$\frac{(d+e\,x)^{m-1}\,(d\,b-2\,a\,e+(2\,c\,d-b\,e)\,x)\left(a+b\,x+c\,x^2\right)^{p+1}}{(p+1)\left(b^2-4\,a\,c\right)}-\frac{2\,(2\,p+3)\left(c\,d^2-b\,d\,e+a\,e^2\right)}{(p+1)\left(b^2-4\,a\,c\right)}\int (d+e\,x)^{m-2}\left(a+b\,x+c\,x^2\right)^{p+1} dx$$

Program code:

```
Int[(d_.+e_.*x_)^m_*(a_.+b_.*x_+c_.*x_^2)^p_,x_Symbol] :=
  (d+e*x)^(m-1)*(d*b-2*a*e+(2*c*d-b*e)*x)*(a+b*x+c*x^2)^(p+1)/((p+1)*(b^2-4*a*c)) -
  2*(2*p+3)*(c*d^2-b*d*e+a*e^2)/((p+1)*(b^2-4*a*c))*Int[(d+e*x)^(m-2)*(a+b*x+c*x^2)^(p+1),x] /;
FreeQ[{a,b,c,d,e},x] && NeQ[b^2-4*a*c,0] && NeQ[c*d^2-b*d*e+a*e^2,0] && NeQ[2*c*d-b*e,0] && EqQ[m+2*p+2,0] && LtQ[p,-1]
```

```
Int[(d_+e_.*x_)^m_*(a_+c_.*x_^2)^p_,x_Symbol] :=
  (d+e*x)^(m-1)*(a*e-c*d*x)*(a+c*x^2)^(p+1)/(2*a*c*(p+1)) +
  (2*p+3)*(c*d^2+a*e^2)/(2*a*c*(p+1))*Int[(d+e*x)^(m-2)*(a+c*x^2)^(p+1),x] /;
FreeQ[{a,c,d,e},x] && NeQ[c*d^2+a*e^2,0] && EqQ[m+2*p+2,0] && LtQ[p,-1]
```

3: $\int \frac{1}{(d+e x)\sqrt{a+b x+c x^2}}\,dx$ when $b^2-4 a c \neq 0 \land 2 c d-b e \neq 0$

Reference: G&R 2.266.1, CRC 258

Reference: G&R 2.266.3, CRC 259

Derivation: Integration by substitution

Basis: $\frac{1}{(d+e x)\sqrt{a+b x+c x^2}} = -2\,\text{Subst}\left[\frac{1}{4 c d^2-4 b d e+4 a e^2-x^2},\ x,\ \frac{2 a e-b d-(2 c d-b e) x}{\sqrt{a+b x+c x^2}}\right] \partial_x \frac{2 a e-b d-(2 c d-b e) x}{\sqrt{a+b x+c x^2}}$

Rule 1.2.1.2.9.3: If $b^2-4 a c \neq 0 \land 2 c d-b e \neq 0$**, then**

$$\int \frac{1}{(d+e x)\sqrt{a+b x+c x^2}}\,dx \to -2\,\text{Subst}\left[\int \frac{1}{4 c d^2-4 b d e+4 a e^2-x^2}\,dx,\ x,\ \frac{2 a e-b d-(2 c d-b e) x}{\sqrt{a+b x+c x^2}}\right]$$

Program code:

```
Int[1/((d_.+e_.*x_)*Sqrt[a_.+b_.*x_+c_.*x_^2]),x_Symbol] :=
  -2*Subst[Int[1/(4*c*d^2-4*b*d*e+4*a*e^2-x^2),x],x,(2*a*e-b*d-(2*c*d-b*e)*x)/Sqrt[a+b*x+c*x^2]] /;
FreeQ[{a,b,c,d,e},x] && NeQ[b^2-4*a*c,0] && NeQ[2*c*d-b*e,0]
```

```
Int[1/((d_+e_.*x_)*Sqrt[a_+c_.*x_^2]),x_Symbol] :=
  -Subst[Int[1/(c*d^2+a*e^2-x^2),x],x,(a*e-c*d*x)/Sqrt[a+c*x^2]] /;
FreeQ[{a,c,d,e},x]
```

4: $\int (d+e x)^m \left(a+b x+c x^2\right)^p\,dx$ when $b^2-4 a c \neq 0 \land c d^2-b d e+a e^2 \neq 0 \land 2 c d-b e \neq 0 \land m+2 p+2 == 0$

Rule 1.2.1.2.9.4: If $b^2-4 a c \neq 0 \land c d^2-b d e+a e^2 \neq 0 \land 2 c d-b e \neq 0 \land p \notin \mathbb{Z} \land m+2 p+2 == 0$**, then**

$$\int (d+e x)^m \left(a+b x+c x^2\right)^p\,dx \to$$

$$-\left(\left(b-\sqrt{b^2-4 a c}+2 c x\right)(d+e x)^{m+1}\left(a+b x+c x^2\right)^p\right) \Big/$$

$$\left((m+1)\left(2 c d-b e+e\sqrt{b^2-4 a c}\right)\left(\frac{\left(2 c d-b e+e\sqrt{b^2-4 a c}\right)\left(b+\sqrt{b^2-4 a c}+2 c x\right)}{\left(2 c d-b e-e\sqrt{b^2-4 a c}\right)\left(b-\sqrt{b^2-4 a c}+2 c x\right)}\right)^p\right)$$

$$\text{Hypergeometric2F1}\left[m+1,\,-p,\,m+2,\,-\frac{4\,c\,\sqrt{b^2-4\,a\,c}\,\,(d+e\,x)}{\left(2\,c\,d-b\,e-e\,\sqrt{b^2-4\,a\,c}\right)\,\left(b-\sqrt{b^2-4\,a\,c}\,+2\,c\,x\right)}\right]$$

Program code:

```
Int[(d_.+e_.*x_)^m_*(a_.+b_.*x_+c_.*x_^2)^p_,x_Symbol] :=
  -(b-Rt[b^2-4*a*c,2]+2*c*x)*(d+e*x)^(m+1)*(a+b*x+c*x^2)^p/
    ((m+1)*(2*c*d-b*e+e*Rt[b^2-4*a*c,2])*
      ((2*c*d-b*e+e*Rt[b^2-4*a*c,2])*(b+Rt[b^2-4*a*c,2]+2*c*x)/((2*c*d-b*e-e*Rt[b^2-4*a*c,2])*(b-Rt[b^2-4*a*c,2]+2*c*x)))^p)*
    Hypergeometric2F1[m+1,-p,m+2,-4*c*Rt[b^2-4*a*c,2]*(d+e*x)/((2*c*d-b*e-e*Rt[b^2-4*a*c,2])*(b-Rt[b^2-4*a*c,2]+2*c*x))] /;
FreeQ[{a,b,c,d,e,m,p},x] && NeQ[b^2-4*a*c,0] && NeQ[c*d^2-b*d*e+a*e^2,0] && NeQ[2*c*d-b*e,0] && Not[IntegerQ[p]] && EqQ[m+2*p+2,0]
```

```
Int[(d_+e_.*x_)^m_*(a_+c_.*x_^2)^p_,x_Symbol] :=
  (Rt[-a*c,2]-c*x)*(d+e*x)^(m+1)*(a+c*x^2)^p/
    ((m+1)*(c*d+e*Rt[-a*c,2])*((c*d+e*Rt[-a*c,2])*(Rt[-a*c,2]+c*x)/((c*d-e*Rt[-a*c,2])*(-Rt[-a*c,2]+c*x)))^p)*
    Hypergeometric2F1[m+1,-p,m+2,2*c*Rt[-a*c,2]*(d+e*x)/((c*d-e*Rt[-a*c,2])*(Rt[-a*c,2]-c*x))] /;
FreeQ[{a,c,d,e,m,p},x] && NeQ[c*d^2+a*e^2,0] && Not[IntegerQ[p]] && EqQ[m+2*p+2,0]
```

10. $\int (d+e\,x)^m\,\left(a+b\,x+c\,x^2\right)^p\,dx$ when $b^2-4\,a\,c\neq 0\,\wedge\,c\,d^2-b\,d\,e+a\,e^2\neq 0\,\wedge\,2\,c\,d-b\,e\neq 0\,\wedge\,m+2\,p+3==0$

1: $\int (d+e\,x)^m\,\left(a+b\,x+c\,x^2\right)^p\,dx$ when $b^2-4\,a\,c\neq 0\,\wedge\,c\,d^2-b\,d\,e+a\,e^2\neq 0\,\wedge\,2\,c\,d-b\,e\neq 0\,\wedge\,m+2\,p+3==0\,\wedge\,p<-1$

Derivation: Quadratic recurrence 2a with $A=1$, $B=0$ and $m+2\,p+3==0$

Rule 1.2.1.2.10.1: If $b^2-4\,a\,c\neq 0\,\wedge\,c\,d^2-b\,d\,e+a\,e^2\neq 0\,\wedge\,2\,c\,d-b\,e\neq 0\,\wedge\,m+2\,p+3==0\,\wedge\,p<-1$, then

$$\int (d+e\,x)^m\,\left(a+b\,x+c\,x^2\right)^p\,dx\;\longrightarrow$$

$$\frac{(d+e\,x)^m\,(b+2\,c\,x)\,\left(a+b\,x+c\,x^2\right)^{p+1}}{(p+1)\,\left(b^2-4\,a\,c\right)}+\frac{m\,(2\,c\,d-b\,e)}{(p+1)\,\left(b^2-4\,a\,c\right)}\int (d+e\,x)^{m-1}\,\left(a+b\,x+c\,x^2\right)^{p+1}\,dx$$

Program code:

```
Int[(d_.+e_.*x_)^m_*(a_.+b_.*x_+c_.*x_^2)^p_,x_Symbol] :=
  (d+e*x)^m*(b+2*c*x)*(a+b*x+c*x^2)^(p+1)/((p+1)*(b^2-4*a*c)) +
  m*(2*c*d-b*e)/((p+1)*(b^2-4*a*c))*Int[(d+e*x)^(m-1)*(a+b*x+c*x^2)^(p+1),x] /;
FreeQ[{a,b,c,d,e,m,p},x] && NeQ[b^2-4*a*c,0] && NeQ[c*d^2-b*d*e+a*e^2,0] && NeQ[2*c*d-b*e,0] && EqQ[m+2*p+3,0] && LtQ[p,-1]
```

```
Int[(d_+e_.*x_)^m_*(a_+c_.*x_^2)^p_,x_Symbol] :=
  -(d+e*x)^m*(2*c*x)*(a+c*x^2)^(p+1)/(4*a*c*(p+1)) -
  m*(2*c*d)/(4*a*c*(p+1))*Int[(d+e*x)^(m-1)*(a+c*x^2)^(p+1),x] /;
FreeQ[{a,c,d,e,m,p},x] && NeQ[c*d^2+a*e^2,0] && EqQ[m+2*p+3,0] && LtQ[p,-1]
```

2: $\int (d+e\,x)^m\,\left(a+b\,x+c\,x^2\right)^p\,dx$ when $b^2-4\,a\,c\neq 0\,\wedge\,c\,d^2-b\,d\,e+a\,e^2\neq 0\,\wedge\,2\,c\,d-b\,e\neq 0\,\wedge\,m+2\,p+3==0\,\wedge\,p\nleq -1$

Reference: G&R 2.176, CRC 123

Derivation: Quadratic recurrence 3b with $A=1$, $B=0$ and $m+2\,p+3==0$

Rule 1.2.1.2.10.2: If $b^2-4\,a\,c\neq 0\,\wedge\,c\,d^2-b\,d\,e+a\,e^2\neq 0\,\wedge\,2\,c\,d-b\,e\neq 0\,\wedge\,m+2\,p+3==0\,\wedge\,p\nleq -1$, then

$$\int (d+e\,x)^m\,\left(a+b\,x+c\,x^2\right)^p\,dx\,\longrightarrow$$

$$\frac{e\,(d+e\,x)^{m+1}\,\left(a+b\,x+c\,x^2\right)^{p+1}}{(m+1)\,\left(c\,d^2-b\,d\,e+a\,e^2\right)}+\frac{(2\,c\,d-b\,e)}{2\,\left(c\,d^2-b\,d\,e+a\,e^2\right)}\int (d+e\,x)^{m+1}\,\left(a+b\,x+c\,x^2\right)^p\,dx$$

Program code:

```
Int[(d_.+e_.*x_)^m_*(a_.+b_.*x_+c_.*x_^2)^p_,x_Symbol] :=
  e*(d+e*x)^(m+1)*(a+b*x+c*x^2)^(p+1)/((m+1)*(c*d^2-b*d*e+a*e^2)) +
  (2*c*d-b*e)/(2*(c*d^2-b*d*e+a*e^2))*Int[(d+e*x)^(m+1)*(a+b*x+c*x^2)^p,x] /;
FreeQ[{a,b,c,d,e,m,p},x] && NeQ[b^2-4*a*c,0] && NeQ[c*d^2-b*d*e+a*e^2,0] && NeQ[2*c*d-b*e,0] && EqQ[m+2*p+3,0]
```

```
Int[(d_+e_.*x_)^m_*(a_+c_.*x_^2)^p_,x_Symbol] :=
  e*(d+e*x)^(m+1)*(a+c*x^2)^(p+1)/((m+1)*(c*d^2+a*e^2)) +
  c*d/(c*d^2+a*e^2)*Int[(d+e*x)^(m+1)*(a+c*x^2)^p,x] /;
FreeQ[{a,c,d,e,m,p},x] && NeQ[c*d^2+a*e^2,0] && EqQ[m+2*p+3,0]
```

11. $\int (d+e\,x)^m\,\left(a+b\,x+c\,x^2\right)^p\,dx$ when $b^2-4\,a\,c\neq 0\,\wedge\,c\,d^2-b\,d\,e+a\,e^2\neq 0\,\wedge\,2\,c\,d-b\,e\neq 0\,\wedge\,p>0$

1: $\int (d+e\,x)^m\,\left(a+b\,x+c\,x^2\right)^p\,dx$ when $b^2-4\,a\,c\neq 0\,\wedge\,c\,d^2-b\,d\,e+a\,e^2\neq 0\,\wedge\,2\,c\,d-b\,e\neq 0\,\wedge\,p>0\,\wedge\,m<-1\,\wedge\,m+2\,p+1\notin\mathbb{Z}^-$

Derivation: Quadratic recurrence 1a with $A=1$ and $B=0$

Rule 1.2.1.2.11.1: If $b^2-4\,a\,c\neq 0\,\wedge\,c\,d^2-b\,d\,e+a\,e^2\neq 0\,\wedge\,2\,c\,d-b\,e\neq 0\,\wedge\,p>0\,\wedge\,m<-1\,\wedge\,m+2\,p+1\notin\mathbb{Z}^-$, then

$$\int (d+e\,x)^m\,\left(a+b\,x+c\,x^2\right)^p\,dx\,\longrightarrow$$

$$\frac{(d+e\,x)^{m+1}\,\left(a+b\,x+c\,x^2\right)^p}{e\,(m+1)}-\frac{p}{e\,(m+1)}\int(d+e\,x)^{m+1}\,(b+2\,c\,x)\,\left(a+b\,x+c\,x^2\right)^{p-1}\,dx$$

Program code:

```
Int[(d_.+e_.*x_)^m_*(a_.+b_.*x_+c_.*x_^2)^p_,x_Symbol] :=
  (d+e*x)^(m+1)*(a+b*x+c*x^2)^p/(e*(m+1)) -
  p/(e*(m+1))*Int[(d+e*x)^(m+1)*(b+2*c*x)*(a+b*x+c*x^2)^(p-1),x] /;
FreeQ[{a,b,c,d,e,m},x] && NeQ[b^2-4*a*c,0] && NeQ[c*d^2-b*d*e+a*e^2,0] && NeQ[2*c*d-b*e,0] && GtQ[p,0] &&
  (IntegerQ[p] || LtQ[m,-1]) && NeQ[m,-1] && Not[ILtQ[m+2*p+1,0]] && IntQuadraticQ[a,b,c,d,e,m,p,x]
```

```
Int[(d_+e_.*x_)^m_*(a_+c_.*x_^2)^p_,x_Symbol] :=
  (d+e*x)^(m+1)*(a+c*x^2)^p/(e*(m+1)) -
  2*c*p/(e*(m+1))*Int[x*(d+e*x)^(m+1)*(a+c*x^2)^(p-1),x] /;
FreeQ[{a,c,d,e,m},x] && NeQ[c*d^2+a*e^2,0] && GtQ[p,0] &&
  (IntegerQ[p] || LtQ[m,-1]) && NeQ[m,-1] && Not[ILtQ[m+2*p+1,0]] && IntQuadraticQ[a,0,c,d,e,m,p,x]
```

2: $\int(d+e\,x)^m\,\left(a+b\,x+c\,x^2\right)^p\,dx$ when $b^2-4\,a\,c\neq0\,\land\,c\,d^2-b\,d\,e+a\,e^2\neq0\,\land\,2\,c\,d-b\,e\neq0\,\land\,p>0\,\land\,m+2\,p\notin\mathbb{Z}^-$

Derivation: Quadratic recurrence 1b with $A=1$ and $B=0$

Derivation: Quadratic recurrence 1a with $A=d$, $B=e$ and $m=m-1$

–

Rule 1.2.1.2.11.2: If $b^2-4\,a\,c\neq0\,\land\,c\,d^2-b\,d\,e+a\,e^2\neq0\,\land\,2\,c\,d-b\,e\neq0\,\land\,p>0\,\land\,m+2\,p\notin\mathbb{Z}^-$, then

$$\int(d+e\,x)^m\,\left(a+b\,x+c\,x^2\right)^p\,dx\,\longrightarrow$$

$$\frac{(d+e\,x)^{m+1}\,\left(a+b\,x+c\,x^2\right)^p}{e\,(m+2\,p+1)}-\frac{p}{e\,(m+2\,p+1)}\int(d+e\,x)^m\,(b\,d-2\,a\,e+(2\,c\,d-b\,e)\,x)\,\left(a+b\,x+c\,x^2\right)^{p-1}\,dx$$

Program code:

```
Int[(d_.+e_.*x_)^m_*(a_.+b_.*x_+c_.*x_^2)^p_,x_Symbol] :=
  (d+e*x)^(m+1)*(a+b*x+c*x^2)^p/(e*(m+2*p+1)) -
  p/(e*(m+2*p+1))*Int[(d+e*x)^m*Simp[b*d-2*a*e+(2*c*d-b*e)*x,x]*(a+b*x+c*x^2)^(p-1),x] /;
FreeQ[{a,b,c,d,e,m},x] && NeQ[b^2-4*a*c,0] && NeQ[c*d^2-b*d*e+a*e^2,0] && NeQ[2*c*d-b*e,0] && GtQ[p,0] &&
  NeQ[m+2*p+1,0] && (Not[RationalQ[m]] || LtQ[m,1]) && Not[ILtQ[m+2*p,0]] && IntQuadraticQ[a,b,c,d,e,m,p,x]
```

```
Int[(d_+e_.*x_)^m_*(a_+c_.*x_^2)^p_,x_Symbol] :=
  (d+e*x)^(m+1)*(a+c*x^2)^p/(e*(m+2*p+1)) +
  2*p/(e*(m+2*p+1))*Int[(d+e*x)^m*Simp[a*e-c*d*x,x]*(a+c*x^2)^(p-1),x] /;
FreeQ[{a,c,d,e,m},x] && NeQ[c*d^2+a*e^2,0] && GtQ[p,0] &&
  NeQ[m+2*p+1,0] && (Not[RationalQ[m]] || LtQ[m,1]) && Not[ILtQ[m+2*p,0]] && IntQuadraticQ[a,0,c,d,e,m,p,x]
```

12. $\int (d+e\,x)^m \left(a+b\,x+c\,x^2\right)^p dx$ when $b^2-4\,a\,c\neq 0\,\wedge\,c\,d^2-b\,d\,e+a\,e^2\neq 0\,\wedge\,2\,c\,d-b\,e\neq 0\,\wedge\,p<-1$

1. $\int (d+e\,x)^m \left(a+b\,x+c\,x^2\right)^p dx$ when $b^2-4\,a\,c\neq 0\,\wedge\,c\,d^2-b\,d\,e+a\,e^2\neq 0\,\wedge\,2\,c\,d-b\,e\neq 0\,\wedge\,p<-1\,\wedge\,m>0$

1: $\int (d+e\,x)^m \left(a+b\,x+c\,x^2\right)^p dx$ when $b^2-4\,a\,c\neq 0\,\wedge\,c\,d^2-b\,d\,e+a\,e^2\neq 0\,\wedge\,2\,c\,d-b\,e\neq 0\,\wedge\,p<-1\,\wedge\,0<m<1$

Derivation: Quadratic recurrence 2a with $A=1$ and $B=0$

Derivation: Quadratic recurrence 2b with $A=d$, $B=e$ and $m=m-1$

Rule 1.2.1.2.12.1.1: If $b^2-4\,a\,c\neq 0\,\wedge\,c\,d^2-b\,d\,e+a\,e^2\neq 0\,\wedge\,2\,c\,d-b\,e\neq 0\,\wedge\,p<-1\,\wedge\,0<m<1$, then

$$\int (d+e\,x)^m \left(a+b\,x+c\,x^2\right)^p dx \longrightarrow$$

$$\frac{(d+e\,x)^m\,(b+2\,c\,x)\,\left(a+b\,x+c\,x^2\right)^{p+1}}{(p+1)\,\left(b^2-4\,a\,c\right)}-\frac{1}{(p+1)\,\left(b^2-4\,a\,c\right)}\int (d+e\,x)^{m-1}\,(b\,e\,m+2\,c\,d\,(2\,p+3)+2\,c\,e\,(m+2\,p+3)\,x)\,\left(a+b\,x+c\,x^2\right)^{p+1} dx$$

Program code:

```
Int[(d_.+e_.*x_)^m_*(a_.+b_.*x_+c_.*x_^2)^p_,x_Symbol] :=
  (d+e*x)^m*(b+2*c*x)*(a+b*x+c*x^2)^(p+1)/((p+1)*(b^2-4*a*c)) -
  1/((p+1)*(b^2-4*a*c))*Int[(d+e*x)^(m-1)*(b*e*m+2*c*d*(2*p+3)+2*c*e*(m+2*p+3)*x)*(a+b*x+c*x^2)^(p+1),x] /;
FreeQ[{a,b,c,d,e},x] && NeQ[b^2-4*a*c,0] && NeQ[c*d^2-b*d*e+a*e^2,0] && NeQ[2*c*d-b*e,0] &&
  LtQ[p,-1] && GtQ[m,0] && (LtQ[m,1] || ILtQ[m+2*p+3,0] && NeQ[m,2]) && IntQuadraticQ[a,b,c,d,e,m,p,x]
```

```
Int[(d_+e_.*x_)^m_*(a_+c_.*x_^2)^p_,x_Symbol] :=
  -x*(d+e*x)^m*(a+c*x^2)^(p+1)/(2*a*(p+1)) +
  1/(2*a*(p+1))*Int[(d+e*x)^(m-1)*(d*(2*p+3)+e*(m+2*p+3)*x)*(a+c*x^2)^(p+1),x] /;
FreeQ[{a,c,d,e},x] && NeQ[c*d^2+a*e^2,0] &&
  LtQ[p,-1] && GtQ[m,0] && (LtQ[m,1] || ILtQ[m+2*p+3,0] && NeQ[m,2]) && IntQuadraticQ[a,0,c,d,e,m,p,x]
```

2: $\int (d+e\,x)^m \left(a+b\,x+c\,x^2\right)^p\,dx$ when $b^2-4\,a\,c\neq 0\,\wedge\,c\,d^2-b\,d\,e+a\,e^2\neq 0\,\wedge\,2\,c\,d-b\,e\neq 0\,\wedge\,p<-1\,\wedge\,m>1$

Derivation: Quadratic recurrence 2a with $A=d$, $B=e$ and $m=m-1$

Rule 1.2.1.2.12.1.2: If $b^2-4\,a\,c\neq 0\,\wedge\,c\,d^2-b\,d\,e+a\,e^2\neq 0\,\wedge\,2\,c\,d-b\,e\neq 0\,\wedge\,p<-1\,\wedge\,m>1$, then

$$\int (d+e\,x)^m \left(a+b\,x+c\,x^2\right)^p\,dx\,\longrightarrow$$

$$\frac{(d+e\,x)^{m-1}\,(d\,b-2\,a\,e+(2\,c\,d-b\,e)\,x)\,\left(a+b\,x+c\,x^2\right)^{p+1}}{(p+1)\,\left(b^2-4\,a\,c\right)}+$$

$$\frac{1}{(p+1)\,\left(b^2-4\,a\,c\right)}\int (d+e\,x)^{m-2}\,\left(e\,(2\,a\,e\,(m-1)+b\,d\,(2\,p-m+4))-2\,c\,d^2\,(2\,p+3)+e\,(b\,e-2\,d\,c)\,(m+2\,p+2)\,x\right)\,\left(a+b\,x+c\,x^2\right)^{p+1}\,dx$$

Program code:

```
Int[(d_.+e_.*x_)^m_*(a_.+b_.*x_+c_.*x_^2)^p_,x_Symbol] :=
  (d+e*x)^(m-1)*(d*b-2*a*e+(2*c*d-b*e)*x)*(a+b*x+c*x^2)^(p+1)/((p+1)*(b^2-4*a*c)) +
  1/((p+1)*(b^2-4*a*c))*
    Int[(d+e*x)^(m-2)*
      Simp[e*(2*a*e*(m-1)+b*d*(2*p-m+4))-2*c*d^2*(2*p+3)+e*(b*e-2*d*c)*(m+2*p+2)*x,x]*
      (a+b*x+c*x^2)^(p+1),x] /;
FreeQ[{a,b,c,d,e},x] && NeQ[b^2-4*a*c,0] && NeQ[c*d^2-b*d*e+a*e^2,0] && NeQ[2*c*d-b*e,0] && LtQ[p,-1] && GtQ[m,1] && IntQuadraticQ[
```

```
Int[(d_+e_.*x_)^m_*(a_+c_.*x_^2)^p_,x_Symbol] :=
  (d+e*x)^(m-1)*(a*e-c*d*x)*(a+c*x^2)^(p+1)/(2*a*c*(p+1)) +
  1/((p+1)*(-2*a*c))*
    Int[(d+e*x)^(m-2)*Simp[a*e^2*(m-1)-c*d^2*(2*p+3)-d*c*e*(m+2*p+2)*x,x]*(a+c*x^2)^(p+1),x] /;
FreeQ[{a,c,d,e},x] && NeQ[c*d^2+a*e^2,0] && LtQ[p,-1] && GtQ[m,1] && IntQuadraticQ[a,0,c,d,e,m,p,x]
```

2: $\int (d+e\,x)^m \left(a+b\,x+c\,x^2\right)^p\,dx$ when $b^2-4\,a\,c\neq 0\,\wedge\,c\,d^2-b\,d\,e+a\,e^2\neq 0\,\wedge\,2\,c\,d-b\,e\neq 0\,\wedge\,p<-1$

Derivation: Quadratic recurrence 2b with $A=1$ and $B=0$

Rule 1.2.1.2.12.2: If $b^2-4\,a\,c\neq 0\,\wedge\,c\,d^2-b\,d\,e+a\,e^2\neq 0\,\wedge\,2\,c\,d-b\,e\neq 0\,\wedge\,p<-1$, then

$$\int (d+e\,x)^m \left(a+b\,x+c\,x^2\right)^p\,dx\,\longrightarrow$$

$$\frac{(d+e\,x)^{m+1}\,\left(b\,c\,d-b^2\,e+2\,a\,c\,e+c\,(2\,c\,d-b\,e)\,x\right)\,\left(a+b\,x+c\,x^2\right)^{p+1}}{(p+1)\,\left(b^2-4\,a\,c\right)\,\left(c\,d^2-b\,d\,e+a\,e^2\right)}+\frac{1}{(p+1)\,\left(b^2-4\,a\,c\right)\,\left(c\,d^2-b\,d\,e+a\,e^2\right)}\cdot$$

$$\int(d+e\,x)^m\,\left(b\,c\,d\,e\,(2\,p-m+2)+b^2\,e^2\,(m+p+2)-2\,c^2\,d^2\,(2\,p+3)-2\,a\,c\,e^2\,(m+2\,p+3)-c\,e\,(2\,c\,d-b\,e)\,(m+2\,p+4)\,x\right)\,\left(a+b\,x+c\,x^2\right)^{p+1}\,dx$$

Program code:

```
Int[(d_.+e_.*x_)^m_*(a_.+b_.*x_+c_.*x_^2)^p_,x_Symbol] :=
  (d+e*x)^(m+1)*(b*c*d-b^2*e+2*a*c*e+c*(2*c*d-b*e)*x)*(a+b*x+c*x^2)^(p+1)/((p+1)*(b^2-4*a*c)*(c*d^2-b*d*e+a*e^2)) +
  1/((p+1)*(b^2-4*a*c)*(c*d^2-b*d*e+a*e^2))*
    Int[(d+e*x)^m*
      Simp[b*c*d*e*(2*p-m+2)+b^2*e^2*(m+p+2)-2*c^2*d^2*(2*p+3)-2*a*c*e^2*(m+2*p+3)-c*e*(2*c*d-b*e)*(m+2*p+4)*x,x]*
      (a+b*x+c*x^2)^(p+1),x] /;
FreeQ[{a,b,c,d,e,m},x] && NeQ[b^2-4*a*c,0] && NeQ[c*d^2-b*d*e+a*e^2,0] && NeQ[2*c*d-b*e,0] && LtQ[p,-1] && IntQuadraticQ[a,b,c,d,e,
```

```
Int[(d_+e_.*x_)^m_*(a_+c_.*x_^2)^p_,x_Symbol] :=
  -(d+e*x)^(m+1)*(a*e+c*d*x)*(a+c*x^2)^(p+1)/(2*a*(p+1)*(c*d^2+a*e^2)) +
  1/(2*a*(p+1)*(c*d^2+a*e^2))*
    Int[(d+e*x)^m*Simp[c*d^2*(2*p+3)+a*e^2*(m+2*p+3)+c*e*d*(m+2*p+4)*x,x]*(a+c*x^2)^(p+1),x] /;
FreeQ[{a,c,d,e,m},x] && NeQ[c*d^2+a*e^2,0] && LtQ[p,-1] && IntQuadraticQ[a,0,c,d,e,m,p,x]
```

13: $\int (d+e\,x)^m \left(a+b\,x+c\,x^2\right)^p dx$ when $b^2-4\,a\,c\neq 0\,\land\,c\,d^2-b\,d\,e+a\,e^2\neq 0\,\land\,2\,c\,d-b\,e\neq 0\,\land\,m>1\,\land\,m+2\,p+1\neq 0$

Reference: G&R 2.160.3, G&R 2.174.1, CRC 119

Derivation: Quadratic recurrence 3a with $A=d$, $B=e$ and $m=m-1$

Note: G&R 2.174.1 is a special case of G&R 2.160.3.

Rule 1.2.1.2.13: If $b^2-4\,a\,c\neq 0\,\land\,c\,d^2-b\,d\,e+a\,e^2\neq 0\,\land\,2\,c\,d-b\,e\neq 0\,\land\,m>1\,\land\,m+2\,p+1\neq 0$, then

$$\int (d+e\,x)^m \left(a+b\,x+c\,x^2\right)^p dx\,\longrightarrow$$

$$\frac{e\,(d+e\,x)^{m-1}\left(a+b\,x+c\,x^2\right)^{p+1}}{c\,(m+2\,p+1)}+\frac{1}{c\,(m+2\,p+1)}\int (d+e\,x)^{m-2}\left(c\,d^2\,(m+2\,p+1)-e\,(a\,e\,(m-1)+b\,d\,(p+1))+e\,(2\,c\,d-b\,e)\,(m+p)\,x\right)\left(a+b\,x+c\,x^2\right)^p dx$$

Program code:

```
Int[(d_.+e_.*x_)^m_*(a_.+b_.*x_+c_.*x_^2)^p_,x_Symbol] :=
  e*(d+e*x)^(m-1)*(a+b*x+c*x^2)^(p+1)/(c*(m+2*p+1)) +
  1/(c*(m+2*p+1))*
    Int[(d+e*x)^(m-2)*
      Simp[c*d^2*(m+2*p+1)-e*(a*e*(m-1)+b*d*(p+1))+e*(2*c*d-b*e)*(m+p)*x,x]*
      (a+b*x+c*x^2)^p,x] /;
FreeQ[{a,b,c,d,e,m,p},x] && NeQ[b^2-4*a*c,0] && NeQ[c*d^2-b*d*e+a*e^2,0] && NeQ[2*c*d-b*e,0] &&
  If[RationalQ[m], GtQ[m,1], SumSimplerQ[m,-2]] && NeQ[m+2*p+1,0] && IntQuadraticQ[a,b,c,d,e,m,p,x]
```

```
Int[(d_+e_.*x_)^m_*(a_+c_.*x_^2)^p_,x_Symbol] :=
  e*(d+e*x)^(m-1)*(a+c*x^2)^(p+1)/(c*(m+2*p+1)) +
  1/(c*(m+2*p+1))*
    Int[(d+e*x)^(m-2)*Simp[c*d^2*(m+2*p+1)-a*e^2*(m-1)+2*c*d*e*(m+p)*x,x]*(a+c*x^2)^p,x] /;
FreeQ[{a,c,d,e,m,p},x] && NeQ[c*d^2+a*e^2,0] &&
  If[RationalQ[m], GtQ[m,1], SumSimplerQ[m,-2]] && NeQ[m+2*p+1,0] && IntQuadraticQ[a,0,c,d,e,m,p,x]
```

14: $\int (d+e\,x)^m \left(a+b\,x+c\,x^2\right)^p dx$ when $b^2-4\,a\,c\neq 0\,\land\,c\,d^2-b\,d\,e+a\,e^2\neq 0\,\land\,2\,c\,d-b\,e\neq 0\,\land\,m<-1$

Reference: G&R 2.176, CRC 123

Derivation: Quadratic recurrence 3b with $A=1$ and $B=0$

Rule 1.2.1.2.14: If $b^2-4\,a\,c\neq 0\,\land\,c\,d^2-b\,d\,e+a\,e^2\neq 0\,\land\,2\,c\,d-b\,e\neq 0\,\land\,m<-1$, then

$$\int (d+e\,x)^m\,\left(a+b\,x+c\,x^2\right)^p\,dx\;\longrightarrow$$

$$\frac{e\,(d+e\,x)^{m+1}\,\left(a+b\,x+c\,x^2\right)^{p+1}}{(m+1)\,\left(c\,d^2-b\,d\,e+a\,e^2\right)}+$$

$$\frac{1}{(m+1)\,\left(c\,d^2-b\,d\,e+a\,e^2\right)}\int (d+e\,x)^{m+1}\,(c\,d\,(m+1)-b\,e\,(m+p+2)-c\,e\,(m+2\,p+3)\,x)\,\left(a+b\,x+c\,x^2\right)^p\,dx$$

– **Program code:**

```
Int[(d_.+e_.*x_)^m_*(a_.+b_.*x_+c_.*x_^2)^p_,x_Symbol] :=
  e*(d+e*x)^(m+1)*(a+b*x+c*x^2)^(p+1)/((m+1)*(c*d^2-b*d*e+a*e^2)) +
  1/((m+1)*(c*d^2-b*d*e+a*e^2))*
    Int[(d+e*x)^(m+1)*Simp[c*d*(m+1)-b*e*(m+p+2)-c*e*(m+2*p+3)*x,x]*(a+b*x+c*x^2)^p,x] /;
FreeQ[{a,b,c,d,e,m,p},x] && NeQ[b^2-4*a*c,0] && NeQ[c*d^2-b*d*e+a*e^2,0] && NeQ[2*c*d-b*e,0] && NeQ[m,-1] &&
  (LtQ[m,-1] && IntQuadraticQ[a,b,c,d,e,m,p,x] || SumSimplerQ[m,1] && IntegerQ[p] || ILtQ[Simplify[m+2*p+3],0])
```

```
Int[(d_+e_.*x_)^m_*(a_+c_.*x_^2)^p_,x_Symbol] :=
  e*(d+e*x)^(m+1)*(a+c*x^2)^(p+1)/((m+1)*(c*d^2+a*e^2)) +
  c/((m+1)*(c*d^2+a*e^2))*
    Int[(d+e*x)^(m+1)*Simp[d*(m+1)-e*(m+2*p+3)*x,x]*(a+c*x^2)^p,x] /;
FreeQ[{a,c,d,e,m,p},x] && NeQ[c*d^2+a*e^2,0] && NeQ[m,-1] &&
  (LtQ[m,-1] && IntQuadraticQ[a,0,c,d,e,m,p,x] || SumSimplerQ[m,1] && IntegerQ[p] || ILtQ[Simplify[m+2*p+3],0])
```

15. $\int \frac{\left(a+b x+c x^2\right)^p}{d+e x} \, dx$ when $b^2-4 a c \neq 0 \wedge c d^2-b d e+a e^2 \neq 0 \wedge 2 c d-b e \neq 0 \wedge 4 p \in \mathbb{Z}$

1. $\int \frac{\left(a+c x^2\right)^p}{d+e x} \, dx$ when $c d^2+a e^2 \neq 0 \wedge 4 p \in \mathbb{Z}$

1: $\int \frac{1}{(d+e x)\left(a+c x^2\right)^{1/4}} \, dx$ when $c d^2+a e^2 \neq 0$

Derivation: Algebraic expansion

Basis: $\frac{1}{d+e x} = \frac{d}{d^2-e^2 x^2} - \frac{e x}{d^2-e^2 x^2}$

Rule 1.2.1.2.15.1.1: If $c d^2+a e^2 \neq 0$, then

$$\int \frac{1}{(d+e x)\left(a+c x^2\right)^{1/4}} \, dx \longrightarrow d \int \frac{1}{\left(d^2-e^2 x^2\right)\left(a+c x^2\right)^{1/4}} \, dx - e \int \frac{x}{\left(d^2-e^2 x^2\right)\left(a+c x^2\right)^{1/4}} \, dx$$

Program code:

```
Int[1/((d_+e_.*x_)*(a_+c_.*x_^2)^(1/4)),x_Symbol] :=
  d*Int[1/((d^2-e^2*x^2)*(a+c*x^2)^(1/4)),x] - e*Int[x/((d^2-e^2*x^2)*(a+c*x^2)^(1/4)),x] /;
FreeQ[{a,c,d,e},x] && NeQ[c*d^2+a*e^2,0]
```

2: $\int \frac{1}{(d+e x)\left(a+c x^2\right)^{3/4}} \, dx$ when $c d^2+a e^2 \neq 0$

Derivation: Algebraic expansion

Basis: $\frac{1}{d+e x} = \frac{d}{d^2-e^2 x^2} - \frac{e x}{d^2-e^2 x^2}$

Rule 1.2.1.2.15.1.2: If $c d^2+a e^2 \neq 0$, then

$$\int \frac{1}{(d+e x)\left(a+c x^2\right)^{3/4}} \, dx \longrightarrow d \int \frac{1}{\left(d^2-e^2 x^2\right)\left(a+c x^2\right)^{3/4}} \, dx - e \int \frac{x}{\left(d^2-e^2 x^2\right)\left(a+c x^2\right)^{3/4}} \, dx$$

Program code:

```
Int[1/((d_+e_.*x_)*(a_+c_.*x_^2)^(3/4)),x_Symbol] :=
  d*Int[1/((d^2-e^2*x^2)*(a+c*x^2)^(3/4)),x] - e*Int[x/((d^2-e^2*x^2)*(a+c*x^2)^(3/4)),x] /;
FreeQ[{a,c,d,e},x] && NeQ[c*d^2+a*e^2,0]
```

2. $\int \frac{(a+b x+c x^2)^p}{d+e x} \, dx$ when $b^2-4 a c \neq 0 \land c d^2-b d e+a e^2 \neq 0 \land 2 c d-b e \neq 0 \land 4 p \in \mathbb{Z}$

1: $\int \frac{(a+b x+c x^2)^p}{d+e x} \, dx$ when $4 a-\frac{b^2}{c}>0 \bigwedge 4 p \in \mathbb{Z}$

Derivation: Integration by substitution

- Basis: If $4 a-\frac{b^2}{c}>0$, then $(a+b x+c x^2)^p F[x] = \frac{1}{2 c \left(-\frac{4 c}{b^2-4 a c}\right)^p} \text{Subst}\left[\left(1-\frac{x^2}{b^2-4 a c}\right)^p F\left[-\frac{b}{2 c}+\frac{x}{2 c}\right], x, b+2 c x\right] \partial_x (b+2 c x)$

- Rule 1.2.1.2.15.2.1: If $4 a-\frac{b^2}{c}>0 \bigwedge 4 p \in \mathbb{Z}$, then

$$\int \frac{(a+b x+c x^2)^p}{d+e x} \, dx \longrightarrow \frac{1}{\left(-\frac{4 c}{b^2-4 a c}\right)^p} \text{Subst}\left[\int \frac{\left(1-\frac{x^2}{b^2-4 a c}\right)^p}{2 c d-b e+e x} \, dx, x, b+2 c x\right]$$

Program code:

```
Int[(a_.+b_.*x_+c_.*x_^2)^p_/(d_.+e_.*x_),x_Symbol] :=
  1/(-4*c/(b^2-4*a*c))^p*Subst[Int[Simp[1-x^2/(b^2-4*a*c),x]^p/Simp[2*c*d-b*e+e*x,x],x],x,b+2*c*x] /;
FreeQ[{a,b,c,d,e,p},x] && GtQ[4*a-b^2/c,0] && IntegerQ[4*p]
```

2: $\int \frac{\left(a+b\,x+c\,x^2\right)^p}{d+e\,x}\,dx$ when $4\,a-\frac{b^2}{c}\not>0\,\bigwedge\,4\,p\in\mathbb{Z}$

Derivation: Piecewise constant extraction

- **Basis:** $\partial_x \frac{\left(a+b\,x+c\,x^2\right)^p}{\left(-\frac{c\,\left(a+b\,x+c\,x^2\right)}{b^2-4\,a\,c}\right)^p}==0$
- **Rule 1.2.1.2.15.2.2: If** $4\,a-\frac{b^2}{c}\not>0\,\bigwedge\,4\,p\in\mathbb{Z}$**, then**

$$\int \frac{\left(a+b\,x+c\,x^2\right)^p}{d+e\,x}\,dx \longrightarrow \frac{\left(a+b\,x+c\,x^2\right)^p}{\left(-\frac{c\,\left(a+b\,x+c\,x^2\right)}{b^2-4\,a\,c}\right)^p}\int \frac{\left(-\frac{a\,c}{b^2-4\,a\,c}-\frac{b\,c\,x}{b^2-4\,a\,c}-\frac{c^2\,x^2}{b^2-4\,a\,c}\right)^p}{d+e\,x}\,dx$$

Program code:

```
Int[(a_.+b_.*x_+c_.*x_^2)^p_/(d_.+e_.*x_),x_Symbol] :=
  (a+b*x+c*x^2)^p/(-c*(a+b*x+c*x^2)/(b^2-4*a*c))^p*
    Int[(-a*c/(b^2-4*a*c)-b*c*x/(b^2-4*a*c)-c^2*x^2/(b^2-4*a*c))^p/(d+e*x),x] /;
FreeQ[{a,b,c,d,e,p},x] && Not[GtQ[4*a-b^2/c,0]] && IntegerQ[4*p]
```

16. $\int \frac{1}{(d+e\,x)\,\left(a+b\,x+c\,x^2\right)^{1/3}}\,dx$ when $b^2-4\,a\,c\neq 0\,\wedge\,c\,d^2-b\,d\,e+a\,e^2\neq 0\,\wedge\,2\,c\,d-b\,e\neq 0$

1. $\int \frac{1}{(d+e\,x)\,\left(a+b\,x+c\,x^2\right)^{1/3}}\,dx$ when $2\,c\,d-b\,e\neq 0\,\wedge\,c^2\,d^2-b\,c\,d\,e+b^2\,e^2-3\,a\,c\,e^2=0$

1: $\int \frac{1}{(d+e\,x)\,\left(a+b\,x+c\,x^2\right)^{1/3}}\,dx$ when $2\,c\,d-b\,e\neq 0\,\wedge\,c^2\,d^2-b\,c\,d\,e+b^2\,e^2-3\,a\,c\,e^2=0\,\wedge\,c\,e^2\,(2\,c\,d-b\,e)>0$

- Derived from formula for this class of Goursat pseudo-elliptic integrands contributed by Martin Welz on 19 September 2016

- Rule 1.2.1.2.16.1.1: If $2\,c\,d-b\,e\neq 0\,\wedge\,c^2\,d^2-b\,c\,d\,e+b^2\,e^2-3\,a\,c\,e^2=0\,\wedge\,c\,e^2\,(2\,c\,d-b\,e)>0$, let $q\to\left(3\,c\,e^2\,(2\,c\,d-b\,e)\right)^{1/3}$, then

$$\int \frac{1}{(d+e\,x)\,\left(a+b\,x+c\,x^2\right)^{1/3}}\,dx\longrightarrow$$

$$-\frac{\sqrt{3}\,c\,e\,\text{ArcTan}\left[\frac{1}{\sqrt{3}}+\frac{2\,(c\,d-b\,e-c\,e\,x)}{\sqrt{3}\,q\,\left(a+b\,x+c\,x^2\right)^{1/3}}\right]}{q^2}-\frac{3\,c\,e\,\text{Log}[d+e\,x]}{2\,q^2}+\frac{3\,c\,e\,\text{Log}\left[c\,d-b\,e-c\,e\,x-q\,\left(a+b\,x+c\,x^2\right)^{1/3}\right]}{2\,q^2}$$

- Rule 1.2.1.2.16.1.1: If $c\,d^2-3\,a\,e^2=0$, let $q\to\left(\frac{6\,c^2\,e^2}{d^2}\right)^{1/3}$, then

$$\int \frac{1}{(d+e\,x)\,\left(a+c\,x^2\right)^{1/3}}\,dx\longrightarrow$$

$$-\frac{\sqrt{3}\,c\,e\,\text{ArcTan}\left[\frac{1}{\sqrt{3}}+\frac{2\,c\,(d-e\,x)}{\sqrt{3}\,d\,q\,\left(a+c\,x^2\right)^{1/3}}\right]}{d^2\,q^2}-\frac{3\,c\,e\,\text{Log}[d+e\,x]}{2\,d^2\,q^2}+\frac{3\,c\,e\,\text{Log}\left[c\,d-c\,e\,x-d\,q\,\left(a+c\,x^2\right)^{1/3}\right]}{2\,d^2\,q^2}$$

Program code:

```
Int[1/((d_.+e_.*x_)*(a_+b_.*x_+c_.*x_^2)^(1/3)),x_Symbol] :=
  With[{q=Rt[3*c*e^2*(2*c*d-b*e),3]},
  -Sqrt[3]*c*e*ArcTan[1/Sqrt[3]+2*(c*d-b*e-c*e*x)/(Sqrt[3]*q*(a+b*x+c*x^2)^(1/3))]/q^2 -
  3*c*e*Log[d+e*x]/(2*q^2) +
   3*c*e*Log[c*d-b*e-c*e*x-q*(a+b*x+c*x^2)^(1/3)]/(2*q^2)] /;
FreeQ[{a,b,c,d,e},x] && NeQ[2*c*d-b*e,0] && EqQ[c^2*d^2-b*c*d*e+b^2*e^2-3*a*c*e^2,0] && PosQ[c*e^2*(2*c*d-b*e)]
```

```
Int[1/((d_+e_.*x_)*(a_+c_.*x_^2)^(1/3)),x_Symbol] :=
  With[{q=Rt[6*c^2*e^2/d^2,3]},
  -Sqrt[3]*c*e*ArcTan[1/Sqrt[3]+2*c*(d-e*x)/(Sqrt[3]*d*q*(a+c*x^2)^(1/3))]/(d^2*q^2) -
  3*c*e*Log[d+e*x]/(2*d^2*q^2) +
   3*c*e*Log[c*d-c*e*x-d*q*(a+c*x^2)^(1/3)]/(2*d^2*q^2)] /;
FreeQ[{a,c,d,e},x] && EqQ[c*d^2-3*a*e^2,0]
```

2: $\int \frac{1}{(d+e\,x)\,\left(a+b\,x+c\,x^2\right)^{1/3}}\,dx$ when $2\,c\,d-b\,e\neq 0\,\wedge\,c^2\,d^2-b\,c\,d\,e+b^2\,e^2-3\,a\,c\,e^2==0\,\wedge\,c\,e^2\,(2\,c\,d-b\,e)\not>0$

Derived from formula for this class of Goursat pseudo-elliptic integrands contributed by Martin Welz on 19 September 2016

Rule 1.2.1.2.16.1.2: If $2\,c\,d-b\,e\neq 0\,\wedge\,c^2\,d^2-b\,c\,d\,e+b^2\,e^2-3\,a\,c\,e^2==0\,\wedge\,c\,e^2\,(2\,c\,d-b\,e)\not>0$**, let** $q\to\left(-3\,c\,e^2\,(2\,c\,d-b\,e)\right)^{1/3}$**, then**

$$\int \frac{1}{(d+e\,x)\,\left(a+b\,x+c\,x^2\right)^{1/3}}\,dx \longrightarrow$$

$$-\frac{\sqrt{3}\,c\,e\,\text{ArcTan}\left[\frac{1}{\sqrt{3}}-\frac{2\,(c\,d-b\,e-c\,e\,x)}{\sqrt{3}\,q\,\left(a+b\,x+c\,x^2\right)^{1/3}}\right]}{q^2}-\frac{3\,c\,e\,\text{Log}[d+e\,x]}{2\,q^2}+\frac{3\,c\,e\,\text{Log}\left[c\,d-b\,e-c\,e\,x+q\,\left(a+b\,x+c\,x^2\right)^{1/3}\right]}{2\,q^2}$$

Program code:

```
Int[1/((d_.+e_.*x_)*(a_+b_.*x_+c_.*x_^2)^(1/3)),x_Symbol] :=
  With[{q=Rt[-3*c*e^2*(2*c*d-b*e),3]},
  -Sqrt[3]*c*e*ArcTan[1/Sqrt[3]-2*(c*d-b*e-c*e*x)/(Sqrt[3]*q*(a+b*x+c*x^2)^(1/3))]/q^2 -
  3*c*e*Log[d+e*x]/(2*q^2) +
   3*c*e*Log[c*d-b*e-c*e*x+q*(a+b*x+c*x^2)^(1/3)]/(2*q^2)] /;
FreeQ[{a,b,c,d,e},x] && NeQ[2*c*d-b*e,0] && EqQ[c^2*d^2-b*c*d*e+b^2*e^2-3*a*c*e^2,0] && NegQ[c*e^2*(2*c*d-b*e)]
```

```
(* Int[1/((d_+e_.*x_)*(a_+c_.*x_^2)^(1/3)),x_Symbol] :=
  With[{q=Rt[-6*c^2*d*e^2,3]},
  -Sqrt[3]*c*e*ArcTan[1/Sqrt[3]-2*(c*d-c*e*x)/(Sqrt[3]*q*(a+c*x^2)^(1/3))]/q^2 -
  3*c*e*Log[d+e*x]/(2*q^2) +
   3*c*e*Log[c*d-c*e*x+q*(a+c*x^2)^(1/3)]/(2*q^2)] /;
FreeQ[{a,c,d,e},x] && EqQ[c*d^2-3*a*e^2,0] && NegQ[c^2*d*e^2] *)
```

2. $\int \frac{1}{(d+e\,x)\,\left(a+b\,x+c\,x^2\right)^{1/3}}\,dx$ when $b^2-4\,a\,c\neq 0\,\wedge\,c^2\,d^2-b\,c\,d\,e-2\,b^2\,e^2+9\,a\,c\,e^2==0$

1. $\int \frac{1}{(d+e\,x)\,\left(a+c\,x^2\right)^{1/3}}\,dx$ when $c\,d^2+9\,a\,e^2==0$

1: $\int \frac{1}{(d+e\,x)\,\left(a+c\,x^2\right)^{1/3}}\,dx$ when $c\,d^2+9\,a\,e^2==0\,\wedge\,a>0$

Derivation: Algebraic expansion

- **Basis: If** $c\,d^2+9\,a\,e^2==0\,\wedge\,a>0$**, then** $\left(a+c\,x^2\right)^{1/3}==a^{1/3}\left(1-\frac{3\,e\,x}{d}\right)^{1/3}\left(1+\frac{3\,e\,x}{d}\right)^{1/3}$

Rule 1.2.1.2.16.2.1.1: If $c\,d^2+9\,a\,e^2==0\,\wedge\,a>0$**, then**

$$\int \frac{1}{(d+e\,x)\,\left(a+c\,x^2\right)^{1/3}}\,dx\,\longrightarrow\,a^{1/3}\int \frac{1}{(d+e\,x)\,\left(1-\frac{3\,e\,x}{d}\right)^{1/3}\left(1+\frac{3\,e\,x}{d}\right)^{1/3}}\,dx$$

Program code:

```
Int[1/((d_+e_.*x_)*(a_+c_.*x_^2)^(1/3)),x_Symbol] :=
  a^(1/3)*Int[1/((d+e*x)*(1-3*e*x/d)^(1/3)*(1+3*e*x/d)^(1/3)),x] /;
FreeQ[{a,c,d,e},x] && EqQ[c*d^2+9*a*e^2,0] && GtQ[a,0]
```

2: $\int \frac{1}{(d+e\,x)\,\left(a+c\,x^2\right)^{1/3}}\,dx$ when $c\,d^2+9\,a\,e^2==0\,\wedge\,a\not>0$

Derivation: Piecewise constant extraction

- **Basis:** $\partial_x \frac{\left(1+\frac{c\,x^2}{a}\right)^{1/3}}{\left(a+c\,x^2\right)^{1/3}}==0$

Rule 1.2.1.2.16.2.1.2: If $c\,d^2+9\,a\,e^2==0\,\wedge\,a\not>0$**, then**

$$\int \frac{1}{(d+e\,x)\,\left(a+c\,x^2\right)^{1/3}}\,dx\,\longrightarrow\,\frac{\left(1+\frac{c\,x^2}{a}\right)^{1/3}}{\left(a+c\,x^2\right)^{1/3}}\int \frac{1}{(d+e\,x)\,\left(1+\frac{c\,x^2}{a}\right)^{1/3}}\,dx$$

Program code:

```
Int[1/((d_+e_.*x_)*(a_+c_.*x_^2)^(1/3)),x_Symbol] :=
  (1+c*x^2/a)^(1/3)/(a+c*x^2)^(1/3)*Int[1/((d+e*x)*(1+c*x^2/a)^(1/3)),x] /;
FreeQ[{a,c,d,e},x] && EqQ[c*d^2+9*a*e^2,0] && Not[GtQ[a,0]]
```

2: $\int \frac{1}{(d+e\,x)\,\left(a+b\,x+c\,x^2\right)^{1/3}}\,dx$ when $b^2-4\,a\,c\neq 0\,\wedge\,c^2\,d^2-b\,c\,d\,e-2\,b^2\,e^2+9\,a\,c\,e^2==0$

Derivation: Piecewise constant extraction

- Basis: Let $q\to\sqrt{b^2-4\,a\,c}$, then $\partial_x\frac{(b+q+2\,c\,x)^{1/3}\,(b-q+2\,c\,x)^{1/3}}{(a+b\,x+c\,x^2)^{1/3}}==0$

- Rule 1.2.1.2.16.2.2: If $b^2-4\,a\,c\neq 0\,\wedge\,c^2\,d^2-b\,c\,d\,e-2\,b^2\,e^2+9\,a\,c\,e^2==0$, let $q\to\sqrt{b^2-4\,a\,c}$, then

$$\int \frac{1}{(d+e\,x)\,\left(a+b\,x+c\,x^2\right)^{1/3}}\,dx\;\longrightarrow\;\frac{(b+q+2\,c\,x)^{1/3}\,(b-q+2\,c\,x)^{1/3}}{\left(a+b\,x+c\,x^2\right)^{1/3}}\int \frac{1}{(d+e\,x)\,(b+q+2\,c\,x)^{1/3}\,(b-q+2\,c\,x)^{1/3}}\,dx$$

Program code:

```
Int[1/((d_.+e_.*x_)*(a_+b_.*x_+c_.*x_^2)^(1/3)),x_Symbol] :=
  With[{q=Rt[b^2-4*a*c,2]},
  (b+q+2*c*x)^(1/3)*(b-q+2*c*x)^(1/3)/(a+b*x+c*x^2)^(1/3)*Int[1/((d+e*x)*(b+q+2*c*x)^(1/3)*(b-q+2*c*x)^(1/3)),x]] /;
FreeQ[{a,b,c,d,e},x] && NeQ[b^2-4*a*c,0] && EqQ[c^2*d^2-b*c*d*e-2*b^2*e^2+9*a*c*e^2,0]
```

17: $\int (d+e\,x)^m\,\left(a+c\,x^2\right)^p\,dx$ when $c\,d^2+a\,e^2\neq 0\,\wedge\,p\notin\mathbb{Z}\,\wedge\,a>0\,\wedge\,c<0$

Derivation: Algebraic expansion

- Basis: If $a>0$, then $\left(a+c\,x^2\right)^p==\left(\sqrt{a}+\sqrt{-c}\,x\right)^p\,\left(\sqrt{a}-\sqrt{-c}\,x\right)^p$

- Rule 1.2.1.2.17: If $c\,d^2+a\,e^2\neq 0\,\wedge\,p\notin\mathbb{Z}\,\wedge\,a>0\,\wedge\,c<0$, then

$$\int (d+e\,x)^m\,\left(a+c\,x^2\right)^p\,dx\;\longrightarrow\;\int (d+e\,x)^m\,\left(\sqrt{a}+\sqrt{-c}\,x\right)^p\,\left(\sqrt{a}-\sqrt{-c}\,x\right)^p\,dx$$

Program code:

```
Int[(d_+e_.*x_)^m_*(a_+c_.*x_^2)^p_,x_Symbol] :=
  Int[(d+e*x)^m*(Rt[a,2]+Rt[-c,2]*x)^p*(Rt[a,2]-Rt[-c,2]*x)^p,x] /;
FreeQ[{a,c,d,e,m,p},x] && NeQ[c*d^2+a*e^2,0] && Not[IntegerQ[p]] && GtQ[a,0] && LtQ[c,0]
```

19. $\int (d+e\,x)^m \left(a+b\,x+c\,x^2\right)^p dx$ when $b^2-4\,a\,c\neq 0\,\wedge\,c\,d^2-b\,d\,e+a\,e^2\neq 0\,\wedge\,2\,c\,d-b\,e\neq 0\,\wedge\,p\notin\mathbb{Z}$

1. $\int (d+e\,x)^m \left(a+b\,x+c\,x^2\right)^p dx$ when $b^2-4\,a\,c\neq 0\,\wedge\,c\,d^2-b\,d\,e+a\,e^2\neq 0\,\wedge\,2\,c\,d-b\,e\neq 0\,\wedge\,p\notin\mathbb{Z}\,\wedge\,m\in\mathbb{Z}^-$

1: $\int (d+e\,x)^m \left(a+c\,x^2\right)^p dx$ when $c\,d^2+a\,e^2\neq 0\,\wedge\,p\notin\mathbb{Z}\,\wedge\,m\in\mathbb{Z}^-$

Derivation: Algebraic expansion

Basis: If $m\in\mathbb{Z}$, **then** $(d+e\,x)^m == \left(\frac{d}{d^2-e^2\,x^2}-\frac{e\,x}{d^2-e^2\,x^2}\right)^{-m}$

Note: Resulting integrands are of the form $x^m\left(a+b\,x^2\right)^p\left(c+d\,x^2\right)^q$ **which are integrable in terms of the Appell hypergeometric function .**

Rule 1.2.1.2.18: If $c\,d^2+a\,e^2\neq 0\,\wedge\,p\notin\mathbb{Z}\,\wedge\,m\in\mathbb{Z}^-$**, then**

$$\int (d+e\,x)^m \left(a+c\,x^2\right)^p dx \longrightarrow \int \left(a+c\,x^2\right)^p \text{ExpandIntegrand}\left[\left(\frac{d}{d^2-e^2\,x^2}-\frac{e\,x}{d^2-e^2\,x^2}\right)^{-m},\,x\right] dx$$

Program code:

```
Int[(d_+e_.*x_)^m_*(a_+c_.*x_^2)^p_,x_Symbol] :=
  Int[ExpandIntegrand[(a+c*x^2)^p,(d/(d^2-e^2*x^2)-e*x/(d^2-e^2*x^2))^(-m),x],x] /;
FreeQ[{a,c,d,e,p},x] && NeQ[c*d^2+a*e^2,0] && Not[IntegerQ[p]] && ILtQ[m,0]
```

2: $\int (d+e\,x)^m \left(a+b\,x+c\,x^2\right)^p dx$ when $b^2-4\,a\,c\neq 0\,\wedge\,c\,d^2-b\,d\,e+a\,e^2\neq 0\,\wedge\,2\,c\,d-b\,e\neq 0\,\wedge\,p\notin\mathbb{Z}\,\wedge\,m\in\mathbb{Z}^-$

Derivation: Piecewise constant extraction and integration by substitution

- **Basis: Let** $q\to\sqrt{b^2-4\,a\,c}$ **, then** $\partial_x \frac{\left(\frac{1}{d+e\,x}\right)^{2\,p}\left(a+b\,x+c\,x^2\right)^p}{\left(\frac{e\,(b-q+2\,c\,x)}{c\,(d+e\,x)}\right)^p\left(\frac{e\,(b+q+2\,c\,x)}{c\,(d+e\,x)}\right)^p} == 0$
- **Basis:** $F[x] == -\frac{1}{e}\,\text{Subst}\left[\frac{F\left[\frac{1-d\,x}{e\,x}\right]}{x^2},\,x,\,\frac{1}{d+e\,x}\right]\,\partial_x\frac{1}{d+e\,x}$
- **Rule 1.2.1.2.19.1: If** $b^2-4\,a\,c\neq 0\,\wedge\,c\,d^2-b\,d\,e+a\,e^2\neq 0\,\wedge\,2\,c\,d-b\,e\neq 0\,\wedge\,p\notin\mathbb{Z}\,\wedge\,m\in\mathbb{Z}^-$**, let** $q\to\sqrt{b^2-4\,a\,c}$ **, then**

$$\int (d+e\,x)^m \left(a+b\,x+c\,x^2\right)^p dx \longrightarrow$$

$$\frac{\left(\frac{1}{d+e\,x}\right)^{2\,p}\left(a+b\,x+c\,x^2\right)^p}{\left(\frac{e\,(b-q+2\,c\,x)}{c\,(d+e\,x)}\right)^p\left(\frac{e\,(b+q+2\,c\,x)}{c\,(d+e\,x)}\right)^p}\int\frac{\left(\frac{e\,(b-q+2\,c\,x)}{c\,(d+e\,x)}\right)^p\left(\frac{e\,(b+q+2\,c\,x)}{c\,(d+e\,x)}\right)^p}{\left(\frac{1}{d+e\,x}\right)^{m+2\,p}}\,dx \longrightarrow$$

$$-\frac{\left(\frac{1}{d+e\,x}\right)^{2\,p}\left(a+b\,x+c\,x^2\right)^p}{e\left(\frac{e\,(b-q+2\,c\,x)}{2\,c\,(d+e\,x)}\right)^p\left(\frac{e\,(b+q+2\,c\,x)}{2\,c\,(d+e\,x)}\right)^p}\,\text{Subst}\left[\int x^{-m-2\,(p+1)}\left(1-\left(d-\frac{e\,(b-q)}{2\,c}\right)x\right)^p\left(1-\left(d-\frac{e\,(b+q)}{2\,c}\right)x\right)^p dx,\ x,\ \frac{1}{d+e\,x}\right]$$

Program code:

```
Int[(d_.+e_.*x_)^m_*(a_.+b_.*x_+c_.*x_^2)^p_,x_Symbol] :=
  With[{q=Rt[b^2-4*a*c,2]},
  -(1/(d+e*x))^(2*p)*(a+b*x+c*x^2)^p/(e*(e*(b-q+2*c*x)/(2*c*(d+e*x)))^p*(e*(b+q+2*c*x)/(2*c*(d+e*x)))^p)*
    Subst[Int[x^(-m-2*(p+1))*Simp[1-(d-e*(b-q)/(2*c))*x,x]^p*Simp[1-(d-e*(b+q)/(2*c))*x,x]^p,x],x,1/(d+e*x)]] /;
FreeQ[{a,b,c,d,e,p},x] && NeQ[b^2-4*a*c,0] && NeQ[c*d^2-b*d*e+a*e^2,0] && NeQ[2*c*d-b*e,0] && Not[IntegerQ[p]] && ILtQ[m,0]
```

2: $\int (d+e\,x)^m \left(a+b\,x+c\,x^2\right)^p dx$ when $b^2-4\,a\,c\neq 0\,\wedge\,c\,d^2-b\,d\,e+a\,e^2\neq 0\,\wedge\,2\,c\,d-b\,e\neq 0\,\wedge\,p\notin\mathbb{Z}$

Derivation: Piecewise constant extraction and integration by substitution

- Basis: Let $q\to\sqrt{b^2-4\,a\,c}$, then $\partial_x \frac{(a+b\,x+c\,x^2)^p}{\left(1-\frac{d+e\,x}{d-\frac{e\,(b-q)}{2\,c}}\right)^p\left(1-\frac{d+e\,x}{d-\frac{e\,(b+q)}{2\,c}}\right)^p}==0$
- Note: If $c\,d^2-b\,d\,e+a\,e^2\neq 0$ and $q==\sqrt{b^2-4\,a\,c}$, then $d-\frac{e\,(b-q)}{2\,c}\neq 0$ and $d-\frac{e\,(b+q)}{2\,c}\neq 0$.
- Rule 1.2.1.2.19.2: If $b^2-4\,a\,c\neq 0\,\wedge\,c\,d^2-b\,d\,e+a\,e^2\neq 0\,\wedge\,2\,c\,d-b\,e\neq 0\,\wedge\,p\notin\mathbb{Z}$, let $q\to\sqrt{b^2-4\,a\,c}$, then

$$\int (d+e\,x)^m \left(a+b\,x+c\,x^2\right)^p dx \longrightarrow$$

$$\frac{\left(a+b\,x+c\,x^2\right)^p}{\left(1-\frac{d+e\,x}{d-\frac{e\,(b-q)}{2\,c}}\right)^p\left(1-\frac{d+e\,x}{d-\frac{e\,(b+q)}{2\,c}}\right)^p}\int (d+e\,x)^m\left(1-\frac{d+e\,x}{d-\frac{e\,(b-q)}{2\,c}}\right)^p\left(1-\frac{d+e\,x}{d-\frac{e\,(b+q)}{2\,c}}\right)^p dx \longrightarrow$$

$$\frac{\left(a+b\,x+c\,x^2\right)^p}{e\left(1-\frac{d+e\,x}{d-\frac{e\,(b-q)}{2\,c}}\right)^p\left(1-\frac{d+e\,x}{d-\frac{e\,(b+q)}{2\,c}}\right)^p}\,\mathrm{Subst}\left[\int x^m\left(1-\frac{x}{d-\frac{e\,(b-q)}{2\,c}}\right)^p\left(1-\frac{x}{d-\frac{e\,(b+q)}{2\,c}}\right)^p dx,\ x,\ d+e\,x\right]$$

- Program code:

```
Int[(d_.+e_.*x_)^m_*(a_.+b_.*x_+c_.*x_^2)^p_,x_Symbol] :=
  With[{q=Rt[b^2-4*a*c,2]},
  (a+b*x+c*x^2)^p/(e*(1-(d+e*x)/(d-e*(b-q)/(2*c)))^p*(1-(d+e*x)/(d-e*(b+q)/(2*c)))^p)*
    Subst[Int[x^m*Simp[1-x/(d-e*(b-q)/(2*c)),x]^p*Simp[1-x/(d-e*(b+q)/(2*c)),x]^p,x],x,d+e*x]] /;
FreeQ[{a,b,c,d,e,m,p},x] && NeQ[b^2-4*a*c,0] && NeQ[c*d^2-b*d*e+a*e^2,0] && NeQ[2*c*d-b*e,0] && Not[IntegerQ[p]]
```

```
Int[(d_+e_.*x_)^m_*(a_+c_.*x_^2)^p_,x_Symbol] :=
  With[{q=Rt[-a*c,2]},
  (a+c*x^2)^p/(e*(1-(d+e*x)/(d+e*q/c))^p*(1-(d+e*x)/(d-e*q/c))^p)*
    Subst[Int[x^m*Simp[1-x/(d+e*q/c),x]^p*Simp[1-x/(d-e*q/c),x]^p,x],x,d+e*x]] /;
FreeQ[{a,c,d,e,m,p},x] && NeQ[c*d^2+a*e^2,0] && Not[IntegerQ[p]]
```

S: $\int (d+e\,u)^m \left(a+b\,u+c\,u^2\right)^p \, dx$ when $u == f+g\,x$

- **Derivation: Integration by substitution**
- **Rule 1.2.1.2.S: If $u == f+g\,x$, then**

$$\int (d+e\,u)^m \left(a+b\,u+c\,u^2\right)^p \, dx \longrightarrow \frac{1}{g}\,\text{Subst}\left[\int (d+e\,x)^m \left(a+b\,x+c\,x^2\right)^p \, dx,\ x,\ u\right]$$

- **Program code:**

```
Int[(d_.+e_.*u_)^m_.*(a_+b_.*u_+c_.*u_^2)^p_.,x_Symbol] :=
  1/Coefficient[u,x,1]*Subst[Int[(d+e*x)^m*(a+b*x+c*x^2)^p,x],x,u] /;
FreeQ[{a,b,c,d,e,m,p},x] && LinearQ[u,x] && NeQ[u,x]
```

```
Int[(d_.+e_.*u_)^m_.*(a_+c_.*u_^2)^p_.,x_Symbol] :=
  1/Coefficient[u,x,1]*Subst[Int[(d+e*x)^m*(a+c*x^2)^p,x],x,u] /;
FreeQ[{a,c,d,e,m,p},x] && LinearQ[u,x] && NeQ[u,x]
```

```
(* IntQuadraticQ[a,b,c,d,e,m,p,x] returns True iff (d+e*x)^m*(a+b*x+c*x^2)^p is integrable wrt x in terms of non-Appell functions.
IntQuadraticQ[a_,b_,c_,d_,e_,m_,p_,x_] :=
  IntegerQ[p] || IGtQ[m,0] || IntegersQ[2*m,2*p] || IntegersQ[m,4*p] ||
  IntegersQ[m,p+1/3] && (EqQ[c^2*d^2-b*c*d*e+b^2*e^2-3*a*c*e^2,0] || EqQ[c^2*d^2-b*c*d*e-2*b^2*e^2+9*a*c*e^2,0])
```